
第 1章

本書について

本書はプログラミング⾔語 ClojureScript を扱います。ClojureScript での開発ツールの⼿引書と
して利⽤できます。ClojureScript を⽤いた⽇常的な開発で役に⽴つ内容を紹介します。
本書はプログラミングの⼊⾨書ではありません。少なくとも 1 つ以上プログラミング⾔語の経験

がある読者を想定しています。ただし、Clojure や関数型⾔語の経験は想定していません。誰にも馴
染みのない ClojureScript の理論的な基盤について⾔及するときは参考⽂献を紹介します。

ClojureScript の公式ドキュメントは良質ですが少ないので、私たちは ClojureScript ⼊⾨者に役
⽴つ参考情報と広範な例、そして⼀連の実⽤的な⼿法をまとめた全集を書こうと思いました。本書
は ClojureScript ⾔語と共に進化します。⾔語機能の参考⽂献として、また実⽤的なプログラミン
グ例を含む⼿引き書として利⽤できます。
次の条件を満たしていれば、この本から多く学べるでしょう。

- ClojureScript か関数型プログラミングに関⼼があり、経験が少しある。

- JavaScript または JavaScript にコンパイルされる⾔語を書いていて、ClojureScript が提供する
機能について知りたい。

- Clojure をある程度知っていて、ClojureScript とどう違うのか、さらに同じコードベースで両⽅
の⾔語をターゲットにする⽅法など、実⽤的なトピックを学びたい。

どの条件にも合わなかったとしても、気を落とさないでください。。本書を試し読みしていただ
き、どうしたらもっと親しみやすいものになるか、ご意⾒をいただけるとありがたいです。私たち
の⽬標は、ClojureScript を新しい⼈にも親しみやすいものにし、Clojure が普及に貢献したプログ
ラミングに関する考え⽅を広めることです。
翻訳版としては、ウクライナ語版*1と⽇本語版があります。

*1 https://lambdabooks.github.io/clojurescript-unraveled

第 2章

はじめに

なぜこれをしてるかだって？ Clojure が最⾼で JavaScript がリーチしてるからさ。
- Rich Hickey

ClojureScript は JavaScript をターゲットにした Clojure の実装です。JavaScript をターゲット
にすることにより、ClojureScript で書かれたプログラムは、Webブラウザや Node.js、また io.js
や Nashorn など、様々な実⾏環境で動作します。
他の JavaScript へコンパイルされる⾔語（TypeScript、FunScript、CoffeeScript 等）とは異な
り、ClojureScript は JavaScript をバイトコードのように使います。ClojureScript は関数型プロ
グラミングを受け⼊れており、デフォルトでとても安全で⼀貫性があります。そのセマンティクス
は、JavaScript と全く異なります。
もう⼀つの⼤きな違いは (私たちの⾒解では⻑所ですが)、Clojure がゲスト⾔語になるように設
計されていることです。Clojure はそれ⾃体に仮想マシンを持たない⾔語であり、実⾏環境の違いに
応じて順応させることができます。これにより、Clojure（ClojureScript）はホスト⾔語で書かれた
全ての既存ライブラリにアクセスすることができます。
詳細に⼊る前に、ClojureScript が提供する素晴らしいアイデアのうち、中⼼的なアイデアをまと
めます。全てを理解できなかったとしても⼤丈夫です。本書を読み進めるにつれて、各々の内容が
明確になっていきます。

- ClojureScript では、デザインの決定とイディオムにおいて、関数型プログラミングのパラダイム
に従う必要があります。ClojureScript は関数型プログラミングにこだわりますが、理論的な純粋さ
を追求するような⾔語ではなく、実⽤的な⾔語です。

- ClojureScript では、不変なデータでプログラムを書くことが推奨されます。この不変なデータは
パフォーマンスが⾼く、最新の⼿法で実装されています。

- ClojureScript では、identity と state を明確に区別します。変更を不変な値の⼀連の流れとして
管理するために、明⽰的に状態を⽣成します。

第 2章はじめに

- ClojureScript には、値に基づくポリモーフィズムと、型に基づくポリモーフィズムがあり、
Expression Problem をシンプルに解決することができます。

- ClojureScript は Lisp ⽅⾔なので、プログラムはプログラミング⾔語⾃体のデータ構造で書くこ
とができます。同図像性として知られる性質により、メタプログラミング (つまりプログラムを書く
プログラム)を可能な限りシンプルに⾏うことができます。

以上の概念を合わせることにより、たとえ ClojureScript を使わないとしても、ソフトウェアの
実装⽅法は⼤きな影響を受けるでしょう。関数型プログラミング、(不変な)データと値の変更の分
離、変更を管理する明⽰的なイディオム、抽象化へのプログラミングのためのポリモーフィズムの
構成により、私たちが書くシステムをよりシンプルにすることができます。

シンプルなプログラミング⾔語、ツール、テクニック、アプローチを使うことで、⽇々開発し
ているソフトウェアと全く同じものを作ることができるのだ。
- Rich Hickey

私たちが ClojureScript から感じた楽しみやインスピレーションをこの本から受けていただけれ
ば幸いです。

第 3章

ClojureScript ⼊⾨

本章では、ClojureScript の概要を⾒ていきます。Clojure の知識は想定していません。4 章以降
を理解するために必要なことを簡潔に説明します。オンラインの REPL*1 でコードの動作を確認す
ることができます。

3.1 Lisp 構⽂の第⼀歩
Lisp は 1958 年に John McCarthy ⽒ によって発明された最も古いプログラミング⾔語の 1

つです。現在も様々な分野で利⽤されており、Lisp ⽅⾔と呼ばれる多くの派⽣が進化してきま
した。CloureScript も Lisp ⽅⾔の 1 つです。元々 Lisp ではリストは丸括弧で囲われますが、
Clojure(Script) では独⾃のデータ構造が実装されており、読み書きが容易です。

ClojureScript では、リストの 1 番⽬に関数がある場合、関数を呼び出すために使われます。次の
例では加算する関数を 3 つの引数に適⽤させています。他の⾔語とは異なり、+ は演算⼦ではなく
関数であることに注意してください。Lisp には演算⼦はなく、関数しかありません。

(+ 1 2 3)
;; => 6

上の例では、加算するための関数である + を 1, 2, 3 に適⽤しています。ClojureScript では ?

や - のような⽂字をシンボルの名前として利⽤できます。このことにより、ClojureScript で書か
れたプログラムは読みやすくなっています。

(zero? 0)
;; => true

関数の呼び出しとデータとを区別するために、リストの前にクオート ’ をつけます。クオートが
つけられたリストは、関数の呼び出しではなく、データとして処理されます。

*1 https://www.ClojureScript.io

’(+ 1 2 3) ;; => (+ 1 2 3)

ClojureScript ではリスト以外の構⽂も使います。詳細な説明は後の章で⾏いますが、ベクタの使
い⽅を例として取り上げます。ベクタは⾓括弧 [ ] で囲み、ローカルな束縛を定義します。

(let [x 1
y 2
z 3]

(+ x y z))
;; => 6

以上が ClojureScript でプログラムを書く上で必要となる構⽂の全てです。他の Lisp ⽅⾔でプ
ログラムを書く場合も同様です。独⾃のデータ構造でプログラミング⾔語が実装されていることは、
構⽂に統⼀性があり simple なので、素晴らしい特徴だといえます。マクロでコードを⽣成すること
も他の⾔語と⽐べて簡単であり、私たちのニーズに合わせて⾔語を拡張するのに⼗分な⼒を与えて
くれます。

3.2 基本的なデータ型
他のプログラミング⾔語と同様、ClojureScript には多くのデータ型が⽤意されています。Clo-

jureScript には、数値、⽂字列、浮動⼩数などのスカラー型のデータ構造があります。これら以外
にも、シンボル、キーワード、正規表現、変数、アトム、Volatile などのデータ構造もあります。

ClojureScript はホスト⾔語を採⽤し、可能な限りホスト⾔が提供する型を使⽤します。例えば、
数値と⽂字列は、JavaScript の数値と⽂字列と同じように利⽤して、結果を得ることができます。

数値
ClojureScript において、数値は整数と浮動⼩数点数を含みます。ClojureScript は JavaScript に

コンパイルされる⾔語だということを覚えておいてください。整数は JavaScript では浮動⼩数点数
になります。他の⾔語と同じように、ClojureScript で数値は次のように表現されます。

23
+23
-100
1.7
-2
33e8
12e-14
3.2e-4

11

キーワード
ClojureScript においてキーワードは、常にキーワードへ評価されるオブジェクトです。マップの
データ構造でキーを効率よく表すために使われます。

:foobar  ;; => :foobar
:2     ;; => :2
:?     ;; => :?

キーワードはリテラルとしてコロン : を使います。コロンはオブジェクトの名前の⼀部ではあり
ません。keyword 関数によりキーワードを作成することも可能です。以下の例は、今は理解できな
くても構いません。後に関数の章で取り上げます。

(keyword "foo")
;; => :foo

キーワードをダブルコロン :: で始めると、キーワードの前に現在の名前空間が付きます。名前
空間付きのキーワードは、等価性の⽐較に影響を与えるので注意してください。

::foo
;; => :cljs.user/foo

(= ::foo :foo)
;; => false

別の⽅法としては、リテラルに名前空間を含める⽅法もあります。この⽅法は、他の名前空間に
名前空間付きのキーワードを作成する際に役に⽴ちます。

:cljs.unraveled/foo
;; => :cljs.unraveled/foo

keyword 関数は 2 つの引数をとることができます。第 1 引数で名前空間を指定します。

(keyword "cljs.unraveled" "foo")
;; => :cljs.unraveled/foo

12

シンボル
ClojureScript のシンボルはキーワードとよく似ていますが、⾃分⾃⾝に対して評価されるのでは
なく、シンボルが参照している対象 (関数や変数等)に対して評価されます。
シンボルの名前はアルファベットで始まり、記号 (*, +, !, -, _, ’, ?) を含むことができますが、

数字で始めることはできません。

sample-symbol
othersymbol
f1
my-special-swap!

理解できなかったとしても気にしないでください。シンボルは様々な例で使われており、読み進
むにつれて理解が深まります。

⽂字列
⽂字列については特に説明することはありません。ClojureScript は、他の⾔語と同じように動作
します。ただし、⽂字列が不変である点に注意してください。この点は JavaScript と同じです。

"An example of a string"

ClojureScript の⽂字列で特異な点は、Lisp の構⽂に由来するものです。⽂字列の構⽂は 1 ⾏で
も複数⾏でも同じです。

"This is a multiline
string in ClojureScript."

⽂字
リテラルで⽂字を書くこともできます。

\a ; ⼩⽂字の⽂字
\newline ; 改⾏⽂字

ホスト⾔語の JavaScript では⽂字のリテラルを使えないので、ClojureScript の⽂字は、
JavaScript の⽂字列の 1 ⽂字に変換されます。

コレクション
あるプログラミング⾔語について詳しく知るためには、その⾔語でコレクションがどのように抽
象化されているかを知る必要があり、ClojureScript にも当てはまります。ClojureScript のコレク
ションには多くの型があります。ClojureScript が他の⾔語のコレクションと違うのは、コレクショ
ンが永続的で不変であることです。詳しい説明に⼊る前に ClojureScript のコレクションに存在す
る型について概要を⾒ましょう。

■リスト リストは Lisp 系⾔語で典型的なコレクションの型です。リストは ClojureScript で最も
シンプルなコレクションです。リストはどの型の要素を含むことができます。リストに他のコレク
ションを含むこともできます。リストは要素を丸括弧で囲みます。

’(1 2 3 4 5)
’(:foo :bar 2)

上記の例では、リストの先頭にクオート ’ がついています。これは Lisp 系⾔語では、リストが
関数やマクロの呼び出しに使われるためです。リストを関数やマクロの呼び出しに使う場合、初め
の要素は関数やマクロのシンボル、残りの要素はその引数である必要があります。上の例ではリス
トの先頭にクオートがついているため、要素のリストとして扱われます。
リスト先頭にクオート ’ があるかどうかで、動作がどう違うかを確認しましょう。

(inc 1)
;; => 2

’(inc 1)
;; => (inc 1)

先頭にクオートをつけずに (inc 1) として評価した場合、inc シンボルは inc 関数として評価さ
れます。1 が第 1 引数として評価されて、2 という値が返ります。
リストを作るために、明⽰的に list 関数を使うことも可能です。

(list 1 2 3 4 5)
;; => (1 2 3 4 5)

(list :foo :bar 2)
;; => (:foo :bar 2)

リストは、前から順に要素にアクセスするには効率よく処理できますが、リスト内の要素に対し
てランダムに (もしくは特定のインデックスに)アクセスするのは不得意です。

■ベクタ リストのようにベクタも連続する値を保存しますが、インデックスを⽤いて効率よく要
素にアクセスできます。要素が前から順に評価されるリストと対照的です。後に詳しく説明します
が、現状ではこの程度の理解で⼗分です。ベクタはリテラルのために⾓括弧を使います。

[:foo :bar]
[3 4 5 nil]

リストと同様に、ベクタはどの型のオブジェクトを含むことができます。vector 関数を⽤いてベ
クタを作成できますが、あまり⼀般的ではありません。

(vector 1 2 3)
;; => [1 2 3]

(vector "blah" 3.5 nil)
;; => ["blah" 3.5 nil]

■マップ マップはコレクションの 1 つであり、キーと値をセットで保存します。他の⾔語では、
連想配列や辞書と⾔われるデータ構造です。

{:foo "bar", :baz 2 }
{:alphabet [:a :b :c]}

キーと値を分けるためにコンマを使うかどうかは任意です。ClojureScript の構⽂では、コンマは
スペースとして処理されます。
ベクタのように、マップリテラルの各々の要素は、結果がマップに保存される前に毎回評価され
ます。ただし、要素が順に評価されるかは保証されません。

■セット 最後はセットです。セットは任意の型のデータを 0 個以上保存することができますが、
要素間に順序はありません。セットのためには、マップのように波括弧 { } を使います。波括弧の
前に シャープ # をつける点がマップとは異なります。set関数を⽤いてコレクションをセットに変
換することもできます。

#{1 2 3 :foo :bar}
;; => #{1 :bar 3 :foo 2}

(set [1 2 1 3 1 4 1 5])
;; => #{1 2 3 4 5}

後のセクションでは、セットと他のコレクションの型について詳しくみます。

3.3 vars

ClojureScript は関数型プログラミング⾔語であり、不変性に重点をおいています。そのため、他
の⾔語の変数にあたる概念がありません。代数学における変数にあたる概念はあります。つまり数
学で x = 6というときには、xというシンボルが 6 を表すということを意味します。

ClojureScript において var はシンボルで表現されて、メタデータと⼀緒に 1 つの値を保存しま
す。特殊形式の def を使うことで、var を 1 つ定義することができます。

(def x 2 2)
(def y [1 2 3])

var は名前空間において常にトップレベルです。名前空間については後に詳しく説明します。も
し defを関数の呼び出しの中で使う場合、var は名前空間のレベルに定義されますが、これは推奨
されません。関数の中で変数を定義するには letを使うことが推奨されます。

3.4 関数
はじめの⼀歩

ClojureScript には第⼀級関数と呼ばれるものがあります。第⼀級の関数は、他の型と同じように
使えます。例えば、常にレキシカルスコープを重視しながら、関数の引数として渡せたり、関数の
返り値として利⽤することができます。ClojureScript におけるダイナミックスコープについては別
のセクションで扱います。
スコープについて詳しく知るには、http://en.wikipedia.org/wiki/Scope を参考にしてください。

様々なタイプのスコープについて詳細を説明しています。
ClojureScript は Lisp ⽅⾔の 1 つであり、関数の呼び出しは、次のように前置記法を⽤います。

(inc 1)
;; => 2

この例では incは関数であり、実⾏時に利⽤できる関数の 1 つです。この場合、1 は inc関数の
第 1 引数です。

(+ 1 2 3)
;; => 6

+ シンボルは加算を意味します。複数の引数をとることができる点ができ、ALGOL型のプログ
ラミング⾔語が 2 つの引数しかとることができないことと対照的です。
前置記法には⼤きな利点があり、その中には必ずしも明⽩でないものもあります。ClojureScript

では関数とオペレータを区別しません。全てが関数です。すぐに分かる⻑所としては、前置記法では
オペレータが任意の数の引数を許容することです。また、演算⼦の優先順位の問題も解消されます。

⾃作関数の定義
特殊形式の fn を⽤いると、無名関数を定義することができます。次の例では、関数は 2 つの引

数を取り、それらの平均を返します。

(fn [param1 param2]
(/ (+ param1 param2) 2.0))

関数を定義して即時に呼び出すことができます（1 つの式で）。

((fn [x] (* x x)) 5)
;; => 25

次に名前付き関数を作成しましょう。名前付き関数とはどういうことでしょうか。ClojureScript
ではすごくシンプルです。関数は第⼀級クラスであり、他の全ての値と同じように振る舞います。
関数をシンボルに束縛することにより、関数に名前をつけます。

(def square (fn [x] (* x x)))

(square 12)
;; => 144

ClojureScript には、関数定義の糖⾐構⽂として defn マクロがあります。

(defn square
"特定の数字の 2乗を返す"
[x]
(* x x))

関数名と引数の間にある⽂字列は docstring(documentation string) と呼ばれます。Web でド
キュメントを⾃動⽣成するときに docstring が使われます。

多様な Arity をもつ関数
ClojureScript は任意の数の引数をとる関数を定義することができます。( Arity という⽤語は関
数が受ける引数の数を意味します) そのための構⽂は普通の関数定義と⼤体同じですが、複数の本体
をもつ点が異なります。例をみてみましょう。

(defn myinc
"⾃⼰定義版のパラメータ化された inc"
([x] (myinc x 1))
([x increment] (+ x increment)))

([x] (myinc x 1)) は、もし引数が 1 つの場合、myinc 関数に x を第 1 引数、1 を第 2 引数と
して渡すということを表しています。([x increment] (+ x increment)) は、もし引数が 2 つ
の場合、2 つの引数を⾜した結果を返します。ここで定義した関数を使ってみましょう。引数の数
を間違えて関数を呼び出すと、コンパイラからエラーメッセージが出⼒されることに注⽬してくだ
さい。

(myinc 1)
;; => 2

(myinc 1 3)
;; => 4

(myinc 1 3 3)
;; Compiler error

Arityの概念を本書で説明することは、本書の範囲を超えています。詳しくはWikipediaの Arity
のページを参照してください。

可変⻑引数関数
可変⻑引数をとる関数を定義することでも、複数の引数を受け取る関数を定義することができま
す。可変⻑引数をとる関数は、任意の数の引数をとることができます。
可変⻑引数関数を定義するには & シンボルを引数のベクタにつけます。

(defn my-variadic-set
[& params]
(set params))

(my-variadic-set 1 2 3 1)
;; => #{1 2 3}

無名関数の短い構⽂
ClojureScript では、無名関数のためのシンプルな構⽂としてリーダマクロの #() を使うことが

できます (ワンライナーを書くためによく⽤いられます)。リーダーマクロとは、コンパイル時に適
当なフォームに変換される「特別な」表現⽅法です。この場合、fnのフォームに変換されます。

(def average #(/ (+ %1 %2) 2))

(average 3 4)
;; => 3.5

この average の定義は、次のように展開されます。

(def average-longer (fn [a b] (/ (+ a b) 2)))

(average-longer 7 8)
;; => 7.5

%1 や %2 は引数の位置を⽰す印であり、リーダーマクロが fnの式に変換される時に暗黙的に宣
⾔されます。
もし関数が 1 つしか引数を取らない場合、% シンボルの後の数字は省略できます。例えば、#(*

%1 %1) は #(* % %) と書くことができます。
さらに %& シンボルにより可変⻑引数にすることができます。

(def my-variadic-set #(set %&))

(my-variadic-set 1 2 2)
;; => #{1 2}

3.5 フロー制御
ClojureScript のフロー制御は、JavaScript や C などとアプローチが異なります。

if による条件分岐
ClojureScript において if は⽂ではなく式であり、3 つの引数をとります。1 つ⽬は条件式、2

つ⽬は条件が真の場合に評価される式、3 つ⽬は条件が偽の場合に評価される式です。

(defn discount
"1点以上の購⼊で 5%割引されます"
[quantity]
(if (>= quantity 100)

0.05
0))

(discount 30)
;; => 0

(discount 130)
;; => 0.05

if の条件分岐の中で複数の式を実⾏するには do を使います。doについては次のセクションで説
明します。

cond による条件分岐
if式には複数の条件を追加するための else if の部分がないため、多少制限されますが、cond

マクロでこの点を解決できます。condでは複数の条件を定義できます。

(defn mypos?
[x]
(cond

(> x 0) "positive"
(< x 0) "negative"
:else "zero"))

(mypos? 0)  ;; => "zero"

(mypos? -2) ;; => "negative"

また、condには condp と呼ばれる別の形式もあり、単純な condと⾮常によく似た働きをします
が、条件 (述語とも呼ばれます)が全ての条件に対して同じ場合は、より綺麗に⾒えます。
以下の各⾏では、ClojureScript は (keyword code) を評価した結果に = 関数を適⽤します。

(defn translate-lang-code
[code]
(condp = (keyword code)

:es "Spanish"
:en "English"
"Unknown"))

(translate-lang-code "en")
;; => "English"

(translate-lang-code "fr")
;; => "Unknown"

case による条件分岐
case は、前例の condp と同様の使い⽅をします。主な違いは、常に = 述語 (関数) が使⽤され

て、その分岐値がコンパイル時に評価されることです。そのため condや condpよりもパフォーマ
ンスが良いですが、条件の値が静的でなければならないという⽋点があります。

(defn translate-lang-code
[code]
(case code

"es" "Spanish"
"en" "English"
"Unknown"))

(translate-lang-code "en")
;; => "English"

(translate-lang-code "fr")
;; => "Unknown"

3.6 真偽判定
真偽判定はプログラミング⾔語によりセマンティクスが異なります。多くの⾔語では、空のコレ
クションや整数の 0 、またこれらと同様の値を偽と⾒なします。ClojureScript では nil と false

だけが偽と⾒なされ、その他は論理的に真として扱われます。
callable プロトコル (IFn については後述)を実装する能⼒と⼀緒に、集合のようなデータ構造は、
関数の中でそれらを追加的にラップする必要なしに、述語として使うことができます。

(def valid? #{1 2 3})

(filter valid? (range 110))
;; => (1 2 3)

このように動作するのは、セットが全ての要素の値⾃体か nil を返すためです。

(valid? 1)
;; => 1

(valid? 4)
;; => nil

3.7 ローカル、ブロック、ループ
ローカル

ClojureScript には ALGOL のような変数の概念がありませんが、ローカル (local)はあります。
ローカルは不変であり、変更しようとするとエラーが発⽣します。
ローカルは let 式で定義されます。let式は、最初のパラメータとしてベクタで始まり、その後
に任意の数の式が続きます。最初のパラメータのベクタには束縛フォームを与えて、その let 内の
ローカルで有効な名前と値のペアを宣⾔します。

(let [x (inc 1)
y (+ x 1)]

(println "Simple message from the body of a let")
(* x y))

上の例では、シンボル x が (inc 1) の値に束縛されて、シンボル y が x と 1 の合計 (つまり 3
)に束縛されます。これらの束縛を受けて、 (println "Simple message from the body of a

let") と (* x y) が評価されます。

ブロック
JavaScript において波括弧 { } は「共に属する」コードのブロックを決めます。ClojureScript

では do を⽤いてブロックを作成します。doはコンソールの何かの結果を出⼒したり、ログを出⼒
したりするような、副作⽤を伴う場合に使います。副作⽤とは、戻り値には不要なものをいいます。

doは任意の数の式を含むことができますが、最後に評価された式の値が返り値となります。

(do
(println "hello world")
(println "hola mundo")
;; この値は返却されずに捨てられます。
(* 3 5)
(+ 1 2))

;; hello world
;; hola mundo
;; => 3

先ほど説明した let の本体は、複数の式をもつことができる点において do とよく似ています。
実際、let は暗黙の do をもちます。

ループ
ClojureScript は関数型のアプローチを採⽤しているので、JavaScript での for のような⼀般的

なループがありません。ClojureScript のループは再帰を使って処理されます。再帰を⽤いてプログ
ラムを書くためには、命令型のプログラミング⾔語とは少し違った⽅法で問題をモデル化する必要
があります。
他の⾔語で for が使われるパターンの多くは、⾼階関数を⽤いて置き換えることができます。⾼

階関数とは関数を引数として受け取る関数です。

■loop/recurによるループ では、loop と recur を⽤いた再帰でループを表現する⽅法を⾒てみ
ましょう。loop は、空の可能性のある束縛のリスト (letとの対称性に注⽬してください)を定義
して、繰り返し実⾏すると、それらの束縛の新しい値を使ってループの開始点に戻ります。例を⾒
てみましょう。

(loop [x 0]
(println "Looping with " x)
(if (= x 2)

(println "Done looping!")
(recur (inc x))))

;; Looping with 0
;; Looping with 1
;; Looping with 2
;; Done looping!
;; => nil

上の例では、まず x を 0 に束縛して本体を実⾏します。if の条件が満たされていないので、
(recur (inc x)) で inc で x に 1 が加算されてループが再実⾏されます。条件が満たされると
recur の呼び出しが⽌まり、ループが終了します。

recur を使うことができるのは loop だけではありません。関数の中で recur を使⽤すると、新
しい束縛値で本体が再帰的に実⾏されます。

(defn recursive-function
[x]
(println "Looping with" x)
(if (== x 2)

(println "Done looping!")
(recur (inc x))))

(recursive-function 0)
;; Looping with 0
;; Looping with 1
;; Looping with 2
;; Done looping!
;; => nil

■⾼階関数によるループの置き換え 命令型プログラミング⾔語では、for で反復的にデータを変
形しますが、次のような⽬的で利⽤することが多いです。命令型プログラミング⾔語では、for ルー
プを使⽤してデータを繰り返し変換することが⼀般的です。通常、次のいずれかを⽬的とします。

- イテラブルの全ての値を、別のイテラブルに変形する
- ある条件でイテラブルの要素をフィルターする
- イテラブルの要素を順に処理をして 1 つの値に変換する
- イテラブルにある各々の値を計算する

上記の実⾏は、⾼階関数と ClojureScript の構⽂を⽤いて書くことができます。まず最初の 3つ
の例を⾒ていきましょう。
イテラブルにある全ての値を変換するには map 関数を使います。map 関数は、関数とシーケンス
を引数にとり、関数をシーケンスの各要素に適⽤します。

(map inc [0 1 2])
;; => (1 2 3)

map 関数の初めのパラメータには、1 つの引数をとり 1 つの値を返す関数を指定します。例えば、
グラフ作成のアプリケーションがあるとします。y = 3x + 5 の式を x の値のセットに対して適⽤
して y の値のセットをえるには、次のように書きます。

(defn y-value [x] (+ (* 3 x) 5))

(map y-value [1 2 3 4 5])
;; => (8 11 14 17 20)

⾼階関数に渡す関数が短い場合は、無名関数を使うこともできます。無名関数には fn か # のど
ちらを⽤いても構いません。

(map (fn [x] (+ (* 3 x) 5)) [1 2 3 4 5])
;; => (8 1 1 1 4 1 7 2 0)

(map #(+ (* 3 %) 5) [1 2 3 4 5])
;; => (8 1 1 1 4 1 7 2 0)

データ構造の値をフィルターするには、filter 関数を使います。filter 関数は、述語とシーケ
ンスをとり、述語に対して真を返す要素だけを持つ新しいシーケンスを⽣成します。

(filter odd? [1 2 3 4])
;; => (1 3)

ここでも、filter 関数に適⽤する最初の引数として true または false を返す任意の関数を
使⽤できます。5 ⽂字未満の単語のみを保つ例をみてみましょう (count 関数は引数の⻑さを返し
ます)。

(filter (fn [word] (< (count word) 5)) ["ant" "baboon" "crab" "duck" "echidna" "fox"])
;; => ("ant" "crab" "duck" "fox")

イテラブルの要素を 1 つずつ処理をして蓄積しながら 1 の値を返すには reduce を⽤います。
reduce は値を蓄積するための関数、初期値、コレクションを取りますが、初期値については任意
です。

(reduce + 0 [1 2 3 4])
;; => 10

また、reduce の最初の引数として⾃作の関数を指定することもできますが、その関数には 2 つの
引数が必要です。最初の引数は「蓄積値」で、 2番⽬の引数は処理されるコレクションのアイテムで
す。この関数は、リスト内の次のアイテムのアクミュレータになる値を返します。例えば、数値の
集合の平⽅和を求めましょう（統計では重要な計算です）。関数を分ける場合は次のようにします。

(defn sum-squares
[accumulator item]
(+ accumulator (* item item)))

(reduce sum-squares 0 [3 4 5])
;; => 50

無名関数を使う場合は次のようにします。

(reduce (fn [acc item] (+ acc (* item item))) 0 [3 4 5])
;; => 50

単語セット内の総⽂字数を検出する reduce 関数を次に⽰します。

(reduce (fn [acc word] (+ acc (count word)))
0 ["ant" "bee" "crab" "duck"])

;; => 14

上の例では、短い構⽂を使っていません。なぜなら、短い構⽂を使うとタイピング数を減らせます
が、コードの可読性が低下する可能性があるからです。新しい⾔語を始めるときは、⾃分が書いたも
のを後で読めることは重要です。もし短い構⽂に慣れている場合は、この構⽂を使⽤してください。
アキュムレータに渡す開始値は慎重に選ぶ必要があることを覚えておいてください。⼀連の数の
積を求めるために reduce を使うときは、0 ではなく 1 を開始値として使う必要があります。0 を
⽤いて掛け算をすることができないためです。

;; wrong starting value
(reduce * 0 [3 4 5])  ;; => 0

;; correct starting accumulator
(reduce * 1 [3 4 5])  ;; => 60

■for によるシーケンス内包表記 ClojureScript の for は、反復のためではなくシーケンスを⽣
成するために⽤いられます。これはシーケンス内包表記として知られています。このセクションで
は、シーケンス内包表記がどのように動作するかを学び、シーケンスを宣⾔的に構築するためにシー
ケンス内包表記を使います。

for は束縛のためのベクタと式をとり、式が評価された結果をシーケンスとして⽣成します。

(for [x [1 2 3]]
[x (* x x)])

;; => ([1 1] [2 4] [3 9])

この例では、x は [1 2 3] の各要素に順に束縛され、元の要素と それを 2 乗した値をもつ 2 要
素ベクタの新しいシーケンスを返します。

forは複数の束縛をサポートしているため、命令型⾔語での for ループのネストと同様に、コレ
クションがネストされて反復されます。最も内側の束縛が「最も速く」繰り返されます。

(for [x [1 2 3]
y [4 5]]

[x y])
;; => ([1 4] [1 5] [2 4] [2 5] [3 4] [3 5])

さらに 3 つの修飾⼦があります。:let はローカルな束縛の作成のために、:while はシーケンス
⽣成からの脱出のために、:when は値のフィルタリングのために⽤います。
次は :let を⽤いたローカル束縛の例です。:let で定義された束縛はその式でのみ利⽤可能
です。

(for [x [1 2 3]
y [4 5]
:let [z (+ x y)]]

z)
;; => (5 6 6 7 7 8)

:while 修飾⼦は、条件が合わなくなったときにシーケンスの⽣成を⽌める条件を表すために使い
ます。次の例を⾒てください。

(for [x [1 2 3]
y [4 5]
:while (= y 4)]

[x y])
;; => ([1 4] [2 4] [3 4])

⽣成された値をフィルターするには、次のように :when 修飾⼦を使います。

(for [x [1 2 3]
y [4 5]
:when (= (+ x y) 6)]

[x y])
;; => ([1 5] [ 2 4])

上記のような修飾語を組み合わせることで、複雑な配列の⽣成を表現したり、理解の意図をより
明確に表現することができます。

(for [x [1 2 3]
y [4 5]
:let [z (+ x y)]
:when (= z 6)]

[x y])
;; => ([1 5] [2 4])

命令型プログラミング⾔語における for 構⽂の最も⼀般的な使⽤⽅法を説明した際に、結果を気
にせずにシーケンスのすべての値に対して計算を実⾏したい場合があると述べました。おそらく、
シーケンスの値を⽤いて何らかの副次的効果を得るために、このようなことをするのでしょう。

同様の⽬的を達成するために、 ClojureScript には doseq 構⽂があります。doseq は for と似
ていますが、式を実⾏した後、結果の値を捨てて nil を返します。

(doseq [x [1 2 3]
y [4 5]

:let [z (+ x y)]]
;; 1 + 4 = 5
;; 1 + 5 = 6
;; 2 + 4 = 6
;; 2 + 5 = 7
;; 3 + 4 = 7
;; 3 + 5 = 8
;; => nil

コレクション内の各アイテムに対して副作⽤のある操作 (println 等) を効率よく繰り返し適⽤
したい場合は、それに特化した関数 run! を使います。内部的には⾼速に reduce を使⽤します。

(run! println [1 2 3])
;; 1
;; 2
;; 3
;; => nil

この関数は明⽰的に nil を返します。

3.8 コレクションの型
不変性と永続性

ClojureScript のコレクションは永続的で不変だと⾔及しましたが、詳しく説明していませんでし
た。不変のデータ構造とは、変更ができないデータ構造です。不変なデータ構造においては、部分
的な更新は許されません。conj を⽤いてベクタに値を付け加える例を⾒てみましょう。

(let [xs [1 2 3]
ys (conj xs 4)]

(println "xs:" xs)
(println "ys:" ys))

;; xs: [1 2 3]
;; ys: [1 2 3 4]
;; => nil

上の例では xs に要素を 1 つ追加して新たなベクタ ys を作成しています。ベクタ xs は不変なの
で変更されません。
永続的なデータ構造とは、変換時に元のバージョンを残したまま新しいバージョンを返すデータ
構造のことです。ClojureScript では、構造共有 (structural sharing) と呼ばれる実装技術を⽤い
て、メモリと時間の効率化を図っており、2 つのバージョンの値の間で共有されるデータのほとん
どは重複しません。
構造共有がどのように動作しているかを知りたい場合は読み進めてください。もし興味がない場
合は、読み⾶ばして次のセクション (シーケンスの抽象化)に進んでください。

ClojureScript におけるデータ構造の構造共有を説明するために、古いデータ構造と新しいデータ
構造の⼀部が同じオブジェクトかどうかを identical? を⽤いて⽐較してみます。ここではリスト
を例に説明します。

(let [xs (list 1 2 3)
ys (cons 0 xs)]

(println "xs:" xs)
(println "ys:" ys)
(println "(rest ys):" (rest ys))
(identical? xs (rest ys)))

;; xs: (1 2 3)
;; ys: (0 1 2 3)
;; (rest ys): (1 2 3)
;; => true

リストの xs に cons 関数を使って値を追加して新たなリスト ys を作成しています。リスト ys

の rest はリスト xs とメモリでは等しくなります。この挙動を⾒ると、リスト xs と リスト ys が
データ構造を共有していることがわかります。

シーケンスの抽象化
ClojureScript による抽象化で中⼼的な概念の 1 つに シーケンス (sequence) があります。シー

ケンスはリストとして⾒なされ、どのコレクションの型もシーケンスとして⾒なすことができます。
シーケンスは全てのコレクションの型のように永続的で変更不可能です。⼤半の ClojureScript の
関数はシーケンスを返します。
シーケンスを⽣成するために使うことができる型はシーカブル (seqables)と呼ばれます。seq を

シーカブルに対して呼び出して、シーケンスをえることができます。シーケンスは 基本的な関数
first と rest をサポートします。どちらも first と rest に与える引数に対して seq を呼び出
します。

(first [1 2 3])
;; => 1

(rest [1 2 3])
;; => (2 3)

seq 関数をシーカブルに対して呼ぶとき、シーカブルが空かどうかで結果が異なります。空の場
合には nil を、そうでなければシーケンスを返します。

(seq [])
;; => nil

(seq [1 2 3])
;; => (1 2 3)

next は rest と似たシーケンス操作関数ですが、シーケンスの要素が空か 1 つの場合に nil を
返す点が rest とは異なります。前述のシーケンスの 1 つが与えられたとき、rest により返される
空のシーケンスは論理値で真を返します。⼀⽅、next により返される nil の値は偽と評価されま
す。真偽判定の章で詳しく説明します。

(rest [])
;; => ()

(next [])
;; => nil

(rest [1 2 3])
;; => (2 3)

(next [1 2 3])
;; => (2 3)

■nil パンニング seq はコレクションが空のときに nil を返し、nil は論理値として偽と評価さ
れるので、コレクションが空かどうかを seq 関数を⽤いて確認することができます。このようなテ
クニックは nil パンニングと呼ばれます。

(defn print-coll
[coll]
(when (seq coll)

(println "Saw " (first coll))
(recur (rest coll))))

(print-coll [1 2 3])
;; Saw 1
;; Saw 2
;; Saw 3
;; => nil

(print-coll #{1 2 3})
;; Saw 1
;; Saw 3
;; Saw 2
;; => nil

nil は シーカブル でもシーケンスでもありませんが、これまで⾒てきた全ての関数でサポートさ
れています。

(seq nil)
;; => nil

(first nil)
;; => nil

(rest nil)
;; => ()

■シーケンス操作関数 ClojureScript コレクションを変換するための中⼼的な関数は、引数から
シーケンスの⽣成を⾏い、前章で学んだ⼀般的なシーケンス操作関数において使えるように実装さ
れています。このことにより、それらの関数は、全ての seqable なデータ型に対して使うことがで
きます。次の例で map 関数が様々な seqables に対して使えることを⾒てみましょう。

(map inc [1 2 3])
;; => (2 3 4)

(map inc #{1 2 3})
;; => (2 4 3)

(map count {:a 41 :b 40})

(map inc ’(1 2 3))
;; => (2 3 4)

map 関数をマップのコレクションに対して使うとき、⾼階関数はキーと値を含む 2 つのアイテム
からなるベクタを受け取ります。次の例では、キーと値にアクセスするために destructuring を使
います。

(map (fn [[key value]] (* value value))
{:ten 10 :seven 7 :four 4})

;; => (100 49 16)

値のシーケンスを取得するために限れば、同じ操作をより慣⽤的な⽅法で⾏うことも可能です。

(map (fn [value] (* value value))
(vals {:ten 10 :seven 7 :four 4}))

;; => (100 49 16)

気がついたかもしれませんが、シーケンスを処理する関数は、空のシーケンスを返すだけでいい
ので、空のコレクションや nil 値であっても安全に使⽤することができます。

(map inc [])  ; ==> ()

(map inc #{}) ; ==> ()

(map inc nil) ; ==> ()

これまで map, filter, reduce などの⼀般的な使⽤例はすでに紹介しましたが、 ClojureScript
の core 名前空間には、さらに多くの汎⽤のシーケンス操作関数が⽤意されています。ここで説明す
る操作の多くは、シーカブルで動作するか、もしくはユーザ定義型に拡張可能です。

coll? 述語を⽤いて、データがコレクション型かどうかを判別できます。

(coll? nil) 
;; => false

(coll? [1 2 3])
;; => true

(coll? {:language "ClojureScript" :file-extension "cljs"})
;; => true

(coll? "ClojureScript")
;; => false

同様に、データ構造がシーケンスかどうかを判別する seq? 述語、シーカブルかどうかを判別し
たりする seqable? 述語があります。

(seq? nil)
;; => false
(seqable? nil)
;; => false

(seq? [])
;; => false
(seqable? [])
;; => true

(seq? #{1 2 3})
;; => false
(seqable? #{1 2 3})
;; => true

(seq? "ClojureScript")
;; => false
(seqable? "ClojureScript")
;; => false

⼀定時間内にカウントできるコレクションの場合は、count 操作を使⽤できます。この操作は⽂
字列に対しても機能しますが、これまで⾒てきたように、⽂字列はコレクション、シーケンス、また
はシーカブルではありません。

(count nil)
;; => 0

(count [1 2 3])
;; => 3

(count {:language "ClojureScript" :file-extension "cljs"})
;; => 2

(count "ClojureScript")
;; => 13

また empty 関数を使うことで、与えられたコレクションの空の variant を得ることもできます。

(empty nil)
;; => nil

(empty [1 2 3])
;; => []

(empty #{1 2 3})
;; => #{}

empty? 述語は、コレクションが空の場合に true を返します。

(empty? nil)
;; => true

(empty? [])
;; => true

(empty? #{1 2 3})
;; => false

conj 操作は要素をコレクションに追加しますが、コレクションの型に応じて異なる「場所」に追
加することができます。コレクションの型で最もパフォーマンスが⾼い場所に要素を追加します。
ただし、全てのコレクションに順序が定義されているわけではありません。

conj には追加する要素をいくつでも渡すことができます。実際の動作を⾒てみましょう。

(conj nil 4 2)
;; => (4 2)

(conj [1 2] 3)
;; => [1 2 3]

(conj [1 2] 3 4 5)
;; => [1 2 3 4 5]

(conj ’(1 2) 0)
;; => (0 1 2)

(conj #{1 2 3} 4)
;; => #{1 3 2 4}

(conj {:language "ClojureScript"} [:file-extension "cljs"])
;; => {:language "ClojureScript", :file-extension "cljs"}

■遅延 ClojureScript のシーケンスを返す⼤半の関数は、全く新たなシーケンスを⽣成するのでは
なく、遅延シーケンスを⽣成します。遅延シーケンスは、要求がされたときに内容を⽣成します。通
常は、それらに反復作業を⾏うときに⽣成されます。遅延により、必要以上のことをしないことが
保証して、潜在的な無限シーケンスを正規のシーケンスとして扱うことができます。
整数の範囲を⽣成する range 関数を⾒てみましょう。

(range 5)
;; => (0 1 2 3 4)

(range 1 10)
;; => (1 2 3 4 5 6 7 8 9)

(range 10 100 15)
;; (10 25 40 55 70 85)

(range) だけで呼び出した場合、整数の無限シーケンスを⽣成します。REPLで (range) を試
さないでください。REPLはその式を完全に評価しようとするので、とても⻑い間待たされます。
少し例を⾒てみましょう。グラフ作成のプログラムを書いていて、y = 2 x ^ 2 + 5 のグラフ

を書くとします。y の値が 100未満の場合の x の値を求めます。0 から 100 までの全ての数を⽣
成して、take-while を⽤いて条件に合うものを保ちます。ここに例を⾒てみましょう。グラフ作
成プログラムを作成していて、⽅程式 y = 2 x ^ 2 + 5 をグラフ化していて、x の値のうち、 y

の値が 100 未満の値だけが必要だとします。0 から 100 までの全ての数を⽣成して、条件が満たさ
れている間だけ値をとります。

(take-while (fn [x] (< (+ (* 2 x x) 5) 100))
(range 0 100))

;; => (0 1 2 3 4 5 6)

コレクション詳細
ここまで、ClojureScript におけるシーケンスの抽象化とシーケンス操作関数について理解を深め
ました。次は、コレクション型の詳細と、それらがサポートする関数について⾒ていきます。

■リスト ClojureScript では、シンボルをグループ化してプログラムにするためのデータ構造と
して、主にリストが使⽤されます。他の Lisp ⽅⾔とは異なり、ClojureScript ではリストとは異な
るデータ構造体を使⽤します (ベクタとマップ)。このことによりコードの統⼀感が低下しますが、
コードの読みやすさは向上します。

ClojureScript のリストは、各ノードに値とリストの残りの部分へのポインタが含まれる連結リス
トだと考えることができます。リストの最後に項⽬を追加するにはリスト全体を横断する必要があ
りますが、リストの先頭に項⽬を追加するのは⾃然に (⾼速に)できます。リストの最初に要素を追
加するには、cons 関数を⽤います。

(cons 0 (cons 1 (cons 2 ())))
;; => (0 1 2)

リストのリテラル表現として () を使いました。シンボルを 1 つも含んでいないので、関数の呼
び出しとしては⾒なされません。もし要素がある場合には、式が評価されないようにクオート ’ を
つける必要があります。

(cons 0 ’(1 2))
;; => (0 1 2)

リストコレクションにおいては先頭部は⼀定の時間で追加できる場所なので、リストに対する
conj 操作は要素を⾃然に先頭部に追加します。

(conj ’(1 2) 0)
;; => (0 1 2)

リストと ClojureScript の他のデータ構造において、peek, pop, conj 関数はスタック操作のため
に使えます。スタックの最上部は conj が要素を追加する場所であることから、conj 操作はスタッ
クへのプッシュ操作と同等です。リストの場合、conj は要素をリストの先頭に追加して、peek は
リストの最初の要素を返し、pop は先頭の要素以外の全ての要素を返します。
スタックを操作する conj と pop がスタックのために使われるコレクションの型に変更を加えな
いことに注意してください。

(def list-stack ’(0 1 2))

(peek list-stack)
;; => 0

(pop list-stack)
;; => (1 2)

(type (pop list-stack))
;; => cljs.core/List

(conj list-stack -1)
;; => (-1 0 1 2)

(type (conj list-stack -1))
;; => cljs.core/List

リストが特に不得意なのは、インデックスによるランダムアクセスです。リストはメモリ内で連
結されたリストのような構造体に格納されるので、要求されたアイテムを取り出すためにも、また
インデックスが有効範囲にない場合にエラーを投げるためにも、ランダムに要素にアクセスするに
は先頭から順に検索する必要があります。遅延シーケンスのようなインデックス付けされていない
順序付きコレクションも、この制限を受けます。

■ベクタ ベクタは ClojureScript で最も⼀般的なデータ構造の 1 つです。ベクタは、伝統的な
Lisp ⽅⾔がリストを使⽤する多くの場所、例えば、関数の引数宣⾔や let 束縛での構⽂内で使われ
ます。

ClojureScript のベクタは、リテラル表現として⾓括弧 [] を使います。vector 関数で⽣成する
ことも可能です。他のコレクションからベクタを作成するには vec 関数を使います。

(vector? [0 1 2])
;; => true

(vector 0 1 2)
;; => [0 1 2]

(vec ’(0 1 2))
;; => [0 1 2]

ベクタは様々な種類の値が順に並べられたコレクションです。リストとは異なり、ベクタは末尾
から⾃然に拡張することができ、conj 関数はアイテムをベクタの末尾に追加します。ベクタの末尾
へ要素を追加する操作は、実質的に⼀定時間で⾏えます。

(conj [0 1] 2)
;; => [0 1 2]

36



第 3章 ClojureScript ⼊⾨ 3.8 コレクションの型

リストとベクタを区別するもう 1 つの点は、ベクタはインデックス付きコレクションであり、効
率的なランダムアクセスや⾮破壊更新をサポートする点です。nth 関数を使うと、指定されたイン
デックスの値を取得できます。

(nth [0 1 2] 0)
;; => 0

ベクタは連続した数値キー (インデックス)を値に関連付けるので、連想データ構造として扱うこ
とができます。ClojureScript が提供する assoc 関数は、連想データの構造体とキーと値のペアの
セットを指定すると、変更されたキーに対応する値を持つ新しいデータの構造体を⽣成します。イ
ンデックスは、ベクタの最初の要素の 0 から始まります。

(assoc ["cero" "uno" "two"] 2 "dos")
;; => ["cero" "uno" "dos"]

ベクタにすでに含まれているキー、またはベクタの最後の位置にあるキーにのみ assoc を使うこ
とができます。

(assoc ["cero" "uno" "dos"] 3 "tres")
;; => ["cero" "uno" "dos" "tres"]

(assoc ["cero" "uno" "dos"] 4 "cuatro")
;; Error: Index 4 out of bounds [0,3]

連想データ構造は関数として使うこともできます。これらは、関連付けられている値に対するキー
の関数です。ベクタの場合、指定されたキーが存在しないと例外が発⽣します。

(["cero" "uno" "dos"] 0)
;; => "cero"

(["cero" "uno" "dos"] 2)
;; => "dos"

(["cero" "uno" "dos"] 3)
;; Error: Not item 3 in vector of length 3

ベクタはリストと同様にスタックとして使うことが可能であり、peek, pop, conj を使うことがで
きます。ベクタとリストは要素を追加するときに、先頭と末尾が逆であることに注意してください。

(def vector-stack [0 1 2])

(peek vector-stack)
;; => 2

(pop vector-stack)
;; => [0 1]

(type (pop vector-stack))
;; => cljs.core/PersistentVector

(conj vector-stack 3)
;; => [0 1 2 3]

(type (conj vector-stack 3))
;; => cljs.core/PersistentVector

map と filter の操作は遅延シーケンスを返しますが、これらの操作の後には完全に実現された
シーケンスが必要なことが⼀般的なため、ベクタを返す同等のものとして mapv や filterv 等があ
ります。これらの関数には、遅延シーケンスからベクタを構築するよりも⾼速であり、意図をより
明確にするという利点があります。

(map inc [0 1 2])
;; => (1 2 3)

(type (map inc [0 1 2]))
;; => cljs.core/LazySeq

(mapv inc [0 1 2])
;; => [1 2 3]

(type (mapv inc [0 1 2]))
;; => cljs.core/PersistentVector

■マップ マップは ClojureScript で広く使われています。ベクタと同様に、マップはメタデータ
を vars に付与するために使われます。ClojureScript のデータ構造体は全てマップのキーとして使
えますが、キーワードは関数としても呼び出すことができるため、キーワードを使うことが⼀般的
です。ClojureScript のマップは、リテラル表現として、キーと値のペアを中括弧 { } で囲んで記
述します。代わりに hash-map 関数で⽣成することもできます。

(map? {:name "Cirilla"})
;; => true

(hash-map :name "Cirilla")
;; => {:name "Cirilla"}

(hash-map :name "Cirilla" :surname "Fiona")
;; => {:name "Cirilla" :surname "Fiona"}

通常のマップには特定の順番がないため、conj の操作は 1 つ以上のキーと値のペアをマップに追
加するだけです。conj をマップに使うには、最後の引数として 1 つ以上のキーと値のペアのシーケ
ンスを想定しています。

(def ciri {:name "Cirilla"})

(conj ciri [:surname "Fiona"])
;; => {:name "Cirilla", :surname "Fiona"}

(conj ciri [:surname "Fiona"] [:occupation "Wizard"])
;; => {:name "Cirilla", :surname "Fiona", :occupation "Wizard"}

上の例では要素の順が保たれていますが、もし多くのキーを含む場合、順序が保持されていない
ことがわかります。マップはキーと値を関連付ける連想型のデータ構造です。マップに要素を追加
するには assoc を使い、マップから要素を取り除くには dissoc を使います。assoc はすでにある
キーの値を更新することができます。これらの関数を試してみましょう。

(assoc {:name "Cirilla"} :surname "Fiona")
;; => {:name "Cirilla", :surname "Fiona"}

(assoc {:name "Cirilla"} :name "Alfonso")
;; => {:name "Alfonso"}

(dissoc {:name "Cirilla"} :name)
;; => {}

マップはキーの関数でもあり、指定されたキーに関連する値を返します。ベクタとは異なり、マッ
プに存在しないキーを指定すると、nil が返されます。

({:name "Cirilla"} :name)
;; => "Cirilla"

({:name "Cirilla"} :surname)
;; => nil

ClojureScript はソートされたハッシュマップも提供しています。ソートされていないバージョン
と同じように動作しますが、繰り返しの際に順序を保持します。sorted-map を使⽤すると、デフォ
ルトの順序でソートされたマップを作成できます。

(def sm (sorted-map :c 2 :b 1 :a 0))
;; => {:a 0, :b 1 , :c 2 }

(keys sm)
;; => (:a :b :c)

独⾃の順序付けが必要な場合は、sorted-map-by に⽐較関数を与えられます。組み込みの
compare 関数から返される値を反転する例を⾒てみましょう。compare 関数は 2 つの要素を⽐較
して、最初の項⽬が 2 番⽬の項⽬より⼩さい場合は -1 、等しい場合は 0 、最初の項⽬が 2番⽬の
項⽬よりも⼤きい場合 1 を返します。

(defn reverse-compare [a b] (compare b a))

(def sm (sorted-map-by reverse-compare :a 0 :b 1 :c 2))
;; => {:c 2 , :b 1 , :a 0}

(keys sm)
;; => (:c :b :a)

■セット ClojureScript のセットは #{} のリテラル表現を⽤いて作成されます。set コンストラ
クタを⽤いて作成することもできます。セットは、要素間の順番がないコレクションであり、同じ
要素の重複は許されません。

(set? #{\a \e \i \o \u})
;; => true

(set [1 1 2 3])
;; => #{1 2 3}

セットのリテラル表現は同じ値を重複して含むことができません。もし要素が重複する場合、エ
ラーが発⽣します。

#{1 1 2 3}
;; clojure.lang.ExceptionInfo: Duplicate key: 1

セットを操作するための関数は多くあります。clojure.set の名前空間にあるため、インポート
する必要があります。名前空間については後に詳しく学びます。現時点では、clojure.set の名前
空間をインポート後、シンボル s に束縛していることが分かれば⼗分です。

(require ’[clojure.set :as s])

(def danish-vowels #{\a \e \i \o \u \æ \ø \å })
;; => #{"a" "e" "å " "æ" "i" "o" "u" "ø "}

(def spanish-vowels #{\a \e \i \o \u})
;; => #{"a" "e" "i" "o" "u"}

(s/difference danish-vowels spanish-vowels)
;; => #{"å " "æ" "ø "}

(s/union danish-vowels spanish-vowels)
;; => #{"a" "e" "å " "æ" "i" "o" "u" "ø "}

(s/intersection danish-vowels spanish-vowels)
;; => #{"a" "e" "i" "o" "u"}

不変であるセットの優れた特性はネストできることです。可変であるセットをもつプログラ
ミング⾔語では重複した値を含みますが、ClojureSript では重複を許しません。実際、全ての
ClojureScript のデータ構造は不変性の性質のおかげで任意にネストすることができます。セットは
他のコレクションのデータ型と同様に conj を使うことも可能です。

(def spanish-vowels #{\a \e \i \o \u})
;; => #{"a" "e" "i" "o" "u"}

(def danish-vowels (conj spanish-vowels \æ \ø \å ))
;; => #{"a" "e" "i" "o" "u" "æ" "ø " "å "}

(conj #{1 2 3} 1)
;; => #{1 3 2}

セットは、読み取り専⽤の連想型データとして動作して、セットに含まれる値をセット⾃体に関連
づけます。ClojureScript では nil と false 以外の全ての値が真であるため、述語関数としてセッ
トを使⽤できます。

(def vowels #{\a \e \i \o \u})
;; => #{"a" "e" "i" "o" "u"}

(get vowels \b)
;; => nil

(contains? vowels \b)
;; => false

(vowels \a)
;; => "a"

(vowels \z)
;; => nil

(filter vowels "Hound dog")
;; => ("o" "u" "o")

セットにもマップのようにソート版があり、マップの sorted-map と sorted-map-by に相当す
る sorted-set と sorted-set-by 関数で作成されます。

(def unordered-set #{[0] [1] [2]})
;; => #{[0] [2] [1]}

(seq unordered-set)
;; => ([0] [2] [1])

(def ordered-set (sorted-set [0] [1] [2]))
;; =># {[0] [1] [2]}

(seq ordered-set)
;; => ([0] [1] [2])

■キュー さらに ClojureScript は、永続的で不変であるキューを提供します。キューは他のコレ
クションのように広くは使われていません。キューは #queue [] というリテラル構⽂で作ること
ができますが、便利なコンストラクタ関数はありません。

(def pq #queue [1 2 3])
;; => #queue [1 2 3]

値をキューに追加するために conj を使うと、アイテムは末尾に追加されます。

(def pq #queue [1 2 3])
;; => #queue [1 2 3]

(conj pq 4 5)
;; => #queue [1 2 3 4 5]

キューに関して留意すべき点は、スタック操作が通常のスタックのセマンティクス (同じ終点か
ら出し⼊れすること) に従わないことです。pop は前の位置から値を取り、conj は要素を後ろに
push(もしくは append) します。

(def pq #queue [1 2 3])
;; => #queue [1 2 3]

(peek pq)
;; => 1

(pop pq)
;; => #queue [2 3]

(conj pq 4)
;; => #queue [1 2 3 4]

キューはリストやベクタほど頻繁には使われませんが、ClojureScript では⼿軽に利⽤できること
を知っておくと便利です。

3.9 Destructuring

destructuring（⾮構造化）という⾔葉が⽰すように、destructuring はコレクションのような構造
化されたデータを分解して、個別の要素に焦点を当てる⽅法です。ClojureScript は分割のための構
⽂はシンプルです。destructuring は、インデックスがついたシーケンスに対しても、連想型のデー
タ構造に対しても、束縛が宣⾔された場所であればどこでも使うことができます。
前の説明を理解するために、destructuring の使⽤例を⾒てみましょう。シーケンスがあり、1 番

⽬と 3 番⽬の要素だけに関⼼があるとします。これらへの参照は、nth 関数で簡単に取得できます。

(let [v [0 1 2]
fst (nth v 0)
thrd (nth v 2)]

[thrd fst])
;; => [2 0]

しかし、前のコードは冗⻑すぎます。destructuring では、束縛の左側のベクタを使⽤して、イン
デックスがついたシーケンスからよりシンプルに値を取り出すことができます。

(let [[fst _ thrd] [0 1 2]]
[thrd fst])

;; => [2 0]

上の例では、[fst _ thrd] が destructuring の箇所です。これはベクタとして表され、fst と
thrd に対応するインデックス 0 と 2 の値がそれぞれ束縛されます。シンボルの _ は、興味のない
値 (この場合は 1)のプレースホルダとして使われます。

destructuring は let での束縛に限定されないことに注意してください。for や doseq の特殊形
式、また関数の引数部など、シンボルに値を束縛する全ての場所で動作します。destructuring 構⽂
を関数の引数部に使うことで、ペアをとり、その位置をスワップする関数を作成できます。

(defn swap-pair [[fst snd]]
[snd fst])

(swap-pair [1 2])
;; => [2 1]

(swap-pair ’(3 4))
;; => [4 3]

ベクタを⽤いた位置の destructuring は、シーケンスからインデックス付きの値を取り出すため
に⾮常に便利ですが、シーケンス内の残りの要素を破棄したくない場合があります。可変⻑引数関
数の引数を & で受け取る⽅法と同様に、& をベクタの destructuring 内で使⽤して、シーケンスの
残りの部分をグループ化することができます。

(let [[fst snd & more] (range 10)]
{:first fst
:snd snd
:rest more})

;; => {:first 0, :snd 1, :rest (2 3 4 5 6 7 8 9)}

0 のインデックスの値が fst に束縛され、1 のインデックスの値が snd に束縛され、2 以降のイ
ンデックスのシーケンスがシンボル more に束縛されていることに注⽬してください。
データ構造を destructutring する際に、データ構造全体に興味がある場合があります。これに
は :as キーワードを使うことで、任意の名前でデータ構造をシンボルに束縛することができます。
destructuring 内で使⽤される場合、元のデータ構造は :as の後のシンボルに束縛されます。

(let [[fst snd & more :as original] (range 10)]
{:first fst
:snd snd
:rest more
:original original})

;; => {:first 0, :snd 1,
;; :rest (2 3 4 5 6 7 8 9),
;; :original (0 1 2 3 4 5 6 7 8 9)}

インデックス付きのシーケンスだけでなく、連想型のデータも destructuring することができま
すが、ベクタではなくマップを⽤いて束縛を⾏います。キーは値を束縛するためのシンボルであり、
値は連想型データ構造で検索するキーです。例を⾒てみましょう。

(let [{language :language} {:language "ClojureScript"}]
language)

;; => "ClojureScript"

上の例では、:language キーと関連づけられた値を取り出して、language シンボルに束縛して
います。もし探しているキーが存在しない場合は nil が束縛されます。

(let [{name :name} {:language "ClojureScript"}]
name)

;; => nil

連想型での destructuring では、束縛にデフォルト値を指定できます。このデフォルト値は、
destructuring  を⾏うデータ構造でキーが⾒つからない場合に使⽤されます。:or キーワードを
マップの後につけることで、デフォルトの値が束縛されます。次の例を⾒てください。

(let [{name :name :or {name "Anonymous"}}
{:language "ClojureScript"}]

name)
;; => "Anonymous"

(let [{name :name :or {name "Anonymous"}} {:name "Cirilla"}]
name)

;; => "Cirilla"

連想型の destructuring では、元のデータ構造を :as キーワードの後に置かれたシンボルに束縛
することも可能です。

(let [{name :name :as person} {:name "Cirilla" :age 49}]
[name person])

;; => ["Cirilla" {:name "Cirilla" :age 49}]

連想型データ構造のキーとして使えるのは、キーワードだけではありません。数字、⽂字列、シ
ンボル、およびその他の多くのデータ構造もキーとして使えるため、これらを使⽤して構造を変更
することもできます。シンボルが var 検索として解決されないようにするには、シンボルをクオー
トする必要があります。

(let [{one 1} {0 "zero" 1 "one"}]
one)

;; => "one"

(let [{name "name"} {"name" "Cirilla"}]
name)

;; => "Cirilla"

(let [{lang ’language} {’language "ClojureScript"}]
lang)

;; => "ClojureScript"

キーに対応する値はシンボルと等しいことが多く (例えば :language の値を language シンボル
に束縛する等)、キーはキーワード、⽂字列、シンボルであることが多いため、ClojureScript には
これらのために略記法があります。

:keys を⽤いて キーワードの destructuring を始めます。以下に全ての例を⽰します。

(let [{:keys [name surname]} {:name "Cirilla" :surname "Fiona"}]
[name surname])

;; => ["Cirilla" "Fiona"]

例の通り、:keys を⽤いて束縛内でシンボルのベクタと関連づけた場合、シンボルのキーワード
に対応する値はそれらに対応づけられます。つまり、{:keys [name surname]} は {name :name

surname :surname} の略記法であり、同じ意味です。

(let [{:strs [name surname]} {"name" "Cirilla" "surname" "Fiona"}]
[name surname])

;; => ["Cirilla" "Fiona"]

(let [{:syms [name surname]} {’name "Cirilla" ’surname "Fiona"}]
[name surname])

;; => ["Cirilla" "Fiona"]

また、destructuring の興味深い特性は、任意で destructuring をネストできることです。これに
より、コレクション上のネストされたデータにアクセスするコードは、コレクションの構造を模倣
するため、⾮常に理解しやすくなります。

(let [{[fst snd] :languages} {:languages ["ClojureScript" "Clojure"]}]
[snd fst])

;; => ["Clojure" "ClojureScript"]

3.10 Threading マクロ
Threading マクロはアロー関数として知られています。Threading マクロを使うことで、ネスト

された複数の関数呼び出しを実⾏するときに、読みやすいコードを記述できます。
例えば (f (g (h x))) というコードにおいて、f は g の結果を第 1 引数として受け取ります。

Threading マクロ -> を使うことで、(-> x (h) (g) (f)) と書くことができます。こちらのほう
が読みやすいですね。

Threading マクロはマクロとして定義されているため、実⾏時のパフォーマンスに影響を与えま
せん。(-> x (h) (g) (f)) はコンパイル時に (f (g (h x))) に変換されます。

h, g, f の括弧はなくてもかまいません。(f (g (h x))) は (-> x h g f) と省略して書くこと
もできます。

thread-firstマクロ ( -> )

-> は thread-firstマクロ と呼ばれますが、複数の式が実⾏されていく中で、最初の引数に値を挿
⼊していくことから、そのような名前がつけられています。
では具体例として、まずは Threading マクロを使わない場合を⾒てみましょう。

(def book {:name "Lady of the Lake"
:readers 0})

(update (assoc book :age 1999) :readers inc)
;; => {:name "Lady of the lake" :age 1999 :readers 1}

thread-firstマクロ (->) を使うと、次のように書くことができます。

(-> book
(assoc :age 1999)
(update :readers inc))

;; => {:name "Lady of the lake" :age 1 999 :readers 1}

このリーダマクロはデータ構造を変換するときに特に役にたちます。これは ClojureScript
(Clojure) のデータ構造を変換する関数は、⼀貫して最初の引数でデータ構造を受け取るからです。

thread-last マクロ ( ->> )

thread-last マクロ (->>) と thread-first マクロ (->) の主な違いは、最初の引数にデータを⼊れ
るのではなく、最後の引数にデータを⼊れる点です。例を⾒てみましょう。

(def numbers [1 2 3 4 5 6 7 8 9 0])

(take 2 (filter odd? (map inc numbers)))
;; => (3 5)

上の例は thread-last マクロ (->>) を⽤いて、次のように書き直すことができます。

(->> numbers
(map inc)
(filter odd?)
(take 2))

;; => (3 5)

thread-last マクロ (->>) は、シーケンスのデータを変形するときに役にたちます。ClojureScript
のシーケンスとコレクションを操作する関数は、シーケンスやコレクションを最後の引数として受
け取るためです。

thread-as マクロ ( as-> )

最後に -> と ->> どちらにも適⽤できない場合があります。この場合は、as-> マクロを使いま
す。as-> マクロは、最初と最後に限らず、好きな場所にデータを⼊れることができます。

as-> マクロは 2 つの引数を固定でもち、その後に任意の数の式が続きます。-> と同様に、最初
の引数は以降の式に渡されていく値です。2 番⽬の引数は束縛の名前です。以降の各々の式で、前
の式の結果のためにその束縛の名前が使われます。
例を⾒てみましょう。

(as-> numbers $
(map inc $)
(filter odd? $)
(first $)
(hash-map :result $ :id 1))

;; => {:result 3 :id 1}

thread-some マクロ ( some-> と some->> )

ClojureScript のより特殊なスレッド系マクロを 2 つ紹介します。これらは -> と ->> と同じよ
うに動作します。式の 1 つが nil と評価された場合に式を短絡することをサポートします。
例を⾒てみましょう。

(some-> (rand-nth [1 nil])
(inc))

;; => 2

(some-> (rand-nth [1 nil])
(inc))

;; => nil

これにより、ヌル・ポインタ・エクセプションを簡単に回避することができます。

thread-cond マクロ ( cond-> と cond->> )

cond-> と cond->> マクロは -> と ->> と似ており、パイプラインからの処理を条件によりス
キップできます。例を⾒てみましょう。

(defn describe-number
[n]
(cond-> []

(odd? n) (conj "odd")
(even? n) (conj "even")
(zero? n) (conj "zero")
(pos? n) (conj "positive")))

(describe-number 3)
;; => ["odd" "positive"]

(describe-number 4)
;; => ["even" "positive"]

対応する条件が論理的に真と評価された場合のみ、値が引き渡されます。

追加資料
詳しい情報は次の⽂献を参照してください。
- Lesser known Clojure: variants of threading macro*2

- Threading Macros Guide*3

*2 http://www.spacjer.com/blog/2015/11/09/lesser-known-clojure-variants-of-threading-macro
*3 http://clojure.org/guides/threading_macros

3.11 リーダーコンディショナル (Reader Conditionals)

この⾔語の特徴として、Clojure の異なる⽅⾔ごとに、コードを共有したり、プラットフォームに
依存するコードを書くことができます。
リーダーコンディショナル (Reader Conditionals)を使うには、ファイルの拡張⼦を .cljs から

.cljc に変更する必要があります。リーダの条件分けは拡張⼦が .cljc の場合にのみ動作します。

Standard (#?)

リーダーコンディショナルには、standard と splicing の 2 タイプがあります。standard の場合
は、cond のように書くことができます。

(defn parse-int
[v]
#?(:clj (Integer/parseInt v)

:cljs (js/parseInt v)))

リーダマクロの #? は condと同じように⾒えますが、条件部にキーワードを⽤いて、ClojureScript
には :cljs 、Clojure には :clj をプラットフォームの特定のために使います。リーダマクロはコ
ンパイル時に評価されるので、実⾏時にオーバーヘッドが発⽣しません。

Splicing ( #?@ )

リーダーコンディショナルの splicing は、standard と同じように使いますが、リストを結合
(splice)できます。このためにリーダマクロ #?@ を使います。

(defn make-list
[]
(list #?@(:clj [5 6 7 8]

:cljs [1 2 3 4])))

;; ClojureScript の場合
(make-list)
;; => (1 2 3 4)

ClojureScript のコンパイラは、上のコードをこのように解釈します。

(defn make-list
[]
(list 1 2 3 4))

リーダーコンディショナルの splicing では、複数のトップレベルのフォームを結合することはで
きないため、次のコードは不正です。

#?@(:cljs [(defn func-a [] :a)
(defn func-b [] :b)])

;; => #error "Reader conditional splicing not allowed at the top level."

そのようにするには、複数のフォームを使うか、do ブロックを使う必要があります。

#?(:cljs (defn func-a [] :a))
#?(:cljs (defn func-b [] :b))

;; または、

#?(:cljs
(do

(defn func-a [] :a)
(defn func-b [] :b)))

参考⽂献
詳しい内容は以下の⽂献を参考にしてください。
- Reader Conditionals Guide*4

- Clojure Reader Conditionals by Example*5

- cuerdas - string manipulation library*6

*4 http://clojure.org/guides/reader_conditionals
*5 https://danielcompton.net/2015/06/10/clojure-reader-conditionals-by-example
*6 https://github.com/funcool/cuerdas

3.12 名前空間
名前空間の定義
名前空間は、ClojureScript でコードをモジュール化する基本単位です。ClojureScript の名前空
間は、Java のパッケージ、Ruby や Python のモジュールに似ており、ns マクロで定義できます。
ClojureScript のソースを少し⾒たことがある⼈は、ファイルの最初に次のように書かれていること
に気付いたかもしれません。

(ns myapp.core
"Some docstring for the namespace.")

(def x "hello")

名前空間は動的であり、いつでも⽣成することができます。ただし、ファイルごとに 1 つの名前
空間を使⽤するのが規則です。通常、名前空間の定義はファイルの先頭にあり、その後に オプショ
ンの docstring が続きます。

var とシンボルについては前に説明しました。定義する全ての var は、その名前空間に関連づけ
られます。具体的な名前空間を定義しない場合は、デフォルトの cljs.user が名前空間として使わ
れます。

(def x "hello")
;; => #’cljs.user/x

他の名前空間の読み込み
名前空間を定義して、その中で変数を定義するのは⾮常に簡単ですが、他の名前空間のシンボル
が使えなければ便利でありません。この⽬的のために、ns マクロは他の名前空間をロードする簡単
な⽅法を提供します。

(ns myapp.main
(:require myapp.core

clojure.string))

(clojure.string/upper-case myapp.core/x)
;; => "HELLO"

ご覧のとおり、別の名前空間から var や関数にアクセスするためには、完全な修飾名 (名前空間
+ 変数名)を使⽤します。

これにより他の名前空間にアクセスできますが、繰り返しが多く、冗⻑になりすぎます。名前空
間の名前が⻑い場合は特に不便です。これを解決するには、:as ディレクティブを利⽤して、名前
空間に追加の (通常より短い)エイリアスを作成します。その⽅法は次のとおりです。

(ns myapp.main
(:require [myapp.core :as core]

[clojure.string :as str]))

(str/upper-case core/x)
;; => "HELLO"

さらに ClojureScript では、:refer ディレクティブを⽤いることで、特定の名前空間から var や
関数を参照することが容易になります。:refer の後には、特定の名前空間内の var を参照するシ
ンボルのシーケンスを書きます。事実上、これらの var と関数はあなたの名前空間の⼀部のように
なるため、これらに修飾⼦をつける必要はありません。

(ns myapp.main
(:require [clojure.string :refer [upper-case]]))

(upper-case x)
;; => "HELLO"

最後に、cljs.core の中にある全てのものを把握する必要があります。cljs.core の名前空間
は⾃動的にロードされるので、明⽰的に require すべきではありません。cljs.core で定義されて
いる var と衝突する var を宣⾔する場合、ns マクロは特定のシンボルを除外して⾃動的にロード
されないようにする別のディレクティブを提供します。
次の例を⾒てください。

(ns myapp.main
(:refer-clojure :exclude [min]))

(defn min
[x y]
(if (> x y)

y
x))

ns マクロは、ホストのクラスを読み込むために :import ディレクティブ、マクロを読み込むた
めに :refer-macros ディレクティブを⽤意しています。これらについては、別のセクションで説
明します。

名前空間とファイル名
myapp.core のような名前空間がある場合、コードは myapp ディレクトリ内で core.cljs とい

う名前のファイルにしなければなりません。上記の例では myapp.core と myapp.main を使⽤して
いますが、ファイル構造は次のようになります。

myapp
└── src

└── myapp
├── core.cljs
└── main.cljs

3.13 抽象化とポリモーフィズム
「インターフェイス等を⽤いてビジネスロジックの抽象化を上⼿く定義できたが、コントロールで
きない別のモジュールを処理する必要がでてきた」
過去にこのような状況を経験したことがあるかと思います。アダプターやプロキシ等の⼿法で対
処することが考えられますが、複雑な作業を追加で⾏う必要があります。
⼀部の動的⾔語では「モンキーパッチ」が可能です。そのような⾔語では、クラスは開かれてい
て、いつでもメソッドを定義したり再定義したりできます。このテクニックは⾮常に悪い慣習であ
ることもよく知られています。
サードパーティのライブラリをインポートするときに、使⽤しているメソッドを暗黙のうちに上
書きできる⾔語は信頼できません。この場合、⼀貫した動作は期待できません。
これらの問題は Expression Problem として知られています。
詳細は、http://en.wikipedia.org/wiki/Expression_problem を参照してください。

プロトコル
「インターフェイス」を定義するための ClojureScript のプリミティブは、プロトコルと呼ばれて
います。プロトコルは名前と関数のセットで構成されます。全ての関数には、JavaScript の this、
Python の self に対応する引数が少なくとも 1 つはあります。
プロトコルは、型に基づくポリモーフィズムを提供して、常に最初の引数によりディスパッチさ
れます（前述の通り、JavaScript の this に相当します）。プロトコルは次のように記述します。

(ns myapp.testproto)

(defprotocol IProtocolName
"プロトコルを記述する docstring"
(sample-method [this] "この関数に関連付けられている doc ⽂字列"))

プロトコルを型と区別するために、接頭辞に I をつける慣習があります。Clojure のコミュニティ
では、この接頭辞の使い⽅について様々な意⾒があります。私たちの意⾒では、名前衝突と混乱の
可能性を避けるために許容できる解決策です。しかし、この接頭辞を使⽤しないことは悪しき慣⾏
とはみなされません。
ユーザーからすると、プロトコル関数は、プロトコルが定義されている名前空間で定義された単
純な関数です。これを⽤いると、競合する関数名を持つ同じ型のために実装された異なるプロトコ
ル間での競合を回避するために、容易でシンプルなアプローチをとることができます。
次の例では、IInvertible という名前で、反転できるデータのためのプロトコルを作成します。

(defprotocol IInvertible
"これは「反転可能な」データ型のためのプロトコルです。"
(invert [this] "指定した項⽬を反転します"))

既存の型の拡張
プロトコルの⼤きな強みの 1 つは、既存の型やサードパーティの型を拡張できることです。この
操作は、様々な⽅法で実⾏できます。ほとんどの場合、extend-type マクロか extend-protocol

マクロを使います。extend-type の構⽂は次のように記述します。

(extend-type TypeA
ProtocolA
(function-from-protocol-a [this]

;; ここに実装を記述する
)

ProtocolB
(function-from-protocol-b-1 [this parameter 1]

;; ここに実装を記述する
)

(function-from-protocol-b-2 [this parameter 1 parameter 2]
;; ここに実装を記述する
))

extend-type を使⽤すると、異なるプロトコルを使⽤して 1 つの型を 1 つの式で拡張できます。
先ほど定義した IInvertibleプロトコルを試してみましょう。

(extend-type string
IInvertible
(invert [this] (apply str (reverse this))))

(extend-type cljs.core.List
IInvertible
(invert [this] (reverse this)))

(extend-type cljs.core.PersistentVector
IInvertible
(invert [this] (into [] (reverse this))))

プロトコルを⽂字列に対して拡張するために、js/String を使わずに、特別なシンボル string

が使われていることに注⽬してください。これは、組み込みの JavaScript の型には特別な扱いがあ
り、string を js/String に置き換えると、コンパイラは警告を発⽣するからです。
プロトコルを JavaScriptのプリミティブ型に拡張する場合、js/Number, js/String, js/Object,

js/Array, js/Boolean, js/Function を使う代わりに、それぞれに特別なシンボル (number,
string, object, array, boolean, function)を使う必要があります。
では、私たちのプロトコル実装を試してみましょう。

(invert "abc")
;; => "cba"

(invert 0)
;; => 0

(invert ’(1 2 3))
;; => (3 2 1)

(invert [1 2 3])
;; => [3 2 1]

⽐較すると、extend-protocol はその逆を⾏います。つまり、あるプロトコルに対して、複数の
型の実装を追加します。構⽂は次のようになります。

(extend-protocol ProtocolA
TypeA
(function-from-protocol-a [this]

;; ここに実装を記述する
)

TypeB
(function-from-protocol-a [this]

;; ここに実装を記述する
))

したがって、前例は次のように書くこともできます。

(extend-protocol IInvertible
string
(invert [this] (apply str (reverse this)))

cljs.core.List
(invert [this] (reverse this))

cljs.core.PersistentVector
(invert [this] (into [] (reverse this))))

ClojureScript での抽象化
ClojureScript ⾃体は、プロトコルとして定義された抽象に基づいて構築されています。Clojure-

Script ⾔語⾃体のほぼ全ての動作は、サードパーティ製のライブラリに適⽤できます。実際の例を
⾒てみましょう。
前のセクションでは、様々な組み込みのコレクションについて説明しました。次の例では、セッ
トを使⽤します。次のコードを⾒てください。

(def mynums #{1 2})

(filter mynums [1 2 4 5 1 3 4 5])
;; => (1 2 1)

何が起こったのでしょうか。セット型は ClojureScript 内部の IFn プロトコルを実装しています。
IFn プロトコルは、関数や呼び出し可能なもののための抽象化を表現します。このように、filter

で呼び出し可能な述部のように使⽤できます。では、⽂字列のコレクションを filter する述語関
数として正規表現を使⽤したい場合はどうなるでしょうか。

(filter #"^foo" ["haha" "foobar" "baz" "foobaz"])
;; TypeError: Cannot call undefined

RegExp の型は IFn プロトコルを実装していないので、例外が発⽣します。正規表現は呼び出せ
ませんが、次のように簡単に修正することができます。

(extend-type js/RegExp
IFn
(-invoke
([this a]

(re-find this a))))

この例を分析してみましょう。IFn プロトコルで invoke 関数を実装するように js/RegExp 型
を拡張しています。正規表現を関数のように呼び出すには、関数のオブジェクトとパターンを指定
して re-find 関数を呼び出します。
これで、正規表現のインスタンスを filter 操作の述語として使⽤できるようになります。

(filter #"^foo" ["haha" "foobar" "baz" "foobaz"])
;; => ("foobar" "foobaz")

プロトコルを⽤いた内部分析
ClojureScript には、実⾏時のイントロスペクション (introspection) を便利にする関数

satisfies? が⽤意されています。この関数の⽬的は、実⾏時に任意のオブジェクト (任意の
型のインスタンス) が特定のプロトコルの基準を満たすかを判定することです。先ほどの例では、
セットのインスタンスが IFn プロトコルを満たす場合、true を返します。

(satisfies? IFn #{1})
;; => true

マルチメソッド
ポリモーフィズムの⾮常に⼀般的なユースケースを解決するプロトコルについて前に説明しまし
た。つまり、型によるディスパッチです。しかし、いくつかの状況では、プロトコルを⽤いたアプ
ローチには制限があります。そのような場合、マルチメソッドが救いの⼿を差し伸べてくれます。
マルチメソッドは、型によるディスパッチに限定されません。代わりに、複数の引数の型と値に
よるディスパッチを提供します。また、特定の⽬的のために階層性を定義できます。また、プロト
コルと同様にマルチメソッドは「開かれたシステム」なので、ユーザやサードパーティは新しい型
のためにマルチメソッドを拡張できます。
マルチメソッドのための基本的な構⽂は defmulti と defmethod です。defmulti は、イニシャ

ル・ディスパッチ関数 (initial dispatch function)を⽤いてマルチメソッドを作成するために使いま
す。次はそのモデルです。

(defmulti say-hello
"⾔語キーに応じて挨拶のメッセージを返すポリモーフィズム的な関数。
デフォルトの⾔語キーを en に設定します。"

(fn [param] (:locale param))
:default :en)

defmulti 内で定義されてい無名関数は、ディスパッチ関数です。この関数は say-hello 関数を
呼び出すたびに呼び出されて、ディスパッチに使⽤される何らかのマーカーオブジェクトを返しま
す。この例では、最初の引数の :locale キーの内容を返します。
最後に、実装を追加します。defmethod で実装を定義します。

(defmethod say-hello :en
[person]
(str "Hello " (:name person "Anonymous")))

(defmethod say-hello :es
[person]
(str "Hola " (:name person "An ó nimo")))

もし :locale と (オプションの):name キーを含むハッシュマップに対してこの関数を実⾏する
と、マルチメソッドはまずディスパッチ関数を呼び出してディスパッチ値を決定し、次にその値の
実装を検索します。実装が⾒つかった場合、ディスパッチャはそれを実⾏します。それ以外の場合、
ディスパッチはデフォルトの実装 (指定されている場合)を検索して実⾏します。

(say-hello {:locale :es})
;; => "Hola An ó nimo"

(say-hello {:locale :en :name "Ciri"})
;; => "Hello Ciri"

(say-hello {:locale :fr})
;; => "Hello Anonymous"

デフォルトの実装が指定されていない場合は、⼀部の値がそのマルチメソッドの実装をもたない
ことを通知する例外が発⽣します。

ヒエラルキー
ヒエラルキーは、ClojureScript において、ドメインに必要なリレーションを構築することができ

ます。ヒエラルキーは、シンボル、キーワード、型のような名前付きのオブジェクト間のリレーショ
ンとして定義されます。
ヒエラルキーは、必要に応じてグローバルまたはローカルに定義できます。マルチメソッドのよ
うに、ヒエラルキーは単⼀の名前空間に限定されません。ヒエラルキーは、定義されている名前空
間だけでなく、任意の名前空間から拡張できます。
グローバルの名前空間はより制限されていますが、これは正当な理由によるものです。名前空間
を持たないキーワードまたはシンボルは、グローバルのヒエラルキーでは使⽤できません。この動
作は、複数のサードパーティ製のライブラリが異なるセマンティクスに対して同じシンボルを使⽤
する場合、予期しない状況が発⽣するのを防ぐのに役⽴ちます。

59



第 3章 ClojureScript ⼊⾨ 3.13 抽象化とポリモーフィズム

■ヒエラルキーの定義 ヒエラルキーのリレーションは、derive 関数を⽤いて確⽴する必要があり
ます。

(derive ::circle ::shape)
(derive ::box ::shape)

ここでは、名前空間を持つキーワード間のリレーションを定義しました。この場合、::circle は
::shape の⼦であり、::box も ::shape の⼦です。

::circle は :current.ns/circle の省略形なので、REPL で実⾏する場合、::circle は
:current.ns/circle として評価されます。

■ヒエラルキーとイントロスペクション ClojureScript には、グローバルあるいはローカルに定義
されたヒエラルキーのイントロスペクションを実⾏時に可能にする⼩さなツールセットが⽤意され
ています。このツールセットは 3 つの関数で構成されています。isa?, ancestors, descendants

です。前例で定義したヒエラルキーを⽤いて、このツールセットの使い⽅を⾒てみましょう。

(ancestors ::box)
;; => #{:cljs.user/shape}

(descendants ::shape)
;; => #{:cljs.user/circle :cljs.user/box}

(isa? ::box ::shape)
;; => true

(isa? ::rect ::shape)
;; => false

■ローカル環境に定義されたヒエラルキー 前に説明した通り、ClojureScript はローカルのヒエラ
ルキーを定義することもできます。これは、make-hierarchy 関数により⾏うことができます。次
に、前例をローカルのヒエラルキーを⽤いて書き直してみましょう。

(def h (-> (make-hierarchy)
(derive :box :shape)
(derive :circle :shape)))

これで、ローカルに定義されたヒエラルキーで同じ関数を使⽤できます。

(isa? h :box :shape)
;; => true

(isa? :box :shape)
;; => false

ご覧のように、ローカルのヒエラルキーでは、通常の (名前空間が修飾されない)キーワードを使
⽤できます。もし isa? をローカルのヒエラルキーの引数を渡さずに実⾏すると、予想した通りに
false を返します。

■マルチスレッドにおけるヒエラルキー ヒエラルキーの⼤きな利点の 1 つは、マルチメソッドと
上⼿く働くことです。これは、マルチメソッドがデフォルトで isa? 関数をディスパッチの最終ス
テップで使うからです。
その意味を明確に理解するために例を⾒てみましょう。初めに、defmulti を⽤いてマルチメソッ
ドを定義します。

(defmulti stringify-shape
"shape キーワードを⼈が読めるような表現で出⼒する関数"
identity
:hierarchy #’h)

:hierarchy キーワードパラメータを使⽤して、使⽤するヒエラルキーをマルチメソッドに指定
します。指定しない場合は、グローバルのヒエラルキーが使⽤されます。
次に、defmethod を⽤いて、マルチメソッドのための実装を定義します。

(defmethod stringify-shape :box
[_]
"A box shape")

(defmethod stringify-shape :shape
[_]
"A generic shape")

(defmethod stringify-shape :default
[_]
"Unexpected object")

では、関数を :box で呼び出したときに何が起こるかを⾒てみましょう。

(stringify-shape :box)
;; => "A box shape"

全て予想通りに、マルチメソッドは与えられた引数に適する実装を直接実⾏しました。次に
:circle を⽤いて同じ関数を呼び出してみましょう。この場合、:circle に合致する実装はありま
せん。

(stringify-shape :circle)
;; => "A generic shape"

マルチメソッドは指定されたヒエラルキーを⽤いて⾃動的に解決されます。:circle は :shape

の⼦孫 (descendant)であるため、:shape の実装が呼び出されます。
ヒエラルキーの⼀部ではないキーワードを指定すると、:default の実装が呼び出されます。

(stringify-shape :triangle)
;; => "Unexpected object"

3.14 データ型
ここまでは、マップ、セット、リスト、ベクタをデータを表現するために使ってきました。多くの
場合、これは本当に素晴らしいアプローチです。しかし、⾃分⾃⾝でこのような型を定義する必要
があるときもあります。この本では、このような型のことを「データ型」と呼びます。
これまでは、マップ、セット、リスト、ベクタを使ってデータを表現してきました。ほとんどの場
合、これは本当に素晴らしいアプローチです。しかし、場合によっては、独⾃に型を定義する必要
があります。本書ではそれらをデータ型と呼ぶことにします。
データ型は、次の機能を提供します。
- ホスト環境が⽀援する独⾃の型 (名前付きまたは無名)
- プロトコルを実装する機能 (インライン)
- フィールドまたはクロージャを⽤いた明⽰的に宣⾔された構造
- マップのような動作 (レコードを経由。詳細は後ほど。)

Deftype

ClojureScript で独⾃の型を作成するための最も低レベルな構⽂は deftype マクロです。例とし
て、User という型を定義してみましょう。

(deftype User [firstname lastname])

型を定義すると、User のインスタンスを作成できます。次の例では、User の後のドット . はコ
ンストラクタを呼び出していることを⽰します。

(def person (User. "Triss" "Merigold"))

フィールドには .- でアクセスできます。

(.-firstname person)
;; => "Triss"

deftypeで定義された型は、ホスト環境に⽀援されるクラスのようなオブジェクトを現在の名前
空間と関連づけて作成します。これは defrecord で定義される型も同様です。defrecord につい
ては後に取り上げます。利便性のために ClojureScript は、:require ディレクティブでインポー
トできる ->User というコンストラクタ関数も定義しています。
私たちは個⼈的にはこのタイプの関数は好きではないため、より慣⽤的な名前で独⾃のコンスト
ラクターを定義したいと思います。

(defn make-user
[firstname lastname]
(User. firstname lastname))

私たちのコードでは ->User の代わりにこちらを使います。

Defrecord

レコード (record) は、ClojureScript で型を定義するための少し⾼いレベルの抽象化であり、型
の定義のために推奨されるべき⽅法です。
私たちが知っているように、ClojureScript ではマップなどのプレーンなデータ型を使⽤する傾向
がありますが、ほとんどの場合、アプリケーションの実体を表す名前付きの型が必要です。そこで
レコードの出番です。
レコードはマッププロトコルを実装するデータ型であるため、マップと同様に使⽤できます。ま
た、レコードも適切な型であるため、プロトコルを介して型ベースのポリモーフィズムをサポート
します。要約すると、レコードを⽤いると、様々な抽象化で働くマップという、両⽅の世界の良い
点をもつことができます。
レコードを使⽤して User 型の定義を始めましょう。

(defrecord User [firstname lastname])

deftype の構⽂とよく似ています。実際には、型を定義するための下位レベルのプリミティブと
して、舞台裏では deftype を使⽤しています。
ここで、フィールドへのアクセスについて、素の型の場合との違いを⾒てみましょう。

(def person (User. "Yennefer" "of Vengerberg"))

(:firstname person)
;; => "Yennefer"

(get person :firstname)
;; => "Yennefer"

レコードはマップであり、マップのように扱うことができます。前述の通り、レコードはマップ
であり、マップのように動作します。

(map? person)
;; => true

また、マップと同様に、レコードは最初に定義されていないフィールドを追加できます。

(def person2 (assoc person :age 92))

(:age person2)
;; => 92

このように、assoc 関数は期待どおりに動作し、同じ型の新しいインスタンスを、新しいキーと
値のペアで返します。ただし、dissoc には注意してください。レコードでの動作は、マップでの動
作とは少し異なります。外すフィールドがオプションのフィールドである場合は新しいレコードを
返しますが、必須のフィールドを外す場合はマップを返します。マップとのもう 1 つの違いは、レ
コードが関数のように動作しないことです。

(def plain-person {:firstname "Yennefer", :lastname "of Vengerberg"})

(plain-person :firstname) ;; => "Yennefer"

(person :firstname)     ;; => person.User does not implement IFn protocol.

利便性のために、deftype と同様に、defrecord マクロは ->User 関数と追加の map->User コ
ンストラクタ関数を公開しています。このコンストラクタについては、deftype で定義したコンス
トラクタと同じ意⾒です。他のコンストラクタを使⽤する代わりに、独⾃のコンストラクタを定義
することをお勧めします。それらが存在する場合、どのように使⽤できるかを⾒てみましょう。

(def cirilla (->User "Cirilla" "Fiona"))
(def yen (map->User {:firstname "Yennefer"

:lastname "of Vengerberg"}))

プロトコルの実装
これまで⾒てきた両⽅の型定義のプリミティブは、(前に説明した通り)プロトコルのインライン
実装を可能にします。例として、次のように定義します。

(defprotocol IUser
"User 型を操作するための共通の抽象化"
(full-name [_] "Get the full name of the user."))

ここでは、抽象化のために、インライン実装 (この場合は IUser) を使⽤して型を定義できます。

(defrecord User [firstname lastname]
IUser
(full-name [_]

(str firstname " " lastname)))

;; Create an instance.
(def user (User. "Yennefer" "of Vengerberg"))

(full-name user)
;; => "Yennefer of Vengerberg"

Reify

reify マクロは、型を事前に定義せずにオブジェクトを作成するために使⽤できる特別なコンス
トラクタです。プロトコルの実装は deftype や defrecord と同じように提供されますが、対照的
に reify はアクセス可能なフィールドをもちません。

IUser の抽象化でうまく機能するユーザ型のインスタンスをエミュレートするには、次のように
します。

(defn user
[firstname lastname]
(reify

IUser
(full-name [_]

(str firstname " " lastname))))

(def yen (user "Yennefer" "of Vengerberg"))
(full-name yen)
;; => "Yennefer of Vengerberg"

Specify

specify! は reify の⾼度な代替⼿段であり、既存の JavaScript オブジェクトにプロトコルの
実装を追加できます。これは、JavaScript のライブラリのコンポーネントにプロトコルを移植する
場合に便利です。

(def obj #js {})

(specify! obj
IUser
(full-name [_]

"my full name"))

(full-name obj)
;; => "my full name"

specify は specify! の不変版です。これは ICloneable を実装している不変でコピー可能な
値 (ClojureScript のコレクション等)に対して使えます。

(def a {})

(def b (specify a
IUser
(full-name [_]

"my full name")))

(full-name a)
;; Error: No protocol method IUser.full-name
;; defined for type cljs.core/PersistentArrayMap: {}

(full-name b)
;; => "my full name"

3.15 ホスト環境の相互運⽤性
ClojureScript は、その兄弟である Clojure と同様に、ゲスト⾔語として設計されています。これ
は、ClojureScript にとっては JavaScript、Clojure にとっては JVM といった既存のエコシステム
の上で、⾔語の設計がうまく機能することを意味します。

型
ClojureScript は、予想と異なるかもしれませんが、プラットホームが提供する全ての型を利⽤し
ようとします。次は、ClojureScript の基礎となるプラットフォームから継承して再利⽤する機能の
⼀覧です。

- ClojureScript の string は、JavaScript の String である。

- ClojureScript の number は、JavaScript の Number である。

- ClojureScript の nil は、JavaScript の null である。

- ClojureScript の正規表現は、JavaScript の RegExp のインスタンスである。

- ClojureScript はインタープリタで実⾏されず、常に JavaScript にコンパイルされる。

- ClojureScript はプラットフォームの API を同じセマンティクスで簡単に呼び出せる。

- ClojureScript のデータ型は、内部的には JavaScript のオブジェクトにコンパイルされる。

ClojureScript はその上に、ベクタ、マップ、セットなど、本章のこれまでのセクションで説明し
たプラットフォームに存在しない独⾃の抽象化や型を構築します。

プラットホームの型との連携
ClojureScript には、オブジェクトメソッドの呼び出し、新しいインスタンスの作成、オブジェク
トプロパティへのアクセスなど、プラットフォームの型との対話を可能にするスペシャルフォーム
の⼩さなセットが付属しています。

プラットホームへのアクセス
ClojureScript には、特殊な名前空間 js/ を通してプラットフォームの環境全体にアクセスでき
ます。JavaScript の組み込み関数 parseInt を実⾏する式は次の通りです。

(js/parseInt "222")
;; => 222

新たにインスタンスを⽣成する
ClojureScript にはインスタンスを作成する⽅法が 2 通りあります。まずは、new を使う⽅法

です。

(new js/RegExp "^foo$")

ドット . を使うこともできます。

(js/RegExp. "^foo$")

後者の⽅法でインスタンスを作成することをお勧めします。2 つの形式の間に違いはありません
が、ClojureScript コミュニティでは後者の形式が最も頻繁に使⽤されています。

インスタンスのメソッド呼び出し
インスタンスメソッドを呼び出すには、JavaScript の⽅法 (obj.method()の形式)とは異なり、

Lisp 系⾔語の他の標準関数と同じように最初に来ますが、関数名は特殊な形式 . で始まります。
正規表現のインスタンスの .test()メソッドを呼び出すには次のようにします。

(def re (js/RegExp "^Clojure"))

(.test re "ClojureScript")
;; => true

JavaScript のオブジェクトに対してインスタンスメソッドを呼び出すことができます。以下の例
において、前者はこれまで⾒てきたパターンに従っています。後者は前者のショートカットです。

(.sqrt js/Math 2)
;; => 1.4142135623730951
(js/Math.sqrt 2)
;; => 1.4142135623730951

オブジェクトのプロパティへのアクセス
オブジェクトのプロパティにアクセスする⽅法は、メソッドを呼び出す⽅法とほぼ同じです。違
いは、ドット . の代わりに ドットハイフン .- を使うことです。例を⾒てみましょう。

(.-multiline re)
;; => false
(.-PI js/Math)
;; => 3.141592653589793

プロパティへアクセスするための略記法
js/ で始まるシンボルは、ネストされたプロパティのアクセスを表⽰するためにドットを含むこ
とができます。次の例では、両者とも同じ関数を呼び出します。

(.log js/console "Hello World")

(js/console.log "Hello World")

次の式は同じプロパティにアクセスします。

(.-PI js/Math)
;; => 3.141592653589793

js/Math.PI
;; => 3.141592653589793

JavaScriptのオブジェクト
ClojureScript には、プレーンな JavaScript のオブジェクトを作成する様々な⽅法があり、各々
に⽬的があります。基本的なものは js-obj 関数です。このメソッドは、可変数のキーと値のペア
を受け⼊れて、JavaScript のオブジェクトを返します。

(js-obj "country" "FR")
;; => #js {:country "FR"}

返り値は、プレーンな JavaScript のオブジェクトを受け⼊れる何らかのサードパーティのライブ
ラリに渡すことができますが、この関数の戻り値の実際の表現を⾒ることができます。まったく同
じことをするための別の形式です。
リーダーマクロの #js を使うことも可能です。マップかベクタの先頭に #js をつけることで、結
果が JavaScript のオブジェクトに変換されます。

(def myobj #js {:country "FR"})

上の例で変換されるものは、次のコードのようになります。

var myobj = {country: "FR"};

前のセクションで説明した通り、オブジェクトのプロパティにはドットハイフン .- を⽤いてアク
セスできます。

(.-country myobj)
;; => "FR"

JavaScript のオブジェクトはミュータブルなので、set! 関数で新たな値を設定できます。

(set! (.-country myobj) "KR")

変換
前述のフォームの不便な点は、再帰的な変換を⾏わないため、ネストされたオブジェクトがある
場合、ネストされたオブジェクトは変換されないことです。ClojureScript のマップを使う場合と、
JavaScript のマップを使う場合を⽐べてみましょう。

(def clj-map {:country {:code "FR" :name "France"}})
;; => {:country {:code "FR", :name "France"}}
(:code (:country clj-map))
;; => "FR"

(def js-obj #js {:country {:code "FR" :name "France"}})
;; => #js {:country {:code "FR", :name "France"}
(.-country js-obj)
;; => {:code "FR", :name "France"}
(.-code (.-country js-obj))
;; => nil

このユースケースを解決するために、clj->js と js->clj が⽤意されています。それぞれ、
ClojureScript と JavaScript のコレクションを相互に変換します。ClojureScript への変換で
:country キーワードが⽂字列に変換されている点に注意してください。

(clj->js {:foo {:bar "baz"}})
;; => #js {:foo #js {:bar "baz"}}
(js->clj #js {:country {:code "FR" :name "France"}}))
;; => {"country" {:code "FR", :name "France"}}

配列の場合、期待通りに動作する into-array という特別な関数があります。

(into-array ["France" "Korea" "Peru"])
;; => #js ["France" "Korea" "Peru"]

配列
前例では、既存の ClojureScript コレクションから配列を作成しましたが、別の⽅法として

make-array 関数を使う⽅法もあります。次の例では、⻑さが 10 の事前に割り当てられた配列を作
成します。

(def a (make-array 10))
;; => #js [nil nil nil nil nil nil nil nil nil nil]

ClojureScript では、配列はシーケンスの抽象化にも適しているため、配列を繰り返し処理した
り、count 関数で単純に要素の数を取得したりすることができます。

(count a)
;; => 10

71



第 3章 ClojureScript ⼊⾨ 3.16 状態管理

JavaScript のプラットフォームでは、配列はミュータブルなコレクション型であるため、特定の
インデックスにアクセスし、その位置に値を設定できます。

(aset a 0 2)
;; => 2
a
;; => #js [2 nil nil nil nil nil nil nil nil nil]

もしくは、値を得るために、インデックスを指定してアクセスできます。

(aget a 0)
;; => 2

JavaScript では、配列においてインデックスでアクセスすることは、オブジェクトのプロパティ
へのアクセスと同じであるため、プレーンなオブジェクトと連携するために同じ関数を使うことが
できます。

(def b #js {:hour 16})
;; => #js {:hour 16}

(aget b "hour")
;; => 16

(aset b "minute" 22)
;; => 22

b
;; => #js {:hour 16, :minute 22}

3.16 状態管理
ClojureScript の根本的なアイデアの 1 つが不変性にあることを学びました。ClojureScript では
スカラー値もコレクションも不変ですが、Date のように JS のホスト環境に存在する可変型は例外
です。
不変性は多くの優れた特性がありますが、時とともに変化する値をモデル化する必要に迫られるこ
ともあります。データ構造を適切に変更できない場合、これを実現するにはどうすればよいでしょ
うか。

var

var は名前空間内で⾃由に再定義できますが、いつ変更されるかを知る⽅法はありません。var を
他の名前空間から再定義できないことは、やや制限的です。また、状態を変更している場合は、そ
れがいつ発⽣するかを知りたいと思うでしょう。

アトム
⾃由に変更できる値を含むオブジェクトとして、ClojureScript は Atom 型を提供します。値を変

更するだけでなく、付加したり切り離したりできる watcher 関数による監視や、アトムに含まれる
値が常に有効であることを確認するバリデーションもサポートします。
もし Ciri という名前の⼈に対応するアイデンティティをモデル化するとすれば、Ciriのデータを
含む不変な値を 1 つのアトムでラップすることができます。アトムの値は、deref 関数またはその
短縮表記 @ を使⽤して取得できます。

(def ciri (atom {:name "Cirilla" :lastname "Fiona" :age 20}))
;; #<Atom: {:name "Cirilla", :lastname "Fiona", :age 20}>

(deref ciri)
;; {:name "Cirilla", :lastname "Fiona", :age 20}

@ciri
;; {:name "Cirilla", :lastname "Fiona", :age 20}

swap! 関数を値を変更するためにアトムに使うことができます。Ciri の誕⽣⽇が今⽇なので、彼
⼥の年齢 age に加算しましょう。

(swap! ciri update :age inc)
;; {:name "Cirilla", :lastname "Fiona", :age 21}

@ciri
;; {:name "Cirilla", :lastname "Fiona", :age 21}

reset! 関数は、アトムに含まれる値を新しい値で置き換えます。

(reset! ciri {:name "Cirilla", :lastname "Fiona", :age 22})
;; {:name "Cirilla", :lastname "Fiona", :age 22}

@ciri
;; {:name "Cirilla", :lastname "Fiona", :age 22}

監視
アトムの監視関数を追加したり削除したりできます。アトムの値が swap! や reset! によって

変更されるたびに、アトムの監視関数を呼び出されます。監視は、add-watch 関数を⽤いて追加で
きます。それぞれの watcher には、後でアトムから監視を削除するために使⽤されるキー (例では
:logger)が関連付けられていることに注意してください。

(def a (atom))
(add-watch a :logger (fn [key the-atom old-value new-value]

(println "Key:" key "Old:" old-value "New:" new-value)))

(reset! a 42)
;; Key: :logger Old: nil New: 42
;; => 42

(swap! a inc)
;; Key: :logger Old: 42 New: 43
;; => 43

(remove-watch a :logger)

Volatile
Volatile は、アトムと同様に、変更可能な値を含むオブジェクトです。しかし、Volatile はアトム

が提供するような監視やバリデーションの機能をもっていません。これにより、多少パフォーマン
スが向上するため、監視やバリデーションが不要なミュータブルなコンテナに向いています。

Volatile の API はアトムのものとよく似ています。それらは、含まれる値を取得するために、参
照を解除して値を取得できます。vswap! でスワップを、vreset! でリセットをサポートします。

(def ciri (volatile! {:name "Cirilla" :lastname "Fiona" :age 20}))
;; #<Volatile: {:name "Cirilla", :lastname "Fiona", :age 20}>
(volatile? ciri)
;; => true
(deref ciri)
;; {:name "Cirilla", :lastname "Fiona", :age 20}

(vswap! ciri update :age inc)
;; {:name "Cirilla", :lastname "Fiona", :age 21}
(vreset! ciri {:name "Cirilla", :lastname "Fiona", :age 22})
;; {:name "Cirilla", :lastname "Fiona", :age 22}

アトムとのもう 1つの違いは、Volatilesのコンストラクタは、最後に ! をつける点です。Volatile
を作成するには volatile! を、アトムを作るには atom を使います。

第 4章

ツールとコンパイラ

本章では、ClojureScript を⽤いた開発を容易にするための既存のツールについて簡単に紹介しま
す。前章とは異なり、本章の内容は、それぞれが独⽴した内容になっています。

4.1 コンパイラの概要
前章の⾔語⾃体の理論的な説明にうんざりしてしまって、何かコードを書いて実⾏したいと思わ
れているかもしれません。このセクションの⽬的は、ClojureScript コンパイラの実⽤的な紹介を少
し⾏うことです。

ClojureScript のコンパイラは、多くのディレクトリと名前空間で分けられたソースコードを取得
して、JavaScript にコンパイルします。今⽇では JavaScript を実⾏できる環境は多くあり、それぞ
れに固有の特性があります。
この章では、ツールを追加せずに ClojureScript を使う⽅法を説明します。これは、他のツール

(Leiningen*1 や *2)が利⽤できない場合において、コンパイラの動作や利⽤法を理解するのに役⽴
ちます。

実⾏環境
実⾏環境とは何でしょうか。実⾏環境は、JavaScript を実⾏できるエンジンです。例えば、最も

⼀般的な実⾏環境はブラウザ (Chrome、Firefox等)であり、次に Node.js*3 が続きます。
他にも、Rhino(JDK6以降)、Nashorn(JDK8以降)、QtQuick(QT)等の実⾏環境がありますが、
どれも最初の 2 つと⼤きな違いはありません。ですので、今のところ ClojureScript は、ブラウザ
や Node.js ですぐに実⾏できるコードにコンパイルするでしょう。

コンパイラのダウンロード
ClojureScriptはセルフホストですが、ClojureScriptを使⽤する最善の⽅法は JVM実装を使⽤す

ることです。これを使⽤するには、JDK8がインストールされている必要があります。ClojureScript

*1 http://leiningen.org
*2 http://boot-clj.com
*3 https://Node.js.org

⾃体に必要なのは JDK7 のみですが、この章で使⽤するスタンドアロンコンパイラには JDK8
が 必 要 で す 。JDK8 は http://www.oracle.com/technetwork/java/javase/downloads/jdk8-
downloads-2133151.html にあります。
最新の ClojureScript コンパイラは、wget を使⽤してダウンロードできます。

wget https://github.com/clojure/clojurescript/releases/download/r1.9.36/cljs.jar

ClojureScript コンパイラはスタンドアロンの実⾏可能な jar ファイルにパッケージされているた
め、ClojureScript のソースコードを JavaScript にコンパイルするために必要なファイルはこれと
JDK8 のみです。

Node.js へのコンパイル
まず、Node.jsをターゲットにコードをコンパイルする例から始めましょう。この例では、Node.js

をインストールする必要があります。
Node.js をインストールする⽅法はいくつかありますが、Node.js のバージョンマネージャーの

nvm を使⽤することをお勧めします。インストール⽅法と nvm の使⽤法については、nvm のド
キュメント*4を参照してください。

nvm をインストールしたら、最新バージョンの Node.js をインストールします。

$ nvm install v6.2.0
$ nvm alias default v6.2.0

次のコマンドで Node.js がインストールされているかをテストしましょう。

$ node --version
v6.2.0

■アプリケーションの作成 実⽤的な例を作成する最初のステップとして、アプリケーションのた
めのディレクトリを作成します。実⽤例に使うコードをそこに取り込んでいきます。
まず、"Hello world" アプリケーションのためのディレクトリを作成しましょう。

$ mkdir -p myapp/src/myapp
$ touch myapp/src/myapp/core.cljs

*4 https://github.com/creationix/nvm

76



第 4章ツールとコンパイラ 4.1 コンパイラの概要

次のようなディレクトリ構造にします。

myapp
└── src

└── myapp
└── core.cljs

次に、前に作成した myapp/src/myapp/core.cljs に次のコードを書き込みます。

(ns myapp.core
(:require [cljs.nodejs :as nodejs]))

(nodejs/enable-util-print!)

(defn -main
[& args]
(println "Hello world!"))

(set! *main-cli-fn* -main)

ファイル内で宣⾔された名前空間が、ディレクトリの構造と正確に⼀致することが⾮常に重要で
す。これは、ClojureScript がソースコードを構築する⽅法です。

■アプリケーションのコンパイル このソース・コードをコンパイルするためには、ClojureScript
コンパイラにソース・ディレクトリと出⼒ファイルを伝える単純なビルド・スクリプトが必要です。
ClojureScript には他にも多くの選択肢がありますが、現時点では取り上げません。

myapp/build.clj を作成します。このファイルの内容は次のとおりです。

(require ’[cljs.build.api :as b])

(b/build "src"
{:main ’myapp.core
:output-to "main.js"
:output-dir "out"
:target :nodejs
:verbose true})

ここでは、この例で使⽤されているコンパイラのオプションについて簡単に説明します。

- :output-to は、コンパイルされたコードの出⼒先を指定します。この場合 main.js です。
- :main は、実⾏時にエントリーポイントとして動作する名前空間を指定します。
- :target は、コンパイルされたコードを実⾏するプラットフォームを⽰します。この例では、
Node.js を使⽤します。このパラメーターを省略した場合、ソースはコンパイルされてブラウザの
環境で実⾏されます。

コンパイルを実⾏するには、次のコマンドを実⾏します。

$ cd myapp
$ java -cp ../cljs.jar:src clojure.main build.clj

コンパイル後、node コマンドでコンパイルされたファイルを実⾏します。

$ node main.js
Hello world!

ブラウザへのコンパイル
このセクションでは、前のセクションの "Hello world" の例と同様のアプリケーションを作成し、
ブラウザの環境で実⾏します。このアプリケーションの最⼩要件は、JavaScript を実⾏できるブラ
ウザだけです。
プロセスはほぼ同じで、ディレクトリの構造も同じです。変更するのは、アプリケーションのエ
ントリーポイントとビルド・スクリプトだけです。前の例のディレクトリ・ツリーを元に、別のディ
レクトリを再⽣成します。

$ mkdir -p mywebapp/src/mywebapp
$ touch mywebapp/src/mywebapp/core.cljs

結果、次のディレクトリ構造になります。

mywebapp
└── src

└── mywebapp
└── core.cljs

次に、新しいコンテンツを mywebapp/src/mywebapp/core.cljs に書き込みます。

(ns mywebapp.core)

(enable-console-print!)

(println "Hello world!")

ブラウザの環境では、アプリケーションのための特定のエントリーポイントは必要ないため、全
体の名前空間がエントリーポイントになります。

■アプリケーションのコンパイル ソースコードをコンパイルしてブラウザで正しく動作させるに
は、mywebapp/build.clj を次のように上書きします。

(require ’[cljs.build.api :as b])

(b/build "src" {:output-to "main.js"
:source-map true
:output-dir "out/"
:main ’mywebapp.core
:verbose true
:optimizations :none})

ここで使⽤するコンパイラオプションについて簡単に説明します。

- :output-to は、コンパイルされたコードの出⼒先を指定します。この場合は main.js です。
- :main は、実⾏時にエントリーポイントとして動作する名前空間を指定します。
- :source-map は、ソースマップの出⼒先を指定します。ソースマップは、ClojureScript のソース
を、⽣成された JavaScript に接続して、エラー・メッセージが元のソースを⽰すようにします。
- :output-dir は、コンパイルで使われる全てのソースを保存するディレクトリを指定します。
ソースマップをソースだけでなく、残り全てのコードと正しく動作させるためのものです。
- :optimizations はコンパイルの最適化を⽰します。このオプションには異なる値がありますが、
これについては後のセクションで詳しく説明します。

コンパイルを実⾏するには、次のコマンドを実⾏します。

cd mywebapp
java -cp ../cljs.jar:src clojure.main build.clj

この処理には時間がかかることがありますが、⼼配せずに少し待ってください。Clojure コンパイ
ラを使⽤した JVM のブートストラップは少し遅いです。以降のセクションでは、このような遅い
プロセスの起動と停⽌を頻繁に⾏わないように、監視プロセスを開始する⽅法を説明します。
コンパイルを待つ間に、ダミーの HTML ファイルを作成して、ブラウザでサンプル・アプリケー
ションを簡単に実⾏できるように、ダミーの HTML ファイルを作成しましょう。以下の内容を
mywebapp ディレクトリの index.html に書き込みます。

<!DOCTYPE html>
<html>

<header>
<meta charset="utf-8" />
<title>Hello World from ClojureScript</title>

</header>
<body>

<script src="main.js"></script>
</body>

</html>

コンパイルが完了して基本的な HTML ファイルができたら、好きなブラウザで開いて、開発ツー
ルのコンソールで調べることができます。"Hello world!" というメッセージが表⽰されます。

監視プロセス
ClojureScript コンパイラの起動時間が遅いことに既に気づいているかもしれません。この問題を
解決するために、ClojureScript のコンパイラには、ソースコードの変更を監視して、ファイルの変
更がディスクに書き込まれると同時に再コンパイルするツールが付属しています。
別のビルド・スクリプトを watch.clj という名前で作成します。

(require ’[cljs.build.api :as b])

(b/watch "src"
{:output-to "main.js"
:output-dir "out/"
:source-map true
:main ’mywebapp.core
:optimizations :none})

ここで、前のセクションと同じようにスクリプトを実⾏します。

$ java -cp ../cljs.jar:src clojure.main watch.clj
Building ...
Reading analysis cache for jar:file:/home/niwi/cljsbook/playground/cljs.jar!/cljs/core.cljs
Compiling src/mywebapp/core.cljs
Compiling out/cljs/core.cljs
Using cached cljs.core out/cljs/core.cljs
... done. Elapsed 0.754487937 seconds
Watching paths: /home/niwi/cljsbook/playground/mywebapp/src

名前空間を mywebapp.core に戻して、出⼒テキストを Hello World, Again! に変更すると、
src/mywebapp/core.cljs がすぐにコンパイルされます。ブラウザで index.html をリロードすると、
新しい出⼒がコンソールに表⽰されます。この⽅法の別の利点は、出⼒が少し増えることです。

最適化レベル
ClojureScript コンパイラにはさまざまな最適化レベルがあります。これらのコンパイルレベルは
舞台裏で⾏われますが、Google Closure Compiler によります。

コンパイルのプロセスは次の通りです。

1. リーダーはコードを読み、何らかの分析を⾏います。この段階でコンパイラは警告を発する場合
があります。
2. その後、ClojureScript コンパイラは ClojureScript の 1 ファイルごとに 1 つの JavaScript の
出⼒ファイルを作成します。
3.⽣成された JavaScript のファイルは Google Closure コンパイラに渡され、最適化レベルや他の
オプション (ソースマップや output デイレクトリ)に応じて最終的な出⼒ファイルが⽣成されます。

最終的な出⼒形式は、選択した最適化レベルによって異なります。

■none この最適化レベルでは、⽣成された JavaScript は名前空間ごとに別々の出⼒ファイルに
書き込まれ、コードへの追加の変換はありません。

■whitespace この最適化レベルでは、⽣成された JavaScript ファイルが依存関係の順で出⼒
ファイルに連結されます。改⾏やその他の空⽩は削除されます。
これによりコンパイル速度が多少低下し、コンパイルが遅くなります。ただし、処理速度はそれ
ほど遅くはなく、中⼩規模のアプリケーションには⼗分に使⽤できます。

■simple 単純なコンパイルレベルは、whitespace 最適化レベルからの作業に基づいて構築され
ます。ローカル変数や関数パラメータの名前をより短い名前に変更するなど、式や関数内でさらに
最適化を実⾏します。

:simple 最適化によるコンパイルは、常に構⽂的に有効な JavaScript の機能を保持するため、コ
ンパイルされた ClojureScript と他の JavaScript との間の相互作⽤に影響はありません。

■advanced ⾼度なコンパイルレベルは simple 最適化レベルの上に構築され、さらに積極的な
最適化とデッドコードの削除を実⾏します。これにより、出⼒ファイルが⼤幅に⼩さくなります。

:advanced 最適化は、Google Closure Compiler ルールに従った JavaScript の厳密なサブセッ
トでのみ動作します。ClojureScript はこの厳密なサブセット内に有効な JavaScript を⽣成します
が、サード・パーティーの JavaScript コードを操作する場合には、全てを期待どおりに動作させる
ために追加作業が必要になります。
サードパーティの JavaScript ライブラリとのやり取りについては、後のセクションで説明します。

4.2 REPLでの作業
導⼊
ソースファイルを作成して、ClojureScript で何かを試したいときに毎回コンパイルすることもで

きますが、REPL を使⽤するのが簡単です。REPL とは、Read(キーボードかた⼊⼒を読みとり)、
Evaluate(⼊⼒を評価して)、Print(結果を出⼒する)、Loop(⼊⼒に戻り繰り返す)の頭⽂字をとった
ものです。⾔い換えると、REPL によって ClojureScript のコンセプトを試して、即座にフィード
バックを得ることができます。

ClojureScript には、様々な実⾏環境で REPL を実⾏するためのサポートが⽤意されていますが、
それぞれに⻑所と短所があります。たとえば、Node.js では REPL を実⾏できますが DOM にアク
セスできません。どの REPL 環境が最適かは、個々のニーズと要件によって異なります。

Nashorn REPL

Nashorn REPLは、特別なものを必要とせず、ClojureScript コンパイラを実⾏するために、以前
の例で使⽤した JVM(JDK8)のみを必要とするため、最も簡単で、おそらく最も苦痛のない REPL
環境です

REPL の Playground のために新規でスクリプトファイル repl.clj を作成して、次のように編集
します。

(require ’[cljs.repl]
’[cljs.repl.nashorn])

(cljs.repl/repl
(cljs.repl.nashorn/repl-env)
:output-dir "out"
:cache-analysis true)

次のコマンドにより REPL を起動して実⾏します。

$ java -cp cljs.jar:src clojure.main repl.clj
To quit, type: :cljs/quit
cljs.user=> (+ 1 2)
3

お気付きかもしれませんが、REPL には、ヒストリーやその他のシェルのような機能がサポート
されていません。これは、デフォルトの REPL では readline がサポートされていないためです。
この問題は rlwrap という名前の簡単なツールを使えば解決できます。このツールはオペレーティ
ングシステムのパッケージマネージャで⾒つけることができるはずです (Ubuntu の場合、sudo apt
install -y rlwrap でインストールします)。

rlwrap により REPL で readline が機能して、コマンドの履歴やコードナビゲーションなどの
シェルのようなユーティリティが使うことができ、REPL をより快適に使うことができます。これ
を使⽤するには、REPL を開始するためには、使⽤した前のコマンドの前に rlwrap を追加します。

$ rlwrap java -cp cljs.jar:src clojure.main repl.clj
To quit, type: :cljs/quit
cljs.user=> (+ 1 2)
3

Node.js REPL

まず初めに Node.js が必要です。この REPL を使⽤するには、Node.js をシステムにインストー
ルしてください。
外部依存関係を持たない Nashorn REPL がすでに使⽤可能になっているのに、なぜ Node.js が

必要なのか不思議に思うかもしれません。答えは⾮常に単純です。Node.js はバックエンドで最も
よく使われている JavaScript 実⾏環境であり、その上に構築された数多くのコミュニティーによる
パッケージが存在するからです。
幸いなことに、Node.js の REPLは、⼀度システムにインストールすれば⾮常に簡単に起動でき
ます。次の内容を tools.clj に追加します。

(require ’[cljs.repl]
’[cljs.repl.node])

(cljs.repl/repl
(cljs.repl.node/repl-env)
:output-dir "out"
:cache-analysis true)

以前 Nashorn REPL で⾏ったように REPL を開始します。

$ rlwrap java -cp cljs.jar:src clojure.main repl.clj
To quit, type: :cljs/quit
cljs.user=> (+ 1 2)
3

Browser REPL

この REPL は、起動と実⾏に最も⼿間がかかります。これは実⾏環境のためにブラウザを使⽤し
ており、追加の要件があるためです。
まず、brepl.clj という名前のファイルを作成して、次のように編集します。

(require
’[cljs.build.api :as b]
’[cljs.repl :as repl]
’[cljs.repl.browser :as browser])

(b/build "src"
{:output-to "main.js"
:output-dir "out/"
:source-map true
:main ’myapp.core
:verbose true
:optimizations :none})

(repl/repl (browser/repl-env)
:output-dir "out")

このスクリプトは、先ほどと同じようにソースをビルドして REPL を開始します。
ブラウザの REPL は、REPL が動作する前にブラウザ内でコードを実⾏することも要求します。

そのために、前のセクションで使⽤したものとよく似たアプリケーションの構造を再作成します。

$ mkdir -p src/myapp
$ touch src/myapp/core.cljs

次に、新しいコンテンツを src/myapp/core.cljs に書き込みます。

(ns myapp.core
(:require [clojure.browser.repl :as repl]))

(defonce conn
(repl/connect "http://localhost:9000/repl"))

(enable-console-print!)

(println "Hello, world!")

最後に、ブラウザ側のコードを実⾏するためのエントリーポイントとして使⽤する index.html を
作成します。

<!DOCTYPE html>
<html>

<header>
<meta charset="utf-8" />
<title>Hello World from ClojureScript</title>

</header>
<body>

<script src="main.js"></script>
</body>

</html>

多くの設定をしましたが、実際に動いているところを⾒てみると、それだけの価値があることは
間違いありません。前例と同じ⽅法で brepl.clj を実⾏します。

$ rlwrap java -cp cljs.jar:src clojure.main brepl.clj
Compiling client js ...
Waiting for browser to connect ...

最後に、お気に⼊りのブラウザを開き、http://localhost:9000/. にアクセスします。ページが
ロードされるた後 (ページは空⽩です)、REPL を実⾏したコンソールに戻り、REPL が実⾏されて
いることを確認しましょう。

[...]
To quit, type: :cljs/quit
cljs.user=> (+ 14 28)
42

ブラウザ REPL の⼤きな利点の 1 つは、ブラウザ環境のすべてにアクセスできることです。たと
えば、REPL に (js/alert "hello world") と⼊⼒すると、ブラウザに警告ボックスが表⽰されます。

4.3 Google Closure Library

Google Closure Library は Google が開発している JavaScript ライブラリです。モジュールの
アーキテクチャを備えており、DOM 操作、イベント処理、Ajax 、JSON を扱うためにクロスブラ
ウザの関数を提供します。

Google Closure Library は、ClojureScript コンパイラで内部的に使⽤されている Closure
Compiler を利⽤するために書かれています。

ClojureScript は Google Closure コンパイラと Closure Library の上に構築されています。実
際、ClojureScript の名前空間は Closure モジュールです。これは、Closure Library を⾮常に簡単
に操作できることを意味します。

(ns yourapp.core
(:require [goog.dom :as dom]))

(def element (dom/getElement "body"))

このコードは、Closure Library の dom モジュールをインポートして、そのモジュールで宣⾔さ
れた関数を使⽤する⽅法を⽰しています。
さらに、Closure Library は、クラスまたはオブジェクトのように動作する「特別な」モジュール
を公開します。これらの機能を使⽤するには、(ns ...) の形式で :import ディレクティブを使⽤
する必要があります。

(ns yourapp.core
(:import goog.History))

(def instance (History.))

Clojure のプログラムでは、Java クラスをインポートするためにホスト環境である Java と相互
運⽤するために、:import ディレクティブを使⽤します。ただし、ClojureScript で型 (クラス)を
定義する場合は、:import ディレクティブではなく、標準の :require ディレクティブを使⽤する
必要があります。

4.4 依存関係の管理
ここまでは、組み込みの ClojureScript のツールを使ってソース・ファイルを JavaScript にコン

パイルしてきました。次は、外部またはサード・パーティ (あるいはその両⽅)の依存関係を管理す
る⽅法について説明します。

このため、この章の残りの部分では、ClojureScript プロジェクトを構築するために、事実上ビル
ドと依存関係管理ツールのデファクトである Leiningen の使い⽅を説明します。boot の⼈気も⾼
まっていますが、本書では Leiningen に限定します。

Leiningen のインストール
Leiningen のインストール⽅法は⾮常に簡単です。次の⼿順に従ってください。

$ mkdir ~/bin
$ cd ~/bin
$ wget https://raw.githubusercontent.com/technomancy/leiningen/stable/bin/lein
$ chmod a+x ./lein
$ export PATH=$PATH:~/bin

~/bin ディレクトリが常にパスに設定されていることを確認します。永続的にするには、export
で始まる⾏を ~/.bashrc に追加します。(bash を使⽤していると仮定します)。
ターミナルを別で開いて lein version を実⾏します。次のように表⽰されます。

$ lein version
Leiningen 2.5.1 on Java 1.8.0_45 OpenJDK 64-Bit Server VM

ここでは、Linuxや BSDのような Unix系のシステムを使っていると仮定しています。Windows
ユーザの⽅は、Leiningen のホームページの説明を参照してください。Linux、Mac OS X、BSD 版
の Leiningen スクリプトは Web サイトから⼊⼿することもできます。

初めてのプロジェクト
ツールの動作を紹介する最良の⽅法は、ツールを使⽤して toy プロジェクトを作成することです。

この例では、閏 (うるう)年かどうかを判別する⼩さなアプリケーションを作成します。
まず、Leiningen の mies テンプレートを使⽤します。
テンプレートは、初めのプロジェクト構造を作成するための Leiningen の機能です。Clojure コ

ミュニティには⾮常に多くのものがあります。この例では、ClojureScript の中⼼的な開発者が作成
した mies テンプレートを使⽤します。テンプレートの詳細については、Leiningen のドキュメント
を参照してください。
プロジェクトのレイアウトを作成します。

$ lein new mies leapyears
$ cd leapyears # ⽣成したプロジェクトのディレクトリに移動する

プロジェクトの構造は次のとおりです。

leapyears
├── index.html
├── project.clj
├── README.md
├── scripts
│ ├── build
│ ├── release
│ ├── watch
│ ├── repl
│ └── brepl
└── src

└── leapyears
└── core.cljs

project.clj には、Leiningen が依存関係をダウンロードしてプロジェクトをビルドするために使
⽤する情報が含まれています。今のところ、project.clj はこういうものだと思ってください。

index.htmlファイルを開き、body の先頭に次の内容を追加します。

<section class="viewport">
<div id="result">

----
</div>
<form action="" method="">

<label for="year">Enter a year</label>
<input id="year" name="year" />

</form>
</section>

次の⼿順では、フォームをインタラクティブにするコードを追加します。次のコードを
src/leapyears/core.cljs に追加します。

(ns leapyears.core
(:require [goog.dom :as dom]

[goog.events :as events]
[cljs.reader :refer (read-string)]))

(enable-console-print!)

(def input (dom/getElement "year"))
(def result (dom/getElement "result"))

(defn leap?
[year]
(or (zero? (js-mod year 400))

(and (pos? (js-mod year 100))
(zero? (js-mod year 4)))))

(defn on-change
[event]
(let [target (.-target event)

value (read-string (.-value target))]
(if (leap? value)

(set! (.-innerHTML result) "YES")
(set! (.-innerHTML result) "NO"))))

(events/listen input "keyup" on-change)

次のようにして、ClojureScript コードをコンパイルします。

$ ./scripts/watch

バックグラウンドでは、監視スクリプトは lein ビルドツールを使⽤して、前のセクションで説明
した Java の build コマンドに似たコマンドを実⾏します。

rlwrap lein trampoline run -m clojure.main scripts/watch.clj

システムに rlwrap がインストールされている必要があります。
最後に、ブラウザで index.html を開きます。テキストボックスに年を⼊⼒すると、うるう年の状

態が表⽰されます。
scripts ディレクトリには、build や release などのファイルがあります。これらは、前のセクショ
ンで⾔及したビルド・スクリプトと同じですが、ここでは watch を使います。

依存関係の管理
ClojureScript のコンパイルプロセスに Leiningen を使⽤する本当の⽬的は、依存関係の取得を

⾃動化することです。これは、⼿動で取得するよりも⼤幅に簡単です。
依存関係は、ほかのパラメータと⼀緒に project.clj で宣⾔されます。このファイルの形式は次の
通りです (miesテンプレートより)。

(defproject leapyears "0.1.0-SNAPSHOT"
:description "FIXME: write this!"
:url "http://example.com/FIXME"
:dependencies [[org.clojure/clojure "1.8.0"]

[org.clojure/clojurescript "1.9.36"]
[org.clojure/data.json "0.2.6"]]

:jvm-opts ^:replace ["-Xmx1g" "-server"]
:node-dependencies [[source-map-support "0.3.2"]]
:plugins [[lein-npm "0.5.0"]]
:source-paths ["src" "target/classes"]
:clean-targets ["out" "release"]
:target-path "target")

ClojureScript に関連するプロパティについて簡単に説明します。
:dependencies は、プロジェクトに必要な依存関係を含むベクタです。
:clean-targets は、lein clean により削除されるパスを含むベクタです。

ClojureScript の依存関係は jar ファイルを使ってパッケージ化されます。Clojure やその他の
JVM ⾔語を使⽤している場合、jar ファイルは⾮常になじみ深いものです。しかし、それらに慣れ
ていなくても⼼配する必要はありません。jarファイルは、プロジェクトを含む単純な zip ファイル
のようなものであり、ライブラリ⽤の project.clj、メタデータの⼀部、および ClojureScript のソー
スを含みます。パッケージングについては別のセクションで説明します。

Clojure パッケージは Clojars で公開されることが多いです。また、ClojureScript Wiki には多
くのサードパーティのライブラリがあります。

外部の依存関係
ClojureScript には存在しませんが、すでに JavaScript に実装されていて、プロジェクトで使い
たいライブラリがある場合があります。
含めるライブラリによって、多くの⽅法があります。いくつか⾒ていきましょう。

Closure 対応のライブラリ
Google Closure モジュールとの互換性を持つように作成されたライブラリをプロジェクトに含め
る場合には、そのライブラリをソース (クラスパス)に配置して、他の Clojure 名前空間と同じよう
にアクセスします。

これは最も単純なケースです。Google Closure モジュールは直接互換性があり、Google Closure
モジュールシステムを使って書かれた JavaScript コードは、あなたの Clojure のコードと、追加の
⼿順なしでミックスできます。

mies テンプレートを使⽤して、新しいプロジェクトを作成してみましょう。

$ lein new mies myextmods
$ cd myextmods

実験⽤に簡単な Google Closure モジュールを作成します。

src/myextmods/myclosuremodule.js

goog.provide("myextmods.myclosuremodule");

goog.scope(function() {
var module = myextmods.myclosuremodule;
module.get_greetings = function() {

return "Hello from google closure module.";
};

});

では、REPL を開き、名前空間を reqire して、expose された関数を使⽤してみます。
you can open the nodejs repl just executing ./scripts/repl on the root of the repository. レポ
ジトリのルートで ./scripts/repl を実⾏するだけで、Node.js の REPL を開くことができます。

(require ’[myextmods.myclosuremodule :as cm])
(cm/get_greetings)
;; => "Hello from google closure module."

CommonJS モジュール互換のライブラリ

Node.js の⼈気により、Node.js で⽤いられる CommonJS の⼈気は、現在 JavaScript のライブ
ラリで最も使⽤されているモジュール形式です。サーバサイド開発で使⽤されるか、ブラウザ側の
アプリケーションで使⽤されるかにより、利⽤法は別々です。

CommonJS のモジュール形式 (Google Closure モジュールを⽤いた前例と類似点が多い)を使⽤
した単純なファイルを作成します。

src/myextmods/mycommonjsmodule.js

function getGreetings() {
return "Hello from commonjs module.";

}

exports.getGreetings = getGreetings;

後に、このシンプルな pet ライブラリを使うために、ClojureScript コンパイラにファイルのパス
と、使⽤するモジュールの型を :foreign-libs 属性で指定する必要があります。

scripts/repl.clj を開いて、次のように変更します。

(require
’[cljs.repl :as repl]
’[cljs.repl.node :as node])

(repl/repl
(node/repl-env)
:language-in :ecmascript5
:language-out :ecmascript5
:foreign-libs [{:file "myextmods/mycommonjsmodule.js"

:provides ["myextmods.mycommonjsmodule"]
:module-type :commonjs}]

:output-dir "out"
:cache-analysis false)

直接的なパスはこの pet ライブラリを指すために使⽤されますが、リモートリソースへの完全な
URI を指定することができ、⾃動的にダウンロードされます。
ここで、./scripts/repl.clj を⽤いた ./scripts/repl を実⾏して、REPL 内で moment を使ってみ

ます。

(require ’[myextmods.mycommonjsmodule :as cm])
(cm/getGreetings)
;; => "Hello from commonjs module."

従来のモジュール化されていない (グローバルスコープの)ライブラリ

今⽇では、何らかのモジュールを使⽤してライブラリをパッケージ化することが⾮常に⼀般的に
なっていますが、グローバルオブジェクトを公開するだけで、モジュールを全く使⽤しないライブ
ラリも数多くありますが、ClojureScript から使うことができます。

グローバルオブジェクトを公開するライブラリを使⽤するには、CommoJS モジュールと同様の
⼿順を実⾏する必要がありますが、:module-type 属性を省略する点が異なります。
これにより、名前空間の js/ を介してグローバルオブジェクトにアクセスするために必要な名前
空間の合成が作成されます。名前空間は、その背後にあるオブジェクトを公開せず、その依存関係
が必要であることをコンパイラに⽰すだけなので、synthetic と呼ばれます。
グローバル関数のみを宣⾔する単純なファイルを作成します。

src/myextmods/myglobalmodule.js

function getGreetings() {
return "Hello from global scope.";

}

scripts/repl.clj を開いて、次のように変更します。

(require
’[cljs.repl :as repl]
’[cljs.repl.node :as node])

(repl/repl
(node/repl-env)
:language-in :ecmascript5
:language-out :ecmascript5
:foreign-libs [{:file "myextmods/mycommonjsmodule.js"

:provides ["myextmods.mycommonjsmodule"]
:module-type :commonjs}

{:file "myextmods/myglobalmodule.js"
:provides ["myextmods.myglobalmodule"]}]

:output-dir "out"
:cache-analysis false)

前の例と同様に、REPL 内で評価してみましょう。

(require ’myextmods.myglobalmodule)
(js/getGreetings)
;; => "Hello from global scope."

4.5 Unit テスト
ご想像のとおり、ClojureScript でのテストは、Clojure、Java、Python、JavaScript などの他の

⾔語で広く使⽤されているものと同じ概念で構成されています。
⾔語に関係なく、Unit テストの主な⽬的は、いくつかのテスト・ケースを実⾏し、テスト対象の
コードが予期したとおりに動作し、予期しない例外を発⽣させずに戻ることを検証することです。

ClojureScript のデータ構造の不変性は、プログラムのエラーを少なくし、テストを容易にするの
に役⽴ちます。ClojureScript のもう⼀つの利点は、複雑なオブジェクトではなく、平易なデータを
使⽤する傾向があることです。このため、テスト⽤の「モック」オブジェクトの作成が⼤幅に簡略
化されます。

初めの⼀歩
公式の ClojureScript のテスト・フレームワークは、名前空間 cljs.test にあります。これは⾮常

に単純なライブラリですが、私たちの⽬的には⼗分すぎるはずです。
他にも、追加の機能を提供したり、テストに対して直接的に異なるアプローチを提供するライブ
ラリ (test.check 等) がありますが、ここでは説明しません。
テストを試すために mies テンプレートを使⽤して、新しいプロジェクトを作成します。

$ lein new mies mytestingapp
$ cd mytestingapp

このプロジェクトには、依存関係管理のセクションで説明したのと同じレイアウトなので、ここ
では説明しません。次に、テスト⽤のディレクトリツリーを作成します。

$ mkdir -p test/mytestingapp
$ touch test/mytestingapp/core_tests.cljs

また、この新しく作成されたテストディレクトリで動作するように、既存の watch.clj を改変する
必要があります。

(require ’[cljs.build.api :as b])

(b/watch (b/inputs "test" "src")
{:main ’mytestingapp.core_tests
:target :nodejs
:output-to "out/mytestingapp.js"
:output-dir "out"
:verbose true})

この新しいスクリプトは、src ディレクトリと test ディレクトリの両⽅をコンパイルして監視し、
新しいエントリーポイントを 名前空間 mytestingapp.core_tests に設定します。
次に、core_tests.cljs にテスト・コードを追加します。

(ns mytestingapp.core-tests
(:require [cljs.test :as t]))

(enable-console-print!)

(t/deftest my-first-test
(t/is (= 1 2)))

(set! *main-cli-fn* #(t/run-tests))

このコードスニペットに関連する部分は以下の通りです。

(t/deftest my-first-test
(t/is (= 1 2)))

deftestマクロは、テストを定義するための基本的な要素です。最初のパラメータとして名前を取
り、続いて is マクロを使⽤した 1 つ以上のアサーションを取ります。この例では、(=1 2) が真で
あることを assert しようとします。
これを実⾏してみましょう。まず、監視プロセスを開始します。

$ ./scripts/watch
Building ...
Copying jar:file:/home/niwi/.m2/repository/org/clojure/clojurescript/1.9.36/clojurescript-1.9.36.jar!/cljs/core.cljs to out/cljs/core.cljs
Reading analysis cache for jar:file:/home/niwi/.m2/repository/org/clojure/clojurescript/1.9.36/clojurescript-1.9.36.jar!/cljs/core.cljs
Compiling out/cljs/core.cljs
... done. Elapsed 3.862126827 seconds
Watching paths: /home/niwi/cljsbook/playground/mytestingapp/test, /home/niwi/cljsbook/playground/mytestingapp/src

コンパイルが完了したら、Node.js を使⽤して、コンパイルされたファイルの実⾏を試みます。

$ node out/mytestingapp.js
Testing mytestingapp.core-tests

FAIL in (my-first-test) (cljs/test.js:374:14)
expected: (= 1 2)

actual: (not (= 1 2))
Ran 1 tests containing 1 assertions.
1 failures, 0 errors.

予想される assert のエラーがコンソールに正常に出⼒されたことがわかります。テストを修正す
るには、not= を付けて = を変更して、ファイルを再度実⾏します。

$ node out/mytestingapp.js

Testing mytestingapp.core-tests

Ran 1 tests containing 1 assertions.
0 failures, 0 errors.

このようなアサーションのテストはあまり役に⽴ちません。アプリケーション・コードをテ
ストしてみましょう。そのために、うるう年かどうかを調べる関数を使います。次の内容を
src/mytestingapp/core.clj に書き込みます。

(defn leap?
[year]
(and (zero? (js-mod year 4))

(pos? (js-mod year 100))
(pos? (js-mod year 400))))

次に、新しい leap? 関数が正しく動作するかを確認するために、新しいテスト・ケースを作成し
ます。core_tests.cljs は次のようになります。

(ns mytestingapp.core-tests
(:require [cljs.test :as t]

[mytestingapp.core :as core]))

(enable-console-print!)

(t/deftest my-first-test
(t/is (not= 1 2)))

(t/deftest my-second-test
(t/is (core/leap? 1980))
(t/is (not (core/leap? 1981))))

(set! *main-cli-fn* #(t/run-tests))

コンパイル済みのファイルを再度実⾏して、2 つのテストが実⾏されていることを確認します。最
初のテストは以前と同じように合格し、2 つの新しいうるう年のテストも合格します。

⾮同期のテスト
ClojureScript の特徴の 1 つは、⾮同期のシングルスレッド実⾏環境で実⾏されることですが、こ

れには課題があります。
⾮同期実⾏環境では、⾮同期関数をテストできるはずです。このため、ClojureScript のテストラ

イブラリには⾮同期マクロが⽤意されており、⾮同期コードでうまく動作するテストを作成するこ
とができます。
まず、⾮同期的に動作する関数を作成する必要があります。この⽬的のために、同じ操作を⾏う
けども、コールバックを使⽤して⾮同期に結果を返す述部 async-leap? を作成します。

(defn async-leap?
[year callback]
(js/setImmediate
(fn []

(let [result (or (zero? (js-mod year 400))
(and (pos? (js-mod year 100))

(zero? (js-mod year 4))))]
(callback result)))))

JavaScript の関数 setImmediate は⾮同期タスクをエミュレートするために使⽤され、コール
バックはその述部の結果によって実⾏されます。
これをテストするには、前述の async マクロを使⽤してテストケースを作成する必要があります。

(t/deftest my-async-test
(t/async done

(core/async-leap? 1980 (fn [result]
(t/is (true? result))
(done)))))

⾮同期マクロによって公開される done 関数は、⾮同期操作が終了して、すべてのアサーション
が実⾏された後に呼び出す必要があります。

done 関数は⼀度だけ実⾏することが⾮常に重要です。省略、または 2 回実⾏すると、異常な動作
が発⽣する可能性があるため、使⽤しないでください。

4.6 CI との連携
⼤半の CI ツールおよびサービスは、提供するテストスクリプトが標準の終了コードを返すこと
を前提としています。しかし、JavaScript には ClojureScript が使⽤できる汎⽤の終了コード API
がないため、ClojureScript のテストフレームワークでは、設定なしにこの終了コードをカスタマイ
ズすることができません。

97



第 4章ツールとコンパイラ 4.6 CI との連携

この問題を解決するため、ClojureScript のテストフレームワークは、テストが完了した後にカス
タムコードを実⾏する⼿段を提供します。ここでは、最終テストのステータス (成功の場合は 0、失
敗の場合は 1) に応じて、環境固有の終了コードを設定する必要があります。
このコードを core_tests.cljs の最後に挿⼊します。

(defmethod t/report [::t/default :end-run-tests]
[m]
(if (t/successful? m)

(set! (.-exitCode js/process) 0)
(set! (.-exitCode js/process) 1)))

これで、テストスクリプトの終了コードを実⾏後に確認できます。

$ node out/mytestingapp.js
$ echo $?

このコード・スニペットは明らかに、Node.js を使⽤してテストを実⾏していることを前提として
います。別の実⾏環境でスクリプトを実⾏している場合は、その環境で終了コードを設定する⽅法
と、それに応じて前のスニペットを変更する⽅法を理解しておく必要があります。

ClojureScript 発展編

5.1 Transducer

データの変換
ClojureScript では、シーケンス抽象化の観点から、データ変換のための豊富な語彙が⽤意されて
います。コレクション処理の関数を組み合わせて新しい関数を作成する⽅法を⾒てみましょう。こ
の章では、ブドウの房 (cluster)を分け、腐った (rotten)房を分別して、きれいにするという単純な
問題を扱います。次のようなブドウの房のコレクションがあります。

(def grape-clusters
[{:grapes [{:rotten? false :clean? false}

{:rotten? true :clean? false}]
:color :green}

{:grapes [{:rotten? true :clean? false}
{:rotten? false :clean? false}]

:color :black}])

私たちは、ブドウの房を個々のブドウに分け、腐ったものを捨て、残りのブドウを⾷べられるよ
うにきれいにしようとしています。このデータ変換の作業を⾏うための装備が ClojureScript には
⼗分に備わっています。おなじみの map 、filter、mapcat 関数を使って実装できます。

(defn split-cluster
[c]
(:grapes c))

(defn not-rotten
[g]
(not (:rotten? g)))

(defn clean-grape
[g]
(assoc g :clean? true))

(->> grape-clusters
(mapcat split-cluster)
(filter not-rotten)
(map clean-grape))
;; => ({rotten? false :clean? true}
;; {:rotten? false :clean? true})

上記の例では、ブドウの選択と洗浄の問題を上⼿に解決していますが、partial application と 関
数合成を使いながら mapcat、filter、map を組み合わせることによって、このような変換を抽象
化することもできます。

(def process-clusters
(comp
(partial map clean-grape)
(partial filter not-rotten)
(partial mapcat split-cluster)))

(process-clusters grape-clusters)
;; => ({rotten? false :clean? true}
;; {:rotten? false :clean? true})

コードは⾮常にきれいですが、いくつか問題があります。たとえば、map filter mapcat の各
呼び出しは、遅延はあっても破棄される中間結果を⽣成するシーケンスを消費して⽣成します。各
シーケンスは次のステップに渡され、シーケンスを返します。ブドウの房のコレクションを 1回の
横断で変換できるとしたら素晴らしいと思いませんか。

process-clusters 関数は、どんなシーケンスにも機能しますが、シーケンス以外には再利⽤
できないことも問題です。ブドウの房のコレクションをメモリ上で利⽤可能にするのではなく、ス
トリームの中で⾮同期にプッシュされることを想像してみてください。このような状況では、map

filter mapcat などは、型に対応した具体的な実装があるため、process-clusters 関数を再利⽤
できなかったのです。

プロセス変換への⼀般化
map filter mapcat を⾏うプロセスは、必ずしも具体的な型に結び付けられているわけではありま
せんが、各々の型に対して再実装していきます。⽂脈に依存しないように、このようなプロセスを
⼀般化する⽅法を⾒てみましょう。まず、map と filter の単純なバージョンを実装して、内部で
どのように動作するかを⾒てみましょう。

(defn my-map
[f coll]
(when-let [s (seq coll)]

(cons (f (first s)) (my-map f (rest s)))))

(my-map inc [0 1 2])
;; => (1 2 3)

(defn my-filter
[pred coll]
(when-let [s (seq coll)]

(let [f (first s)
r (rest s)]

(if (pred f)
(cons f (my-filter pred r))
(my-filter pred r)))))

(my-filter odd? [0 1 2])
;; => (1)

どちらも seqable を受け取り、シーケンスを返すことを前提としています。多くの再帰関数と同
様に、これらはおなじみの reduce 関数として実装できます。reduce 関数はアキュムレータと⼊⼒
を受け取り、次のアキュムレータを返します。今後は、このような関数を reducing 関数 (reduce を
⾏う関数) と呼びます。

(defn my-mapr
[f coll]
;; reducing function
(reduce (fn [acc input]

(conj acc (f input)))
;; initial value
[]
;; collection to reduce
coll))

(my-mapr inc [0 1 2])
;; => [1 2 3]

(defn my-filterr
[pred coll]
;; reducing function
(reduce (fn [acc input]

(if (pred input)
(conj acc input)
acc))

;; initial value
[]
;; collection to reduce
coll))

(my-filterr odd? [0 1 2])
;; => [1]

reduce を使⽤すると、シーケンスだけでなく、還元可能なあらゆるものに対して関数が動作
するようになるので、以前のバージョンはより⼀般的なものになりました。しかし、my-mapr と
my-filterr はソース (coll) について何も知らないにもかかわらず、 reduce の初期値 ( [] ) と
reduce 関数の本体でハードコードされた conj 演算によって、出⼒ (ベクタ)と結びついているこ
とに気づかれたかも知れません。結果を別のデータ構造、例えば遅延シーケンスに蓄積することも
できますが、そのためには関数を書き直さなければなりません。
これらの関数を本当の意味で汎⽤的にするにはどうすればよいのでしょうか。これらの関数は、
変換中の⼊⼒のソースや、⽣成された出⼒について知るべきではありません。conj は reducing 関
数の 1つであることにお気づきでしょうか。アキュムレータと⼊⼒を受け取り、別のアキュムレー
タを返します。したがって、my-mapr と my-filterr が使⽤する reducing 関数をパラメータ化し
ても、ビルドする結果の型については何も知りません。試してみましょう。

(defn my-mapt
;; function to map over inputs
[f]
;; parameterised reducing function
(fn [rfn]

;; transformed reducing function, now it maps f !
(fn [acc input]

(rfn acc (f input)))))

(def incer (my-mapt inc))

(reduce (incer conj) [] [0 1 2])
;; => [1 2 3]

(defn my-filtert
;; predicate to filter out inputs
[pred]
;; parameterised reducing function
(fn [rfn]

;; transformed reducing function,
;; now it discards values based on pred !
(fn [acc input]

(if (pred input)
(rfn acc input)
acc))))

(def only-odds (my-filtert odd?))

(reduce (only-odds conj) [] [0 1 2])
;; => [1]

⾼階関数が多いので、何が起こっているのかを理解するために分解してみましょう。my-mapt が
どのように機能するかを段階的に検証していきます。my-filtert の仕組みは似ているので、ここ
では省略します。
まず、my-mapt はマッピング関数を取ります。この例では、inc を指定して別の関数を返して
います。f を inc に置き換えて、build しているものを確認しましょう。

(def incer (my-mapt inc))
;; (fn [rfn]
;; (fn [acc input]
;; (rfn acc (inc input))))
;; ^^^

結果として得られる関数は、それがデリゲートする reducing 関数を受け取るようにパラメータ化
されています。それを cond と⼀緒に呼んだときに何が起こるか⾒てみましょう。

(incer conj)
;; (fn [acc input]
;; (conj acc (inc input)))
;; ^^^^

⼊⼒の変換のために inc 、結果を蓄積するために conj を使⽤する reducing 関数を取り戻しま
す。本質的に、reducing 関数の変換として map 定義しました。ある reducing 関数を別の reducing
関数に変換する関数のことを ClojureScript では transducers と呼びます。

transducer の⼀般性を⽰すために、reduce の呼び出しにおいて、異なるソースと出⼒先を使⽤し
てみましょう。

(reduce (incer str) "" [0 1 2])
;; => "123"

(reduce (only-odds str) "" ’(0 1 2))
;; => "1"

map 及び filter の transducer版は、⼊⼒をソースから出⼒先に運ぶプロセスを変換しますが、
⼊⼒がどこから来てどこで終わるかは知りません。これらの実装においては、⽂脈に関係なく⽬的
を達成することの本質が含まれています。

transducer についての知識が増えたので、独⾃のバージョンの mapcat を実装できます。私たち
はすでにその基本的な部分を持っています。map transducer です。mapcat が⾏うことは、関数を
⼊⼒上にマップし、結果の構造を 1レベルの平らさにすることです。結合する箇所を transducer と
して実装してみましょう。

(defn my-cat
[rfn]
(fn [acc input]

(reduce rfn acc input)))

(reduce (my-cat conj) [] [[0 1 2] [3 4 5]])
;; => [0 1 2 3 4 5]

my-cat transducer は、その⼊⼒をアキュムレータに結合する reducing 関数を返します。このよ
うにして、rfn 関数で reduce 可能な⼊⼒を reduce して、このような reduce のための初期値とし
てアキュムレータ (acc)を使⽤します。mapcat は map と cat の合成です。transducer が構成さ
れる順序は逆向きに⾒えるかもしれませんが、この点については後に明らかにします。

(defn my-mapcat
[f]
(comp (my-mapt f) my-cat))

(defn dupe
[x]
[x x])

(def duper (my-mapcat dupe))

(reduce (duper conj) [] [0 1 2])
;; => [0 0 1 1 2 2]

ClojureScript core における Transducer

map filter mapcat のような ClojureScript の core 関数は、transducer を返す 1 バージョンの
項数をサポートします。process-cluster の定義を再考して、transducer の観点から定義してみ
ましょう。

(def process-clusters
(comp
(mapcat split-cluster)
(filter not-rotten)
(map clean-grape)))

前の process-clusters の定義からいくつかの点が変わりました。まず、シーケンスの処理に部
分的に適⽤する代わりに、transducer を返すバージョンの mapcat filter map を使⽤します。
合成される順序が逆であり、実⾏された順序で表⽰されることにお気づきでしょうか。map、

filter、mapcat は全て transducer を返すことに注⽬してください。filter は map が返す
reducing 関数を変換して、次に進む前にフィルタリングを適⽤します。mapcat は filter が返す
reducing 関数を変換して、マッピングと結合を適⽤してから処理を進めます。

transducer の強⼒な特性の⼀つは、それらが規則的な関数合成を⽤いて合成されることです。さ
らにエレガントなのは、様々な transducer の合成そのものが transducer であることです。これは、
定義した process-cluster も transducer であることを意味するので、合成可能で⽂脈に依存し
ないアルゴリズムの変換を定義したことになります。

ClojureScript の core 関数の多くは transducer を受けとりますが、新しく作成した process-
cluster でいくつか例を⾒てみましょう。

(into [] process-clusters grape-clusters)
;; => [{:rotten? false, :clean? true}
;; {:rotten? false, :clean? true}]

(sequence process-clusters grape-clusters)
;; => ({:rotten? false, :clean? true} {:rotten? false, :clean? true})

(reduce (process-clusters conj) [] grape-clusters)
;; => [{:rotten? false, :clean? true} {:rotten? false, :clean? true}]

transducer から返される reducing 関数を⽤いて reduce を使⽤することは⾮常に⼀般的である
ため、transduce と呼ばれる変換 (transformation) で reduce を⾏う関数が存在します。これで、
reduce への前の呼び出しを transduce を⽤いて書き直すことができます。

(transduce process-clusters conj [] grape-clusters)
;; => [{:rotten? false, :clean? true} {:rotten? false, :clean? true}]

初期化 (Initialisation)

最後の例では、transduce 関数 ([])に初期値を指定しましたが、省略しても同じ結果が得られ
ます。

(transduce process-clusters conj grape-clusters)
;; => [{:rotten? false, :clean? true} {:rotten? false, :clean? true}]

何が起こっているのでしょうか。アキュムレータとして使⽤する初期値を指定していないのに、
その値が何であるかを transduce はどのようにして知ることができるのでしょうか。引数なしで
conj を呼び出して何が起こるかを確認してみます。

(conj)
;; => []

conj 関数には空のベクタを返す引数が必要ないバージョンがありますが、これだけが引数なしを
許す reducing 関数という訳ではありません。他の関数も⾒てみましょう。

(+)
;; => 0

(*)
;; => 1

(str)
;; => ""

(= identity (comp))
;; => true

transducer によって返される reducing 関数は、同様に引数がない場合をサポートしなければなら
ず、これは通常、変換された reducing 関数にデリゲートされます。これまでに実装した transducer
には、引数なし版の実⽤的な実装はありませんでした。引数なし版で単に reducing 関数を呼び出し
てみます。修正した my-mapt は以下のようになります。

(defn my-mapt
[f]
(fn [rfn]

(fn
;; arity 0 that delegates to the reducing fn
([] (rfn))
([acc input]
(rfn acc (f input))))))

transducer によって返される reducing 関数の引数なし版の呼び出しは、ネストされたすべての
reducing 関数の引数なし版を呼び出し、最終的に最も外側の reducing 関数を呼び出します。定義
済みの process-clusters transducer の例を⾒てみましょう。

((process-clusters conj))
;; => []

引数がない場合の呼び出しは transducer のスタックを経由して、最終的に (conj) を呼び出し
ます。

ステートフルな Transducer

ここまでは純粋に関数的な transducer だけを⾒てきました。つまり、暗黙的な状態を持たず、状
態の予測は容易です。しかし、take のような本質的に状態をもつデータを変換する関数が多くあり
ます。take は保持する n 個の要素と 1 つのコレクションを受け取り、最⼤で n 個の要素を持つコ
レクションを返します。

(take 10 (range 100))
;; => (0 1 2 3 4 5 6 7 8 9)

少し話題を変えて、reduce 関数の早期終了について学びましょう。アキュムレータを reduced と
呼ばれる型にラップして、reduce の処理をすぐに終了するように指⽰することができます。コレク
ション内の⼊⼒を集約し、アキュムレータに 10 個の要素が⼊ると終了する reduce の例を⾒てみま
しょう。

(reduce (fn [acc input]
(if (= (count acc) 10)

(reduced acc)
(conj acc input)))

[]
(range 100))

;; => [0 1 2 3 4 5 6 7 8 9]

transducer は reducing 関数の変形であるため、早期終了のために reduced を使⽤します。状態
をもつ transducer はプロセスが終了する前に、何らかの掃除を⾏う必要があるかもしれないので、
完成のためのステップとして、引数を 1つとる場合をサポートしなければならないことに注意して
ください。通常、引数を取らない場合と同様に引数は、引数が 1 つである変形された reducing 関
数にデリゲートされます。
これを知っていると、take のような状態をもつ transducer を書くことができます。これまでに
⾒てきたような⼊⼒の数を追跡するために、可変な状態を内部的に使⽤して、必要な要素数を確認
したらすぐに、アキュムレータを reduced にラップします。

(defn my-take
[n]
(fn [rfn]

(let [remaining (volatile! n)]
(fn

([] (rfn))
([acc] (rfn acc))
([acc input]
(let [rem @remaining

nr (vswap! remaining dec)
;; we still have items to take
result (if (pos? rem)

;; we’re done,
;; acc becomes the result
(rfn acc input)
acc)]

;; wrap result in reduced if not already
(if (not (pos? nr))

(ensure-reduced result)
result)))))))

これは ClojureScript の core にある take 関数を単純化したものです。注意すべき点がいくつか
あるので、少しずつ分けて⾒ていきましょう。
まず最初に注意しなければならないのは、transducer内で変更可能な値を作成している点です。変
換を⾏う reducing 関数を受け取るまでは値を作成していないことに注意してください。transducer
を返す前に作ってしまうと、my-take 関数を 2 回以上使うことはできません。transducer は、使⽤
の度に変換を⾏う reducing 関数を渡されるので、何度でも使うことができます。また、変更可能な
変数は、毎回使うたびに作成されます。

(fn [rfn]
;; make sure to create mutable variables
;; inside the transducer
(let [remaining (volatile! n)]

(fn
;; ...

)))

(def take-five (my-take 5))

(transduce take-five conj (range 100))
;; => [0 1 2 3 4]

(transduce take-five conj (range 100))
;; => [0 1 2 3 4]

ここで、my-take から返される reducing 関数について詳しく⾒ていきましょう。まず volatile
に deref を⾏い、取得されるべき残りの要素の数を取得して、それをデクリメントして次の残りの
値を取得します。もし、まだ取るべきアイテムが残っているなら、アキュムレータと⼊⼒を渡して
rfn を呼びます。そうでなければ、すでに最終的な結果をもっています。

([acc input]
(let [rem @remaining

nr (vswap! remaining dec)
result (if (pos? rem)

(rfn acc input)
acc)]

;; ...
))

my-take の本体は明らかになっているはずです。次の剰余 (nr) を使⽤して、処理されるアイテ
ムがまだ存在するかどうかをチェックします。存在しない場合は、ensure-reduced 関数を使⽤し
て、結果を reduced にラップします。ensure-reduced は、まだ reduce されていない場合は、値を
reduced でラップして、すでに削減されている場合には単に値を返します。まだ完了していない場
合は、累積した結果を返して処理します。

(if (not (pos? nr))
(ensure-reduced result)
result)

ステートフルな transducer の例を⾒てきましたが、完了の段階では何もしませんでした。累積値
を表⽰するために完了のステップを使⽤する transducer の例を⾒てみましょう。要素の数が n の
要素あれば、サイズが n 個のベクタの⼊⼒を変換する partition-all の簡素版を実装しましょう。
⽬的をよく理解するために、引数を 2 つとる場合で、数字とコレクションを与えると何が得られる
かを⾒てみましょう。

(partition-all 3 (range 10))
;; => ((0 1 2) (3 4 5) (6 7 8) (9))

partition-all の transducer を返す関数は、数値 n を取り、n 個のサイズのベクタで⼊⼒をグ
ループ化する transducer を返しまる。完了する段階では、累積結果があるかどうかを確認して、あ
る場合は結果に追加します。以下は、ClojureScript の core 関数である partition-all を単純化
したもので、array-list は変更可能な JavaScript の配列のラッパーです。

(defn my-partition-all
[n]
(fn [rfn]

(let [a (array-list)]
(fn

([] (rfn))
([result]
;; no inputs accumulated, don’t have to modify result
(let [result (if (.isEmpty a)

result
(let [v (vec (.toArray a))]

;; flush array contents for garbage collection
(.clear a)
;; pass to rfn , removing the reduced wrapper if present
(unreduced (rfn result v))))]

(rfn result)))
([acc input]
(.add a input)
;; got enough results for a chunk
(if (== n (.size a))

(let [v (vec (.toArray a))]
(.clear a)
;; the accumulated chunk
;; becomes input to rfn
(rfn acc v))

acc))))))

109



第 5章 ClojureScript 発展編 5.1 Transducer

(def triples (my-partition-all 3))

(transduce triples conj (range 10))
;; => [[0 1 2] [3 4 5] [6 7 8] [9]]

Eduction

Eduction とは、コレクションと 1 つ以上の変換を組み合わせる⽅法であり、それらに対して
reduce したり、繰り返したりすることができ、そのたびに変換を適⽤します。処理したいコレク
ションとそれに対する変換があり、他に拡張してほしい場合、私たちはソースコレクションと私た
ちの変換をカプセル化した Eduction をそれらに渡すことができます。eduction 関数を使⽤して
eduction を作成できます。

(def ed (eduction (filter odd?) (take 5) (range 100)))

(reduce + 0 ed)
;; => 25

(transduce (partition-all 2) conj ed)
;; => [[1 3] [5 7] [9]]

ClojureScript core における他の Transducer

map 、filter 、mapcat 、take 、partition-all について学びましたが、ClojureScript には
もっと多くの transducer があります。以下は、他の興味深いものの⼀部です。

- drop は take の双対であり、⼊⼒を reducing 関数に渡す前に n の値まで⼩さくします
- distinct は、⼊⼒を 1 回だけ⾏うことを許可します。
- dedupe は⼊⼒値の連続する重複を削除します

他にどんな transducer があるのかを知るために、ClojureScript core を調べてみることをお勧め
します。

Transducerを定義する
独⾃の transducer を作成する前に考慮すべき点がいくつかあるので、この章では、transducer を
適切に実装する⽅法を学びます。まず、transducer の⼀般的な構造は次の通りであることを学びま
した。

(fn [xf]
(fn

;; init
([]
...)

;; completion
([r]
...)

;; step
([acc input]
...)))

通常、transducer の間では ... の箇所のコードのみが変化します。これらは、結果として得られ
る関数のそれぞれの引数で保存されなければならない不変の式です。

- 0 番⽬の引数 (init): ネストされた変換 xf の 0 番⽬の引数を呼び出す必要があります。
- 1 番⽬の引数 (completion): 最終値を⽣成し、潜在的にフラッシュ状態を⽣成するために使⽤さ
れ、ネストされたトランスフォーム xf の 1 番⽬の引数を正確に 1 度だけ呼び出す必要があります。

- 2 番⽬の引数 (ステップ): 結果を導く reducing 関数であり、ネストされたトランスフォーム xf

の 2 番⽬の引数を 0 回以上呼び出します。

トランスデゥシブルなプロセス
トランスデゥシブルなプロセスとは、⼊⼒値を取り込む⼀連のステップによって定義される任意
のプロセスです。⼊⼒のソースは、プロセスによって異なります。私たちの例の⼤半は、コレクショ
ンまたは遅延シーケンスからの⼊⼒を扱っていますが、⾮同期の値のストリーム、または core.async
のチャンネルである可能性があります。各ステップで⽣成される出⼒もプロセスごとに異なります。
into は transducer のすべての出⼒を持つコレクションを作成して、シーケンスは遅延シーケンスを
⽣成し、⾮同期ストリームは出⼒をリスナーに通知します。
変換可能なプロセスについての理解を深めるために、無限の列 (unbounded queue) を実装しま
す。列に値を追加することは、⼊⼒を取り込む⼀連のステップと捉えることができるため、まず、無
限の列を実装するプロトコルとデータ型を定義します。

(defprotocol Queue
(put! [q item] "put an item into the queue")
(take! [q] "take an item from the queue")
(shutdown! [q] "stop accepting puts in the queue"))

(deftype UnboundedQueue [^:mutable arr ^:mutable closed]
Queue
(put! [_ item]

(assert (not closed))
(assert (not (nil? item)))
(.push arr item)
item)

(take! [_]
(aget (.splice arr 0 1) 0))

(shutdown! [_]
(set! closed true)))

私たちは Queue プロトコルを定義しましたが、お気付きかもしれませんが、UnboundedQueue の
実装は transducer をまったく認識しません。そのステップ関数として put! があるので、このイン
ターフェースの上に transducible プロセスを実装します。

(defn unbounded-queue
([]
(unbounded-queue nil))

([xform]
(let [put! (completing put!)

xput! (if xform (xform put!) put!)
q (UnboundedQueue. #js [] false)]

(reify
Queue
(put! [_ item]

(when-not (.-closed q)
(let [val (xput! q item)]

(if (reduced? val)
(do

;; call completion step
(xput! @val)
;; respect reduced
(shutdown! q)
@val)

val))))
(take! [_]

(take! q))
(shutdown! [_]

(shutdown! q))))))

unbounded-queue コンストラクタは、内部的に UnboundedQueue インスタンスを使⽤して、
take! と shutdown! の呼び出しをプロキシして、put! 関数の中のトランスデゥシブルなロジッ
クを実装します。何が起こっているのか理解するために、少しずつ⾒ていきましょう。

(let [put! (completing put!)
xput! (if xform (xform put!) put!)
q (UnboundedQueue. #js [] false)]

;; ...
)

まず、Queue プロトコルの put! 関数に、引数をとらない場合と引数を 1 つとる場合を追加する
ために completing を使います。これにより、この reducing 関数を xform に与えて別のものを引
き出す場合に、transducer とうまく動作するようになります。その後、もし transducer(xform)が
与えられると、transducer を put! に適⽤しながら reducing 関数を引き出します。transducer が
渡されなければ、put! を使います。q は UnboundedQueue の内部インスタンスです。

(reify
Queue
(put! [_ item]

(when-not (.-closed q)
(let [val (xput! q item)]

(if (reduced? val)
(do

;; call completion step
(xput! @val)
;; respect reduced
(shutdown! q)
@val)

val))))
;; ...

)

expose された put! 操作は queue がシャットダウンされていない場合にのみ実⾏されます。
UnboundedQueue の put! の実装では、新たな値を⼊れることができるか、不変性を壊さないかを
検証するために assert を⽤います。もし queue が閉じられておらず、値を⼊れることができる場合
は、変換される可能性がある xput! を⽤います。

put の操作が reduce された値を返した場合、transducible なプロセスを終了する必要があること
を⽰しています。この場合、queue をシャットダウンして、追加の値を受け⼊れないようにします。
もし reduce された値をえられなければ、puts を引き受け続けることができます。

queue が transducer がない状態でどう動作するかを⾒てみましょう。

(def q (unbounded-queue))
;; => #<[object Object]>

(put! q 1)
;; => 1
(put! q 2)
;; => 2

(take! q)
;; => 1
(take! q)
;; => 2
(take! q)
;; => nil

期待していた通りですね。今度はステートレスな transducer で試してみましょう。

(def incq (unbounded-queue (map inc)))
;; => #<[object Object]>

(put! incq 1)
;; => 2
(put! incq 2)
;; => 3

(take! incq)
;; => 2
(take! incq)
;; => 3
(take! incq)
;; => nil

transducible なプロセスを実装したかどうかを確認するために、ステートフルな transducer を使
いましょう。4と等しくない間は値を受け⼊れ、⼊⼒を 2つの要素のチャンクに分割する transducer
を使⽤します。

(def xq (unbounded-queue (comp
(take-while #(not= % 4))
(partition-all 2))))

(put! xq 1)
(put! xq 2)
;; => [1 2]
(put! xq 3)
(put! xq 4) ;; shouldn’t accept more values from here on
(put! xq 5)
;; => nil

(take! xq)
;; => [1 2]
;; seems like partition-all flushed correctly!
(take! xq)
;; => [3]
(take! xq)
;; => nil

queue の例は、core.async チャンネルが内部ステップで transducer を使う⽅法にに⼤きく影響
を受けています。チャンネルと transducer を使ったチャンネルの使い⽅については、後のセクショ
ンで説明します。

Transducible なプロセスは、早期終了のシグナルの伝達⼿段として reduced を反映しなければ
いけません。たとえば、コレクションのビルドは、reduced に遭遇したときや、transducer をもつ
core.async チャンネルが閉じられた場合に停⽌します。reduce された値は必ず deref で開封さ
れて、完了のステップに渡されます。完了ステップは 1 度しか呼び出されてはいけません。

transducible なプロセスは、それ⾃⾝のステップ関数を⽤いて transducer を呼び出す際に⽣成す
る reducing 関数を expose すべきではありません。なぜなら、それは状態をもち、他の場所から使
⽤することは安全でないからです。

5.2 ⼀時性
ClojureScriptの不変で永続的なデータ構造は、それなりのパフォーマンスを発揮しますが、最終
的な結果を共有するだけのために、複数のステップを使って⼤きなデータ構造を変換している状況
もあります。たとえば、core にある into 関数は、コレクションを取得して、シーケンスの内容を
続々と取り込みます。

(into [] (range 100))
;; => [0 1 2 ... 98 99]

上の例では、100 個の要素からなるベクタを⼀度に⼀つずつ conj しながら⽣成しています。最
終的な結果でない全ての中間ベクタは into 関数以外では⾒ることができず、永続化のために必要な
配列のコピーは不要なオーバーヘッドです。
このような状況のために、ClojureScript は transient と呼ばれる永続的なデータ構造を提供して
います。マップ、ベクタ、セットには、transient に相当するものがあります。Transient は、常に
transient 関数を使⽤して、永続的なデータ構造から⽣成されます。これにより、⼀定の時間内に、
transient のバージョンが作成されます。

(def tv (transient [1 2 3]))
;; => #<[object Object]>

Transient は、対応する永続的な読取り API をサポートします。

(def tv (transient [1 2 3]))

(nth tv 0)
;; => 1

(get tv 2)
;; => 3

(def tm (transient {:language "ClojureScript"}))

(:language tm)
;; => "ClojureScript"

(def ts (transient #{:a :b :c}))

(contains? ts :a)
;; => true

(:a ts)
;; => :a

transient には更新のための永続的で不変なセマンティクスがないので、conj や assoc 関数
を使って変換できません。その代わり、transient 上で機能する変換関数は感嘆符 ! をつけます。
conj! を使った例を⾒てみましょう。

(def tv (transient [1 2 3]))

(conj! tv 4)
;; => #<[object Object]>

(nth tv 3)
;; => 4

このように、ベクタの transient バージョンは不変でも永続でもありません。その代わり、ベクタ
はその場で変異します。conj! を使って繰り返し tv を変換することもできますが、永続的なデー
タ構造で使⽤されるイディオムを放棄してはいけません。transient のデータを変換する場合は、次
の例のように、その戻されたバージョンを使⽤して変更を加えます。

(-> [1 2 3]
transient
(conj! 4)
(conj! 5))

;; => #<[object Object]>

transient! を呼ぶことで、transient を元の永続的で不変なデータ構造に変換することもできま
す。この操作は、transient を永続的なデータ構造から派⽣させる操作と同様に、⼀定の時間で実⾏
されます。

(-> [1 2 3]
transient
(conj! 4)
(conj! 5)
persistent!)

;; => [1 2 3 4 5]

transient を永続的な構造に変換する特徴は、⼀時的なものが永続的なデータ構造に変換された後
に無効になり、それ以上の変換ができなくなることです。これは、派⽣した永続的なデータ構造が
transient の内部的な node を使⽤しており、それらを変更することにより不変性と永続的な保証が
壊れるためです。

(def tm (transient {}))
;; => #<[object Object]>

(assoc! tm :foo :bar)
;; => #<[object Object]>

(persistent! tm)
;; => {:foo :bar}

(assoc! tm :baz :frob)
;; Error: assoc! after persistent!

最初の into の例に戻りましょう。ここではパフォーマンスのために transient を使⽤してシン
プルに実装します。内部的には変異を使⽤しますが、永続的なデータ構造を返す純粋な関数型のイ
ンタフェースを公開します。

(defn my-into
[to from]
(persistent! (reduce conj! (transient to) from)))

(my-into [] (range 100))
;; => [0 1 2 ... 98 99]

5.3 メタデータ
ClojureScript のシンボル、変数、永続的なコレクションは、メタデータの添付 (attach) をサ
ポートしています。メタデータは、添付される実体 (entity) に関する情報を含むマップです。
ClojureScript のコンパイラは、型のヒントのような⽬的のためにメタデータを使⽤します。また、
メタデータのシステムは、ツールやライブラリ及びアプリケーションの開発者も利⽤できます。

ClojureScript を⽇常的に書いているなかでは、メタデータが必要になることはあまりないかもし
れませんが、メタデータは知っておくと便利な機能なので、いつか役に⽴つかもしれません。実⾏
時のコード分析やドキュメントの⽣成などが⾮常に簡単になります。次のセクションでは、その理
由を説明します。

Vars

var を定義して、デフォルトでどのメタデータが付加されているかを⾒てみましょう。このコード
は REPL で実⾏されるため、ソースファイルで定義されている var のメタデータは変化するかもし
れません。meta 関数を使⽤して、指定された値のメタデータを取得します。

(def answer-to-everything 42)
;; => 42

#’answer-to-everything
;; => #’cljs.user/answer-to-everyhing

(meta #’answer-to-everything)
;; => {:ns cljs.user,
;; :name answer-to-everything,
;; :file "NO_SOURCE_FILE",
;; :source "answer-to-everything",
;; :column 6,
;; :end-column 26,
;; :line 1,
;; :end-line 1,
;; :arglists (),
;; :doc nil,
;; :test nil}

上記のコードにおいて、#’answer-to-everything は、answer-to-everything シンボルの値
を保持する var への参照を渡します。これには、定義された名前空間 (:ns)、その名前、ファイル
(ここでは REPLで定義されているため、ソースファイルはありません)、ソース、定義されたファ
イルでの位置、引数の⼀覧 (関数の場合のみ意味があります)、ドキュメントの⽂字列、およびテス
ト関数に関する情報が含まれていることがわかります。
関数における var のメタデータを⾒てみましょう。

(defn add
"A function that adds two numbers."
[x y]
(+ x y))

(meta #’add)
;; => {:ns cljs.user,
;; :name add,
;; :file "NO_SOURCE_FILE",
;; :source "add",
;; :column 7,
;; :end-column 10,
;; :line 1,
;; :end-line 1,
;; :arglists (quote ([x y])),
;; :doc "A function that adds two numbers.",
;; :test nil}

引数のリストは :arglists フィールド、マニュアルは :doc フィールドに格納されます。では
テスト関数を定義して、何のために :test フィールドを使うかを⾒ていきましょう。

(require ’[cljs.test :as t])

(t/deftest i-pass
(t/is true))

(meta #’i-pass)
;; => {:ns cljs.user,
;; :name i-pass,
;; :file "NO_SOURCE_FILE",
;; :source "i-pass",
;; :column 12,
;; :end-column 18,
;; :line 1,
;; :end-line 1,
;; :arglists (),
;; :doc "A function that adds two numbers.",
;; :test #<function (){ ... }>}

i-pass var のメタデータ内の :test フィールドがテスト関数です。テスト関数は、指定した名
前空間でテストを検出して実⾏するために cljs.test ライブラリによって使⽤されます。

値
var はメタデータを持つことができ、コンパイラと cljs.test ライブラリが利⽤するために、ど

のようなメタデータを付与されるかを学びました。永続コレクションがメタデータをもつことは可
能ですが、デフォルトではメタデータはありません。with-meta 関数を使⽤すると、指定したメ
タデータが付加された同じ値と型を持つオブジェクトを派⽣させることができます。例を⾒てみま
しょう。

(def map-without-metadata {:language "ClojureScript"})
;; => {:language "ClojureScript"}

(meta map-without-metadata)
;; => nil

(def map-with-metadata (with-meta map-without-metadata
{:answer-to-everything 42}))

;; => {:language "ClojureScript"}

(meta map-with-metadata)
;; => {:answer-to-everything 42}

(= map-with-metadata
map-without-metadata)

;; => true

(identical? map-with-metadata
map-without-metadata)

;; => false

ClojureScript の等価性は値に基づいているため、メタデータが 2 つのデータ構造の等価性に影
響しないことは驚くべきことではありません。もう 1 つ興味深い点は、with-meta を使⽤すると、
指定されたものと同じ型と値をもつ別のオブジェクトが作成され、指定されたメタデータがそれに
付与されることです。
永続データ構造から新しい値を派⽣させるときにメタデータで何が起こるかを確認してみま
しょう。

(def derived-map (assoc map-with-metadata :language "Clojure"))
;; => {:language "Clojure"}

(meta derived-map)
;; => {:answer-to-everything 42}

この例でわかるように、メタデータは永続的なデータ構造の派⽣バージョンで保持されます。微
妙なところもありますが、新しいデータ構造を派⽣する関数が同じ型のコレクションを返す限り、メ
タデータは保存されます。これは、変換によって型が変更される場合には当てはまりません。この
点を理解するために、ベクタから seq や subvector を派⽣させると何が起こるかを⾒てみましょう。

(def v (with-meta [0 1 2 3] {:foo :bar}))
;; => [0 1 2 3]

(def sv (subvec v 0 2))
;; => [0 1]

(meta sv)
;; => nil

(meta (seq v))
;; => nil

メタデータのための構⽂
ClojureScript の reader はメタデータの注釈 (annotation)を構⽂としてサポートしますが、様々
な⽅法で記述できます。var 定義やコレクションの先頭にキャレット (^) とその後にマップをつけ
て、指定したメタデータのマップで注釈をつけることができます。

(def ^{:doc "The answer to Life, Universe and Everything."} answer-to-everything 42)
;; => 42

(meta #’answer-to-everything)
;; => {:ns cljs.user,
;; :name answer-to-everything,
;; :file "NO_SOURCE_FILE",
;; :source "answer-to-everything",
;; :column 6,
;; :end-column 26,
;; :line 1,
;; :end-line 1,
;; :arglists (),
;; :doc "The answer to Life, Universe and Everything.",
;; :test nil}

(def map-with-metadata ^{:answer-to-everything 42} {:language "ClojureScript"})
;; => {:language "ClojureScript"}

(meta map-with-metadata)
;; => {:answer-to-everything 42}

answer-to-everything の var の定義で指定されたメタデータが var のメタデータとどのよう
にマージされるかに注意してください。
メタデータの⼀般的な使⽤法は、特定のキーの値を true に設定することです。たとえば、変数
が動的まか定数かを var のメタデータに追加できます。このような場合には、キャレットに続けて
キーワードを使⽤する省略表記があります。次に例を⽰します。

(def ^:dynamic *foo* 42)
;; => 42

(:dynamic (meta #’*foo*))
;; => true

(def ^:foo ^:bar answer 42)
;; => 42

(select-keys (meta #’answer) [:foo :bar])
;; => {:foo true, :bar true}

メタデータの付与には、別の簡略表記があります。キャレットに続けてシンボルを使⽤する
と、:tag キーの下のメタデータのマップに追加されます。^boolean などのタグを使⽤すると、
ClojureScript コンパイラに、式の型や関数の戻り値の型に関するヒントを与えます。

(defn ^boolean will-it-blend? [_] true)
;; => #<function ... >

(:tag (meta #’will-it-blend?))
;; => boolean

(not ^boolean (js/isNaN js/NaN))
;; => false

メタデータを扱う関数
これまで meta と with-meta について学んできましたが、ClojureScript にはメタデータを変換
するための関数がいくつか⽤意されています。var-meta は、元のオブジェクトと同じ型と値を持
つ新しいオブジェクトを派⽣させるという点で with-meta と似ていますが、直接的に付与するメタ
データをとりません。代わりに、var-meta は、変換する指定のオブジェクトのメタデータに適⽤す
る関数をとり、新たなメタデータを派⽣させます。例を⾒てみましょう。

(def map-with-metadata ^{:foo 40} {:language "ClojureScript"})
;; => {:language "ClojureScript"}

(meta map-with-metadata)
;; => {:foo 40}

(def derived-map (vary-meta map-with-metadata update :foo + 2))
;; => {:language "ClojureScript"}

(meta derived-map)
;; => {:foo 42}

既存の var や値のメタデータを変更する場合、変更のためには alter-meta! 関数を、別のマッ
プに置き換えるには reset-meta! を使うことができます。

(def map-with-metadata ^{:foo 40} {:language "ClojureScript"})
;; => {:language "ClojureScript"}

(meta map-with-metadata)
;; => {:foo 40}

(alter-meta! map-with-metadata update :foo + 2)
;; => {:foo 42}

(meta map-with-metadata)
;; => {:foo 42}

(reset-meta! map-with-metadata {:foo 40})
;; => {:foo 40}

(meta map-with-metadata)
;; => {:foo 40}

5.4 中⼼的なプロトコル
ClojureScript の core に定義されている関数群がプロトコルを中⼼に実装されていることも素晴
らしい特徴の 1 つだといえます。これは、そのようなプロトコルで拡張したどんな型、また私たち
や第三者によって定義された型でも動作するように開かれています。

関数
これまでの章で学んだように、関数だけが呼び出されるものとは限りません。ベクタはインデッ
クス、マップはキー、セットは値を関数のように使うことができます。

IFn プロトコルを実装する関数として呼び出し可能な型に拡張することができます。関数として
の呼び出しをサポートしていないコレクションが queue です。PersistentQueue 型に IFn を実装
して、queue をインデックスの関数として呼び出すことができるようにします。

(extend-type PersistentQueue
IFn
(-invoke

([this idx]
(nth this idx))))

(def q #queue[:a :b :c])
;; => #queue [:a :b :c]

(q 0)
;; => :a

(q 1)
;; => :b

(q 2)
;; => :c

Printing

いくつかの中⼼的なプロトコルについて学ぶために、値のペアを保持する Pair 型を定義します。

(deftype Pair [fst snd])

型の印字⽅法をカスタマイズしたい場合は、IPrintWithWriter プロトコルを実装できます。出⼒
値、writer オブジェクト、オプションを受け取る -pr-writer と呼ばれる関数を定義します。この
関数は、求められる Pair型 の⽂字列表現を書くために、writer オブジェクトの -write 関数を使
⽤します。

(extend-type Pair
IPrintWithWriter
(-pr-writer [p writer _]

(-write writer (str "#<Pair " (.-fst p) "," (.-snd p) ">"))))

シーケンス
前のセクションでは、ClojureScript の主要な抽象概念の 1 つであるシーケンスについて学びま

した。シーケンスを操作するための first 関数と rest 関数を覚えているでしょうか。これらは
ISeq プロトコルで定義されているので、このような関数に対応する型を拡張することができます。

(extend-type Pair
ISeq
(-first [p]

(.-fst p))

(-rest [p]
(list (.-snd p))))

(def p (Pair. 1 2))
;; => #<Pair 1,2>

(first p)
;; => 1

(rest p)
;; => (2)

シーケンスを操作する便利な関数として next があります。next は、与えられた引数がシーケン
スである限り動作しますが、INext プロトコルを使って明⽰的に実装することができます。

(def p (Pair. 1 2))

(next p)
;; => (2)

(extend-type Pair
INext
(-next [p]

(println "Our next")
(list (.-snd p))))

(next p)
;; Our next
;; => (2)

最後に、ISeqable プロトコルを実装する独⾃の型をシーカブルにすることができます。これは、
シーケンスを元に戻すために seq に渡すことができることを意味します。

(def p (Pair. 1 2))

(extend-type Pair
ISeqable
(-seq [p]

(list (.-fst p) (.-snd p))))

(seq p)
;; => (1 2)

これで Pair 型は、ClojureScript のシーケンス操作関数の⼤半で動作します。

(def p (Pair. 1 2))
;; => #<Pair 1,2>

(map inc p)
;; => (2 3)

(filter odd? p)
;; => (1)

(reduce + p)
;; => 3

コレクション
コレクションの関数もプロトコルで定義されています。このセクションの例では、native な

JavaScript の⽂字列をコレクションのように動作させます。コレクションを扱う上で最も重要な関
数は、ICollection プロトコルで定義されている conj です。⽂字列は、⽂字列への conj に意味
を持つ唯⼀の型であるため、⽂字列に対する conj の操作は単なる連結になります。

(extend-type string
ICollection
(-conj [this o]

(str this o)))

(conj "foo" "bar")
;; => "foobar"

(conj "foo" "bar" "baz")
;; => "foobarbaz"

コレクションを操作する empty は便利な関数です。これは IEmptyable Collection プロトコル
の⼀部です。⽂字列の型に実装してみましょう。

(extend-type string
IEmptyableCollection
(-empty [_]

""))

(empty "foo")
;; => ""

ここでは、native な JavaScript の⽂字列を拡張するための特別なシンボルである string を使
いました。詳細については、JavaScript の型を拡張するセクションを参照してください。

■コレクションの特性 すべてのコレクションに備わっているわけではない特性がいくつかありま
す。たとえば、⼀定の時間で数えられることや、可逆的であることなどです。これらの特性がすべ
てのコレクションに対して意味をもつわけではないので、これらの特性は異なるプロトコルに分割
されます。これらのプロトコルを説明するために、前に定義した Pair 型を使⽤します。

count 関数を使⽤して⼀定の時間でカウントできるコレクションの場合、ICounted プロトコル
を実装できます。Pair型には容易に実装できるはずです。

(extend-type Pair
ICounted
(-count [_]

2))

(def p (Pair. 1 2))

(count p)
;; => 2

ベクタやリストなどのコレクション型は nth 関数でインデックス番号を付けることができます。
このような型にインデックスが付けられれば、IIndexed プロトコルを実装することができます。

(extend-type Pair
IIndexed
(-nth

([p idx]
(case idx

0 (.-fst p)
1 (.-snd p)
(throw (js/Error. "Index out of bounds"))))

([p idx default]
(case idx

0 (.-fst p)
1 (.-snd p)
default))))

(nth p 0)
;; => 1

(nth p 1)
;; => 2

(nth p 2)
;; Error: Index out of bounds

(nth p 2 :default)
;; => :default

連想性
キーを値にマップする連想的なデータ構造が数多くあります。私たちはすでにいくつかの関数に
遭遇しており、get、assoc、dissoc 等の関数を知っています。これらの関数を構築しているプロト
コルについて説明します。
まず最初に、連想的なデータ構造においてキーを検索する⽅法が必要です。ILookup プロトコル

は、そのための関数を定義します。ここでは、Pair 型にキーを検索する機能を追加します。これは、
Pair 型がインデックス 0 と 1 を値にマップする連想的なデータ構造であるためです。

(extend-type Pair
ILookup
(-lookup

([p k]
(-lookup p k nil))

([p k default]
(case k

0 (.-fst p)
1 (.-snd p)
default))))

(get p 0)
;; => 1

(get p 1)
;; => 2

(get p :foo)
;; => nil

(get p 2 :default)
;; => :default

データ構造で assoc を使⽤するには、IAssociative プロトコルを実装する必要があります。
Pair 型では、値を関連づけるためのキーとして 0 と 1 だけが許可されます。IAssociative には、
キーの有無を問い合わせる機能もあります。

(extend-type Pair
IAssociative
(-contains-key? [_ k]

(contains? #{0 1} k))

(-assoc [p k v]
(case k

0 (Pair. v (.-snd p))
1 (Pair. (.-fst p) v)
(throw (js/Error. "Can only assoc to 0 and 1 keys")))))

(def p (Pair. 1 2))
;; => #<Pair 1,2>

(assoc p 0 2)
;; => #<Pair 2,2>

(assoc p 1 1)
;; => #<Pair 1,1>

(assoc p 0 0 1 1)
;; => #<Pair 0,1>

(assoc p 2 3)
;; Error: Can only assoc to 0 and 1 keys

assoc を補完する関数として dissoc があり、dissoc は IMapプロトコルの⼀部です。Pair 型
ではあまり意味がありませんが、それでも実装してみましょう。0 または 1 の関連付けを解除する
と、そのような位置に nil が配置され、無効なキーは無視されます。

(extend-type Pair
IMap
(-dissoc [p k]

(case k
0 (Pair. nil (.-snd p))
1 (Pair. (.-fst p) nil)
p)))

(def p (Pair. 1 2))
;; => #<Pair 1,2>

(dissoc p 0)
;; => (nil 2)

(dissoc p 1)
;; => (1 nil)

(dissoc p 2)
;; => (1 2)

(dissoc p 0 1)
;; => (nil nil)

連想的なデータ構造は、エントリー (entries)と呼ばれるキーと値のペアで構成されます。key 関
数と val 関数を使⽤すると、このようなエントリーのキーと値を参照できます。これらの関数は
IMapEntry プロトコルに基づいて構築されています。key 関数と val 関数の例と、マップのエン
トリーを使⽤してマップを構築する⽅法をいくつか⾒てみましょう。

(key [:foo :bar])
;; => :foo

(val [:foo :bar])
;; => :bar

(into {} [[:foo :bar] [:baz :xyz]])
;; => {:foo :bar, :baz :xyz}

Pair 型もマップのエントリーにできます。最初の要素をキー、2 番⽬の要素を値として扱います。

(extend-type Pair
IMapEntry
(-key [p]

(.-fst p))

(-val [p]
(.-snd p)))

(def p (Pair. 1 2))
;; => #<Pair 1,2>

(key p)
;; => 1

(val p)
;; => 2

(into {} [p])
;; => {1 2}

⽐較
2 つ以上の値が等しいかどうかを = を⽤いて判定するには、IEquivプロトコルを実装する必要が

あります。Pair 型に、IEquiv プロトコルを実装してみましょう:

(def p (Pair. 1 2))
(def p’ (Pair. 1 2))
(def p’’ (Pair. 1 2))

(= p p’)
;; => false

(= p p’ p’’)
;; => false

(extend-type Pair
IEquiv
(-equiv [p other]

(and (instance? Pair other)
(= (.-fst p) (.-fst other))
(= (.-snd p) (.-snd other)))))

(= p p’)
;; => true

(= p p’ p’’)
;; => true

型を⽐較できるようにすることもできます。compare 関数は 2 つの値を取り、最初の値が 2 番⽬
の値より⼩さい場合は負の数を返し、両⽅が等しい場合は 0 を返し、最初の値が 2 番⽬の値より⼤
きい場合は 1 を返します。型を⽐較できるようにするには、IComparable プロトコルを実装する必
要があります。

Pair 型の場合、⽐較とは最初の 2 つの値が等しいかどうかをチェックすることを意味します。等
しい場合は、結果は 2 番⽬の値になります。そうでない場合は、最初の⽐較の結果を返します。

(extend-type Pair
IComparable
(-compare [p other]

(let [fc (compare (.-fst p) (.-fst other))]
(if (zero? fc)

(compare (.-snd p) (.-snd other))
fc))))

(compare (Pair. 0 1) (Pair. 0 1))
;; => 0

(compare (Pair. 0 1) (Pair. 0 2))
;; => -1

(compare (Pair. 1 1) (Pair. 0 2))
;; => 1

(sort [(Pair. 1 1) (Pair. 0 2) (Pair. 0 1)])
;; => ((0 1) (0 2) (1 1))

メタデータ
meta 関数と with-meta 関数も、それぞれ IMeta と IWithMeta という 2 つのプロトコルに基

づいています。追加のフィールドをメタデータを保持するために加えて、両プロトコルを実装する
ことにより、メタデータを運ぶ独⾃の型を作成できます。
メタデータをもつことができる Pair を実装して⾒ましょう。

(deftype Pair [fst snd meta]
IMeta
(-meta [p] meta)

IWithMeta
(-with-meta [p new-meta]

(Pair. fst snd new-meta)))

(def p (Pair. 1 2 {:foo :bar}))
;; => #<Pair 1,2>

(meta p)
;; => {:foo :bar}

(def p’ (with-meta p {:bar :baz}))
;; => #<Pair 1,2>

(meta p’)
;; => {:bar :baz}

JavaScriptとの相互運⽤
ClojureScript は JavaScript の VM でホストされるため、ClojureScript のデータ構造を

JavaScript のデータ構造に変換したり、また、その逆を⾏う必要があります。native の JavaScript
の型をプロトコルで表現される抽象化に加えることもできます。

■JavaScript の型を拡張する JavaScript のオブジェクトを拡張する場合は、js/String や
js/Date などの JavaScript のグローバル環境を使⽤せずに、特別なシンボルが使⽤されます。こ
れは、グローバル環境にある JavaScript オブジェクトの変更を避けるために⾏われます。

JavaScriptの型を拡張するシンボルには、object、array、number、string、function、boolean、
nil があります。プロトコルを native オブジェクトにディスパッチするには、Google Closure の
goog.typeOf 関数を使います。全ての型のためのプロトコルを実装するために使われる特別なデ
フォルトのシンボルがあります。

JavaScript の型の拡張を説明するために、mutable? 関数を⽤いて MaybeMutable プロトコルを
実装します。JavaScript の可変性はデフォルトなので、mutable? から true を返すデフォルトの
JavaScript の型を拡張します。:

(defprotocol MaybeMutable
(mutable? [this]

"Returns true if the value is mutable."))

(extend-type default
MaybeMutable
(mutable? [_] true))

;; object
(mutable? #js {})
;; => true

;; array
(mutable? #js [])
;; => true

;; string
(mutable? "")
;; => true

;; function
(mutable? (fn [x] x))
;; => true

幸い、全ての JavaScript オブジェクトの値が可変という訳ではないため、⽂字列と関数に対して
false を返すように MaybeMutable の実装を改良することができます。

(extend-protocol MaybeMutable
string
(mutable? [_] false)

function
(mutable? [_] false))

;; object
(mutable? #js {})
;; => true

;; array
(mutable? #js [])
;; => true

;; string
(mutable? "")
;; => false

;; function
(mutable? (fn [x] x))
;; => false

JavaScript では date に特別な記号はないので、js/Date を直接拡張する必要があります。
JavaScript のグローバルな名前空間にある残りの型にも同じことが当てはまります。

■データの変換 ClojureScript の型から JavaScript の型、あるいはその逆を⾏うために、
IEncodeJS と IEncodeClojure プロトコルにそれぞれ基づいた clj->js と js->clj 関数を
使います。
この例では、ES6 で導⼊された Set 型を使⽤します。これは、全ての JavaScript の実⾏環境で
使⽤できるわけではありません。

■ClojureScript から JavaScript まず、ClojureScript の set 型を拡張して JavaScript に変換
できるようにします。デフォルトでは、set は JavaScript の配列に変換されます。

(clj->js #{1 2 3})
;; => #js [1 3 2]

これを修正しましょう。clj->js は値を再帰的に変換するので、set の要素を全て JavaScript に
変換して、変換された値を使って set を作成します。

(extend-type PersistentHashSet
IEncodeJS
(-clj->js [s]

(js/Set. (into-array (map clj->js s)))))

(def s (clj->js #{1 2 3}))
(es6-iterator-seq (.values s))
;; => (1 3 2)

(instance? js/Set s)
;; => true

(.has s 1)
;; => true
(.has s 2)
;; => true
(.has s 3)
;; => true
(.has s 4)
;; => false

es6-iterator-seq は、ES6 のイテレブルから seq を取得するための ClojureScript core にあ
る実験的な関数です。

■JavaScript から ClojureScript 今度は JavaScript のセットを拡張して ClojureScript に変換
します。clj->js と同様に、js->clj はデータ構造の値を再帰的に変換します。

(extend-type js/Set
IEncodeClojure
(-js->clj [s options]

(into #{} (map js->clj (es6-iterator-seq (.values s))))))

(= #{1 2 3}
(js->clj (clj->js #{1 2 3})))

;; => true

(= #{[1 2 3] [4 5] [6]}
(js->clj (clj->js #{[1 2 3] [4 5] [6]})))

;; => true

ClojureScript と JavaScript の値は、⼀対⼀のマッピングはないことに注意してください。たと
えば、ClojureScript のキーワードは、clj->js に渡されると JavaScript の⽂字列に変換されます。

リダクション
reduce 関数は IReduce プロトコルに基づいており、このプロトコルを使⽤すると、独⾃の型や
サードパーティの型を reducible にすることができます。reduce と⼀緒に使う以外にも、⾃動的に
transduce と⼀緒に動作するので、transducer を使って reduce を⾏うことができます。

JavaScript の配列は ClojureScript でも reduce できます。

(reduce + #js [1 2 3])
;; => 6

(transduce (map inc) conj [] [1 2 3])
;; => [2 3 4]

しかし、新しい ES6 に Set 型はそうではないので、IReduce プロトコルを実装しましょう。Set
の values メソッドを使ってイテレーターを取得して、これを es6-iterator-seq 関数を使って
seq に変換します。その後、元の reduce 関数にデリゲートして、取得したシーケンスを reduce し
ます。

(extend-type js/Set
IReduce
(-reduce

([s f]
(let [it (.values s)]

(reduce f (es6-iterator-seq it))))
([s f init]
(let [it (.values s)]

(reduce f init (es6-iterator-seq it))))))

(reduce + (js/Set. #js [1 2 3]))
;; => 6

(transduce (map inc) conj [] (js/Set. #js [1 2 3]))
;; => [2 3 4]

連想的なデータ構造は IKVReduce プロトコルに基づく reduce‐ kv 関数を⽤いて reduce でき
ます。reduce と reduce-kv の主な違いは、reduce-kv は 3 つの引数をとる関数であり、reducer、
アキュムレータの受け取り、各アイテムのキーと値をとることです。
例を⾒てみましょう。マップをペアからなるベクタに変換します。ベクタはインデックスを値に
関連付けるので、reduce-kv を使って値を reduce することができます。

(reduce-kv (fn [acc k v]
(conj acc [k v]))

[]
{:foo :bar
:baz :xyz})

;; => [[:foo :bar] [:baz :xyz]]

新しい ES6 のマップ型を拡張して、reduce-kv をサポートするようにします。これを⾏うには、
キーと値のペアのシーケンスを取得し、アキュムレータ、キー、値を位置を⽰す引数として使⽤し
て reducing 関数を呼び出します。

(extend-type js/Map
IKVReduce
(-kv-reduce [m f init]

(let [it (.entries m)]
(reduce (fn [acc [k v]]

(f acc k v))
init
(es6-iterator-seq it)))))

(def m (js/Map.))
(.set m "foo" "bar")
(.set m "baz" "xyz")

(reduce-kv (fn [acc k v]
(conj acc [k v]))

[]
m)

;; => [["foo" "bar"] ["baz" "xyz"]]

どちらの例でも、最終的には reduce 関数にデリゲートすることになりました。reduce 関数は、
値の reduce された値を認識して、値が⾒つかると終了します。これらのプロトコルを reduce に関
して実装しない場合は、早期終了のために reduce された値をチェックする必要があることを覚えて
おいてください。

遅延計算
⾮同期処理の概念をもついくつかの型があり、それらが表す値はまだ認識 (realize)されていない

かもしれません。realized? を使って値が認識されたかどうかを確認することができます。ここで
は、計算を⾏い、結果が必要になったときにそれを実⾏する Delay 型を使って⾒ていきましょう。
遅延を逆に参照すると、計算が実⾏されて、遅延が認識されます。

(defn computation []
(println "running!")
42)

(def d (delay (computation)))

(realized? d)  ;; => false

(deref d)
;; running!
;; => 42

(realized? d)  ;; => true

@d  ;; => 42

realized? と deref は IPending と IDeref のプロトコルの上にあります。

状態
Atom や Volatile のような ClojureScript の状態に関する構造体 (state constructs)は、異なる

特性とセマンティクスを持ち、add-watch や reset! または swap! のような操作は、プロトコルに
よってサポートされています。

■Atom このようなプロトコルを理解するために、私たちは独⾃の簡素版の Atom を実装します。
バリデータやメタデータはサポートしませんが、次のようなことは可能です。

- deref で現在の値を取得するためのアトムの参照解除する
- reset でアトムに含まれる値をリセットする
- swap! で atom を状態を変える関数を⽤いて変更する

deref は IDerefプロトコル、reset! は IReset プロトコル、swap! は ISwap プロトコルに
基づいています。まず、atom の実装のために、データ型とコンストラクターを定義します。

(deftype MyAtom [^:mutable state ^:mutable watches]
IPrintWithWriter
(-pr-writer [p writer _]

(-write writer (str "#<MyAtom " (pr-str state) ">"))))

(defn my-atom
([]
(my-atom nil))

([init]
(MyAtom. init {})))

(my-atom)    ;; => #<MyAtom nil>
(my-atom 42)  ;; => #<MyAtom 42>

ここでは、atom の現在の状態と watcher のマップ (watches)の両⽅を {:mutable true} のメタ
データでマークしていることに注⽬してください。これらを修正して、アノテーションを⽤いて明
⽰的にします。
この MyAtom 型はまだあまり有⽤ではありませんが、まず IDeref プロトコルを実装して、現在

の値を逆参照できるようにします。

(extend-type MyAtom
IDeref
(-deref [a]

(.-state a)))

(def a (my-atom 42))

@a  ;; => 42

参照を解除できるようになったので、IWatchable プロトコルを実装して、独⾃の atom に
watches を追加したり削除したりできるようにします。watches を MyAtom のマップに格納して、
キーをコールバックに関連づけます。

(extend-type MyAtom
IWatchable
(-add-watch [a key f]

(let [ws (.-watches a)]
(set! (.-watches a) (assoc ws key f))))

(-remove-watch [a key]
(let [ws (.-watches a)]

(set! (.-watches a) (dissoc ws key))))

(-notify-watches [a oldval newval]
(doseq [[key f] (.-watches a)]

(f key a oldval newval))))

atom に watches を追加できるようになりましたが、まだ変更できないのであまり役に⽴ちませ
ん。変更を取り込むためには、IReset プロトコルを実装して、atom の値をリセットするたびに
watches に通知する必要があります。

(extend-type MyAtom
IReset
(-reset! [a newval]

(let [oldval (.-state a)]
(set! (.-state a) newval)
(-notify-watches a oldval newval)
newval)))

ここで、正しく動作するか確認しましょう。watches を追加してから atom の値を変更して、
watches が呼び出されることを確認してから削除します。

(def a (my-atom 41))
;; => #<MyAtom 41>

(add-watch a :log (fn [key a oldval newval]
(println {:key key

:old oldval
:new newval})))

;; => #<MyAtom 41>

(reset! a 42)
;; {:key :log, :old 41, :new 42}
;; => 42

(remove-watch a :log)
;; => #<MyAtom 42>

(reset! a 43)
;; => 43

私たちが実装を進めている atom にはまだスワップの機能がないので、ここで ISwap プロトコル
を実装します。-swap! には 4 つの arity があります。swap! に渡される関数には、1 つ以上の引数
をとることができます。

(extend-type MyAtom
ISwap
(-swap!

([a f]
(let [oldval (.-state a)

newval (f oldval)]
(reset! a newval)))

([a f x]
(let [oldval (.-state a)

newval (f oldval x)]
(reset! a newval)))

([a f x y]
(let [oldval (.-state a)

newval (f oldval x y)]
(reset! a newval)))

([a f x y more]
(let [oldval (.-state a)

newval (apply f oldval x y more)]
(reset! a newval)))))

これで、独⾃版の atom 抽象の実装ができました。これを REPL でテストし、期待どおりに動作
するかどうかを確認しましょう。

(def a (my-atom 0))  ;; => #<MyAtom 0>

(add-watch a :log (fn [key a oldval newval]
(println {:key key

:old oldval
:new newval})))

;; => #<MyAtom 0>

(swap! a inc)
;; {:key :log, :old 0, :new 1}
;; => 1

(swap! a + 2)
;; {:key :log, :old 1, :new 3}
;; => 3

(swap! a - 2)
;; {:key :log, :old 3, :new 1}
;; => 1

(swap! a + 2 3)
;; {:key :log, :old 1, :new 6}
;; => 6

(swap! a + 4 5 6)
;; {:key :log, :old 6, :new 21}
;; => 21

(swap! a * 2)
;; {:key :log, :old 21, :new 42}
;; => 42

(remove-watch a :log) ;; => #<MyAtom 42>

成功しました! メタデータやバリデータをサポートしないバージョンの Atom を実装しましたが、
このような機能をサポートするように拡張することは、読者の皆さんの課題とします。メタデータ
とバリデーターを保存できるようにするには、MyAtom 型を変更する必要があります。

■Volatile Volatile は atom よりも単純で、変化の監視をサポートしません。⼤半のプログラミン
グ⾔語に存在する可変変数のように、全ての変更は前の値を上書きします。Volatile は IVolatile プ
ロトコルに基づいています。IVolatile プロトコルは、vreset! のためのメソッドのみを定義します。
これは vswap! がマクロとして実装されているからです。まず、独⾃の volatile 型とコンストラク
タを作成します。

(deftype MyVolatile [^:mutable state]
IPrintWithWriter
(-pr-writer [p writer _]

(-write writer (str "#<MyVolatile " (pr-str state) ">"))))

(defn my-volatile
([]

(my-volatile nil))
([v]

(MyVolatile. v)))

(my-volatile) ;; => #<MyVolatile nil>

(my-volatile 42) ;; => #<MyVolatile 42>

私たちの MyVolatile は、参照の解除と再設定をサポートする必要があります。では、IDeref

と IVolatile を実装しましょう。これにより、MyVolatile で deref、vreset!、vswap! が使え
るようになります。

(extend-type MyVolatile
IDeref
(-deref [v]

(.-state v))

IVolatile
(-vreset! [v newval]

(set! (.-state v) newval)
newval))

(def v (my-volatile 0))
;; => #<MyVolatile 42>

(vreset! v 1)
;; => 1

@v
;; => 1

(vswap! v + 2 3)
;; => 6

@v
;; => 6

変更
transient のセクションでは、ClojureScript が提供する不変で永続的なデータ構造に対応する可
変のデータ構造について学びました。これらのデータ構造は可変であり、それらに対する関数は、可
変であることを明⽰するために感嘆符 (!)で終わります。あなたの想像通り、transient に対する全
ての操作はプロトコルに基づいています。

■永続性から⼀時性 (⼀時性から永続性) IEditableCollection プロトコルに基づく transient

関数を使って、永続的なデータ構造を変換できることを学びました。transient を永続的なデータ
構造に変換するには、ITransientCollectionに基づく persistent! を使います。
不変で永続的なデータ構造とそれらの⼀時的な対応物を実装することはこの本の範囲外ですが、
興味があれば ClojureScript のデータ構造の実装を⾒ることをお勧めします。

■Transientのベクタとセット transient のデータ構造のためのプロトコルの⼤部分は学びました
が、assoc! を transient のベクタに使うための ITransientVector と、disj! を transient のセット
に使うための ITransientSet についてまだ取り上げていません。

ITransientVector プロトコルを説明するために、連想的な transient のデータ構造にするため
に、JavaScript の配列型を拡張してみます。

(extend-type array
ITransientAssociative
(-assoc! [arr key val]

(if (number? key)
(-assoc-n! arr key val)
(throw (js/Error.

"Array’s key for assoc! must be a number."))))

ITransientVector
(-assoc-n! [arr n val]

(.splice arr n 1 val)
arr))

(def a #js [1 2 3])  ;; => #js [1 2 3]
(assoc! a 0 42)    ;; => #js [42 2 3]
(assoc! a 1 43)    ;; => #js [42 43 3]
(assoc! a 2 44)    ;; => #js [42 43 44]

ITransientSet プロトコルを説明するために、ES6 の Set 型を拡張して、conj!、disj!、
persistent! をサポートします。以前に Set 型を拡張して ClojureScript に変換できるようにし
たことを思い出してください。

(extend-type js/Set
ITransientCollection
(-conj! [s v]

(.add s v)
s)

(-persistent! [s]
(js->clj s))

ITransientSet
(-disjoin! [s v]

(.delete s v)
s))

(def s (js/Set.))

(conj! s 1)
(conj! s 1)
(conj! s 2)
(conj! s 2)

(persistent! s)  ;; => #{1 2}

(disj! s 1)

(persistent! s)  ;; => #{2}

5.5 CSP (core.asyncを⽤いた場合)

CSP は Communicating Sequential Processes の略で、1978 年に C.A.R.Hoare によって開発
された並⾏システムを記述するための形式であり、チャネルを介したメッセージの受け渡しと同期
に基づく並⾏性のモデルです。CSP の背後にある理論的モデルの詳細については、この本では説明
しません。その代わりに、core.async が提供する並⾏性に関わるプリミティブに焦点を当てます。

core.async は ClojureScript core の⼀部ではありませんが、ライブラリとして実装されていま
す。⾔語の⼀部ではありませんが、広く使⽤されています。多くのライブラリは core.asynx のプリ
ミティブの上に構築されています。そのため、本書で取り上げる価値はあると思います。これは、
ClojureScript のマクロでコードを変換することにより実現できる構⽂的な抽象化の良い例でもあ
ります。このセクションで説明する例を実⾏するには、core.async をインストールする必要があり
ます。

チャンネル
チャンネルはベルトコンベアのようなもので、⼀度に 1 つの値を⼊れたり出したりできます。複

数の reader と writer を持つことができ、core.async の基本的なメッセージパッシングのメカニズ
ムです。これがどのように動作するかを確認するために、いくつかの操作を実⾏するチャンネルを
作成します。

(require ’[cljs.core.async :refer [chan put! take!]])

(enable-console-print!)

(def ch (chan))

(take! ch #(println "Got a value:" %))
;; => nil

;; there is a now a pending take operation,
;; let’s put something on the channel
(put! ch 42)
;; Got a value: 42
;; => 42

上記の例では、chan コンストラクタを使⽤してチャネル ch を作成しました。その後、チャンネ
ルで take の操作を実⾏して、take の操作が成功したとき、呼び出されるコールバックを提供しま
す。値をチャンネル上で設定するために put! を使った後、take の操作が完了して、"Got a value:
42" という⽂字列が出⼒されます。put! は、チャンネルに設定されたばかりの値を返します。

take! とは異なり、put! 関数はコールバックを受け取りませんが、前の例では何も提供していま
せん。put の場合、コールバックは指定した値が取得されるたびに呼び出されます。put と take は
任意の順序で発⽣しますが。ポイントを説明するために、いくつか put と take を実⾏してみます。

(require ’[cljs.core.async :refer [chan put! take!]])

(def ch (chan))

(put! ch 42 #(println "Just put 42")) ;; => true

(put! ch 43 #(println "Just put 43")) ;; => true

(take! ch #(println "Got" %))
;; Got 42
;; Just put 42
;; => nil

(take! ch #(println "Got" %))
;; Got 43
;; Just put 43
;; => nil

あなたはなぜ put! が true を返したか疑問に思っているかもしれません。まだ値が取得されて
いない場合でも、put の操作を実⾏できることを⽰します。チャンネルは閉じることができ、これ
により put の操作が失敗します。

(require ’[cljs.core.async :refer [chan put! close!]])

(def ch (chan))

(close! ch)  ;; => nil

(put! ch 42)  ;; => false

上記の例は最も単純な状況ですが、チャンネルが close されたとき、保留中の操作では何が起こ
るのでしょうか。いくつか take を実⾏してチャンネルを閉じて、何が起こるか⾒てみましょう。

(require ’[cljs.core.async :refer [chan put! take! close!]])

(def ch (chan))

(take! ch #(println "Got value:" %))  ;; => nil
(take! ch #(println "Got value:" %))  ;; => nil

(close! ch)
;; Got value: nil
;; Got value: nil
;; => nil

チャンネルが閉じている場合は、すべての take! の操作は nil の値を受け取ります。チャンネ
ルにおいて nil は、チャネルがクローズされたことを受信者に知らせるための指標値です。そのた
め、チャネルに nil の値を設定することはできません。

(require ’[cljs.core.async :refer [chan put!]])

(def ch (chan))

(put! ch nil)
;; Error: Assert failed: Can’t put nil in on a channel

■バッファ 保留中の take と put の操作は、チャンネルで enqueue されていますが、保留中の
take の操作や put の操作が多数ある場合はどうなるでしょうか。put と take の多いチャンネルを
叩いて確認してみましょう。

(require ’[cljs.core.async :refer [chan put! take!]])

(def ch (chan))

(dotimes [n 1025]
(put! ch n))

;; Error: Assert failed: No more than 1024
;; pending puts are allowed on a single channel.

(def ch (chan))

(dotimes [n 1025]
(take! ch #(println "Got" %)))

;; Error: Assert failed: No more than 1024
;; pending takes are allowed on a single channel.

上記の例が⽰しているように、チャンネルには保留中の put または take には制限があり、現在
は 1024ですが、これは変更される可能性のある実装の詳細です。チャンネルには保留中の put と
take の両⽅が存在することはできません。保留中の take がある場合、またはその逆の場合、put は
すぐに成功するためです。
チャンネルは put の操作のバッファリングをサポートします。バッファーを持つチャンネルを作
成する場合、バッファーに空きがあれば put の操作は即座に成功して、そうでなければキューに⼊
れられます。1 つの要素のバッファでチャンネルを作成するポイントを説明します。chan コンスト
ラクタは最初の引数として数値を受け取り、指定されたサイズのバッファをもちます。

(require ’[cljs.core.async :refer [chan put! take!]])

(def ch (chan 1))

(put! ch 42 #(println "Put succeeded!"))
;; Put succeeded!
;; => true

(dotimes [n 1024]
(put! ch n))

;; => nil

(put! ch 42)
;; Error: Assert failed: No more than 1024
;; pending puts are allowed on a single channel.

上の例では何が起こったのでしょうか。サイズ 1 のバッファーをもつチャンネルを作成して、そ
のチャンネルに対して put の操作を実⾏しましたが、値がバッファーに⼊れられたため、すぐに成
功しました。その後、保留中の put の queue を埋めるためにさらに put の操作を 1024 回⾏い、さ
らに 1 つの値を put しようとすると、チャンネルはそれ以上の put を enqueue できないと警告を
発しました。
チャンネルがどのように機能して、どのバッファがどのように使われるわかったところで、次に

core.async が実装している異なるバッファについて⾒ていきましょう。それぞれのバッファには異
なるポリシーがあり、それぞれのバッファがいつ何を使うべきかを知るのは興味深いことです。デ
フォルトでは、チャンネルはバッファされません。

■Fixed 固定サイズのバッファは、chan コンストラクタに数値を指定したときに作成され、その
数値で指定されたサイズになります。これは可能な限り最も単純なバッファで、いっぱいになると
put が queue に⼊れられます。

chan コンストラクタは、最初の引数として数値またはバッファのいずれかを受け⼊れます。次の
例で作成した 2 つのチャンネルは、どちらもサイズ 32 の固定バッファを使⽤します。

(require ’[cljs.core.async :refer [chan buffer]])

(def a-ch (chan 32))

(def another-ch (chan (buffer 32)))

■Dropping 固定バッファーを使⽤すると、put の操作を enqueue できます。しかし前に⾒たよ
うに、固定バッファがいっぱいになっても put は queue に⼊れられます。バッファーがいっぱいに
なったときに発⽣する put 操作を破棄したくない場合は、dropping バッファを使⽤できます。

dropping バッファのサイズは固定されており、いっぱいになると書き込みは完了しますが、その
値は破棄されます。例を⾒てみましょう。

(require ’[cljs.core.async :refer [chan dropping-buffer put! take!]])

(def ch (chan (dropping-buffer 2)))

(put! ch 40)
;; => true
(put! ch 41)
;; => true
(put! ch 42)
;; => true

(take! ch #(println "Got" %))
;; Got 40
;; => nil
(take! ch #(println "Got" %))
;; Got 41
;; => nil
(take! ch #(println "Got" %))
;; => nil

3 つの put の操作を実⾏して、3 つとも成功しましたが、チャンネルの dropping バッファのサ
イズ が 2 であるため、最初の 2 つの値だけが taker に渡されました。利⽤できる値がないため 3
番⽬の take が enqueue されているのがわかりますが、3 番⽬の出⼒値 (42)は破棄されています。

■スライド sliding バッファは、dropping バッファとは逆のポリシーを持ちます。許容値の put
が完了すると、最も古い値は新しい値に置き換えられます。sliding バッファは、最後の put だけを
処理して、古い値を破棄できる場合に便利です。

(require ’[cljs.core.async :refer [chan sliding-buffer put! take!]])

(def ch (chan (sliding-buffer 2)))

(put! ch 40)
;; => true
(put! ch 41)
;; => true
(put! ch 42)
;; => true

(take! ch #(println "Got" %))
;; Got 41
;; => nil
(take! ch #(println "Got" %))
;; Got 42
;; => nil
(take! ch #(println "Got" %))
;; => nil

3 つの put の操作を⾏い、そのうちの 3 つは成功しましたが、チャンネルの sliding バッファの
サイズが 2 であるため、最後の 2 つの値だけが取得側に渡されました。ご覧のとおり、最初の put
の値が破棄されてから値が使⽤できないため、3 番⽬の take が enqueue されます。

■Transducer transducer の節でも⾒たように、チャンネルに値を⼊れることは transducible な
プロセスと捉えることができます。つまり、チャンネルを作成して transducer を渡すことで、⼊⼒
値を変換してからチャンネルに⼊れることができます。
チャンネルを持つ transducer を使⽤したい場合、transducer によって修正される reducing 関数
がバッファの add 関数となるので、バッファを供給しなければなりません。バッファの add 関数
は、バッファと⼊⼒を受け取り、そのような⼊⼒を組み込みながらバッファを返すため reducing 関
数です。

(require ’[cljs.core.async :refer [chan put! take!]])

(def ch (chan 1 (map inc)))

(put! ch 41)  ;; => true

(take! ch #(println "Got" %))
;; Got 42
;; => nil

reducing 関数が reduce された値を返したとき、チャンネルはどうなるのでしょうか。チャンネ
ルを終了する概念はクローズされているため、reduce された値が検出されるとチャンネルはクロー
ズされます。

(require ’[cljs.core.async :refer [chan put! take!]])

(def ch (chan 1 (take 2)))

(take! ch #(println "Got" %))  ;; => nil
(take! ch #(println "Got" %))  ;; => nil
(take! ch #(println "Got" %))  ;; => nil

(put! ch 41)  
;; => true
(put! ch 42)
;; Got 41
;; => true
(put! ch 43)
;; Got 42
;; Got nil
;; => false

チャンネルに最⼤ 2 回の⼊⼒を許可するので、take、stateful、transducerを使⽤しました。
次に、チャンネルで 3 つの take の操作を実⾏して、値を受け取るのは 2 つだけだと予想します。
上の例でわかるように、3 番⽬の take はチャンネルが閉じていることを⽰す sentinel nil 値を取得
します。また、3 回⽬の put の操作は false を返して、失敗したことを⽰します。

■例外処理 バッファに値を追加するとして例外が発⽣した場合、core.asyncでは操作が失敗し、
例外がコンソールに記録されます。しかし、チャンネルのコンストラクターは 3 番⽬の引数 (例外
を処理する関数)を受け⼊れます。
例外ハンドラを持つチャンネルを作成すると、例外が発⽣するたびに例外ハンドラが呼び出され
ます。ハンドラが nil を返した場合は操作は暗黙的に失敗して、別の値を返した場合はその値で
add の操作が再試⾏されます。

(require ’[cljs.core.async :refer [chan put! take!]])

(enable-console-print!)

(defn exception-xform
[rfn]
(fn [acc input]

(throw (js/Error. "I fail!"))))

(defn handle-exception
[ex]
(println "Exception message:" (.-message ex))
42)

(def ch (chan 1 exception-xform handle-exception))

(put! ch 0)
;; Exception message: I fail!
;; => true

(take! ch #(println "Got:" %))
;; Got: 42
;; => nil

■Offerと Poll これまでに、チャンネルに関する 2 つの基本的な操作である put! と take!につ
いて学びました。それらは、値を取得または設定しますが、すぐに実⾏できない場合は enqueue さ
れます。どちらの関数もその性質的に⾮同期であり、すぐに成功することができますが、後で完了
することもできます。

core.async には値の設定と取得のために 2 つの同期的な操作があります。offer! と poll! で
す。それらがどのように機能するかを例を⾒てみましょう。

offer! はすぐに値を設定できる場合は、チャンネルに値を設定します。チャンネルが値を受け
取った場合は true を返し、それ以外の場合は false を返します。put! とは異なり、offer! は
閉じたチャンネルと開いたチャンネルを区別できない点に注意してください。

(require ’[cljs.core.async :refer [chan offer!]])

(def ch (chan 1))

(offer! ch 42)
;; => true

(offer! ch 43)
;; => false

poll! がすぐに値を取得できる場合は、チャンネルから値を取得します。成功した場合は値を返
し、それ以外の場合は nil を返します。take! とは異なり、poll! は閉じたチャンネルと開いた
チャンネルを区別できません。

(require ’[cljs.core.async :refer [chan offer! poll!]])

(def ch (chan 1))

(poll! ch)
;; => nil

(offer! ch 42)
;; => true

(poll! ch)
;; => 42

プロセス
チャンネルについてはすべて学びましたが、プロセスについては詳しく取り上げていませんでし
た。プロセスは、独⽴して実⾏されて、通信と調整のためにチャンネルを使⽤するロジックです。プ
ロセス内での put および take は、操作が完了するとプロセスが停⽌します。プロセスの停⽌によ
り ClojureScript が実⾏される環境にあるスレッドだけがブロックされるわけではありません。代
わりに、操作の実⾏を待機しているときに再開されます。
プロセスは go マクロを使って起動されて、put と take には <! および >! のプレースホルダを
つけます。go マクロはコールバックを使⽤するようにコードを書き換えますが、go の内部ではす
べてが同期的なコードに⾒えるため、理解が容易です。

(require ’[cljs.core.async :refer [chan <! >!]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(enable-console-print!)

(def ch (chan))

(go
(println [:a] "Gonna take from channel")
(println [:a] "Got" (<! ch)))

(go
(println [:b] "Gonna put on channel")
(>! ch 42)
(println [:b] "Just put 42"))

;; [:a] Gonna take from channel
;; [:b] Gonna put on channel
;; [:b] Just put 42
;; [:a] Got 42

上記の例では、ch から値を取得してコンソールに出⼒するプロセスを goで起動しています。値
がすぐに使⽤可能になるわけではないので、値が使⽤可能になるまで⼀時停⽌します。その後、チャ
ンネルに価値をもたらす別のプロセスを開始します。
保留中の take があるため、put の操作は成功します。値が最初のプロセスに渡されると、両⽅の
プロセスが終了します。
どちらの go ブロックも独⽴して実⾏されて⾮同期に実⾏されますが、同期的なコードに⾒えま

す。上記の go ブロックはかなり単純ですが、チャンネルを介して調整する並⾏プロセスを書けるこ
とは、複雑な⾮同期のワークフローを実装するために⾮常に強⼒なツールとなります。チャンネル
はまた、producer と consumer を分離します。
プロセスは任意の時間だけ待つことができ、与えられたミリ秒後に閉じられるチャンネルを返す

timeout 関数があります。go ブロック内で timeout チャンネルと take の操作を組み合わせると、
スリープさせることができます。

(require ’[cljs.core.async :refer [<! timeout]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(enable-console-print!)

(defn seconds
[]
(.getSeconds (js/Date.)))

(println "Launching go block")

(go
(println [:a] "Gonna take a nap" (seconds))
(<! (timeout 1000))
(println [:a] "I slept one second, bye!" (seconds)))

(println "Block launched")

;; Launching go block
;; Block launched
;; [:a] Gonna take a nap 9
;; [:a] I slept one second, bye! 10

出⼒されたメッセージからわかるように、プロセスは、timeout チャンネルの take の操作でブ
ロックされたときには、1 秒間何もしません。プログラムは続⾏されて、1 秒後にプロセスが再開す
るか終了します。

■選択 go ブロック内で⼀度に 1 つの値を設定や取得するだけでなく、alts! を使って複数のチャ
ンネル操作に⾮決定的な選択をすることもできます。⼀連のチャンネルの put または take の操作
が与えられ (チャンネルで put と take は同時に⾏うこともできます)、準備が整うとすぐに実⾏さ
れます。複数の操作が alts! を呼び出すときに実⾏できれば、擬似的なランダムの選択を⾏います。

timeout 関数と ats! を組み合わせることで、簡単にチャンネルの操作を試して、⼀定時間後に取
り消すことができます。その⽅法を⾒てみましょう。

(require ’[cljs.core.async :refer [chan <! timeout alts!]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(enable-console-print!)

(def ch (chan))

(go
(println [:a] "Gonna take a nap")
(<! (timeout 1000))
(println [:a] "I slept one second, trying to put a value on channel")
(>! ch 42)
(println [:a] "I’m done!"))

(go
(println [:b] "Gonna try taking from channel")
(let [cancel (timeout 300)

[value ch] (alts! [ch cancel])]
(if (= ch cancel)

(println [:b] "Too slow, take from channel cancelled")
(println [:b] "Got" value))))

;; [:a] Gonna take a nap
;; [:b] Gonna try taking from channel
;; [:b] Too slow, take from channel cancelled
;; [:a] I slept one second, trying to put a value on channel

上の例では、1 秒待ってから ch チャンネルに値を⼊れる go ブロックを起動しました。もう⼀
⽅の go ブロックは、300 ミリ秒後に閉じる cancel チャンネルを作成します。その後、alts! を
使って ch からの読み込みとキャンセルを同時に⾏おうとします。これらいずれかのチャンネルか
ら値を取得できる場合は、常に成功します。cancel は 300 ミリ秒後にクローズされて、閉じたチャ
ンネルからの take が  nil sentinel を返すため、alts! は成功します。alts! が 2 つの要素か
らなるベクタを返することに注⽬してください。それらは、操作による返り値と、実⾏されたチャ
ンネルを含みます。
このため、candel チャンネルで read の操作が⾏われたか、ch チャンネルで⾏われたかを検出
することができます。この例をコピーして、最初のプロセスの timeout を 100 ミリ秒に設定して、
ch チャンネルに対する read の操作がどのように成功するかを確認することをお勧めします。

read の操作間での選択⽅法を学んだので、alt! で条件付きの read の操作を表現する⽅法を⾒て
みましょう。チャンネルとその上に置こうとする値を提供する必要があるため、チャンネルと write
の操作を表す値を持つ 2 つの要素のベクタを使⽤します。

(require ’[cljs.core.async :refer [chan <! alts!]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(enable-console-print!)

(def a-ch (chan))
(def another-ch (chan))

(go
(println [:a] "Take a value from a-ch ")
(println [:a] "Got" (<! a-ch))
(println [:a] "I’m done!"))

(go
(println [:b] "Take a value from another-ch ")
(println [:a] "Got" (<! another-ch))
(println [:b] "I’m done!"))

(go
(println [:c] "Gonna try putting in both channels simultaneously")
(let [[value ch] (alts! [[a-ch 42]

[another-ch 99]])]
(if (= ch a-ch)

(println [:c] "Put a value in a-ch ")
(println [:c] "Put a value in another-ch "))))

;; [:a] Take a value from a-ch
;; [:b] Take a value from another-ch
;; [:c] Gonna try putting in both channels simultaneously
;; [:c] Put a value in a-ch
;; [:a] Got 42
;; [:a] I’m done!

上記の例を実⾏すると、a-ch チャンネルでの put の操作だけが成功します。両⽅のチャンネル
は、値を取得する準備ができています。このコードを実⾏すると異なる結果が得られることがあり
ます。

■優先順位 デフォルトで alt! は、複数の操作を実⾏する準備ができている場合は、常に⾮決定
的な選択を⾏いますが、代わりに alts! に :priority のオプションを渡す操作を優先させることがで
きます。:priority が true の場合、もし 複数の操作が準備できていれば、順番に試⾏されます。

(require ’[cljs.core.async :refer [chan >! alts!]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(enable-console-print!)

(def a-ch (chan))
(def another-ch (chan))

(go
(println [:a] "Put a value on a-ch ")
(>! a-ch 42)
(println [:a] "I’m done!"))

(go
(println [:b] "Put a value on another-ch ")
(>! another-ch 99)
(println [:b] "I’m done!"))

(go
(println [:c] "Gonna try taking from both channels with priority")
(let [[value ch] (alts! [a-ch another-ch] :priority true)]

(if (= ch a-ch)
(println [:c] "Got" value "from a-ch ")
(println [:c] "Got" value "from another-ch "))))

;; [:a] Put a value on a-ch
;; [:a] I’m done!
;; [:b] Put a value on another-ch
;; [:b] I’m done!
;; [:c] Gonna try taking from both channels with priority
;; [:c] Got 42 from a-ch

a-ch と another-ch は両⽅とも、いつ alts! を read するかの値を持っており、:priority オ
プションを true に設定すると a-ch が優先されます。:priority オプションを削除してこの例を
複数回実⾏すると、alts! は priority なしで⾮決定論的な選択をします。

■デフォルト alts! についてもう⼀つの興味深いことに、操作の準備ができておらず、デフォル
ト値を指定すれば、すぐに戻ることができます。いずれかの操作が準備できている場合に限り、条
件付きで操作を選択することができます。準備できていない場合にはデフォルト値を返します。

(require ’[cljs.core.async :refer [chan alts!]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(def a-ch (chan))

(def another-ch (chan))

(go
(println [:a] "Gonna try taking from any of the channels without blocking")
(let [[value ch] (alts! [a-ch another-ch] :default :not-ready)]

(if (and (= value :not-ready)
(= ch :default))

(println [:a] "No operation is ready, aborting")
(println [:a] "Got" value))))

;; [:a] Gonna try taking from any of the channels without blocking
;; [:a] No operation is ready, aborting

上の例でわかるように、操作の準備ができていない場合は、alts! による返される値は、呼び出
し時に :default キーの後に指定したものであり、channel は :default キーワード⾃体です。

コンビネーター
チャンネルとプロセスを理解したところで、次に、core.syanc に存在するチャンネルを扱うため
の興味深いコンビネーターについて検討します。このセクションでは、これらすべてについて簡単
に説明して、簡単な使⽤例を⽰します。

■パイプ pipe は⼊⼒チャンネルと出⼒チャンネルを取り、⼊⼒チャンネルに設定されたすべての
値を出⼒チャンネルに渡します。3 番⽬の引数に false を指定しない限り、ソースが閉じられるた
びに出⼒チャンネルが閉じられます。

(require ’[cljs.core.async :refer [chan pipe put! <! close!]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

(def in (chan))
(def out (chan))

(pipe in out)

(go-loop [value (<! out)]
(if (nil? value)

(println [:a] "I’m done!")
(do

(println [:a] "Got" value)
(println [:a] "Waiting for a value")
(recur (<! out)))))

(put! in 0) ;; => true
(put! in 1) ;; => true
(close! in)

;; [:a] Got 0 ;; [:a] Waiting for a value
;; [:a] Got 1 ;; [:a] Waiting for a value
;; [:a] I’m done!

上記の例では、go-loop マクロを使⽤して、out チャンネルが閉じられるまで値を繰り返し読み
取ります。in チャンネルを閉じると、out チャンネルも閉じられて go-loop が終了します。
■pipeline-async pipeline-async は ある数字を受け取りますが、その数字は、並列性の制御、
出⼒チャンネル、⾮同期の関数、⼊⼒チャンネルのためのものです。⾮同期の関数には 2 つの引数
があり、⼊⼒チャンネルに格納された値と、⾮同期操作の結果を格納するチャンネルのために使い
ます。result チャンネルは終了後に閉じられます。その数は、⼊⼒を使⽤して⾮同期関数を呼び出
すために使⽤される並⾏的に動作する go ブロックの数を制御します。
出⼒チャンネルは、⾮同期関数の呼び出しが完了するまでに要する時間に関係なく、⼊⼒チャン
ネルに相対的な順序で出⼒を受け取ります。これにはオプションの last パラメータがあり、⼊⼒
チャンネルが閉じられたときに出⼒チャンネルを閉じるかどうかを制御します。デフォルトは true
です。

(require ’[cljs.core.async :refer [chan pipeline-async put! <! close!]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

(def in (chan))
(def out (chan))
(def parallelism 3)

(defn wait-and-put [value ch]
(let [wait (rand-int 1000)]

(js/setTimeout (fn []
(println "Waiting" wait "miliseconds for value" value)
(put! ch wait)
(close! ch))

wait)))

(pipeline-async parallelism out wait-and-put in)

(go-loop [value (<! out)]
(if (nil? value)

(println [:a] "I’m done!")
(do

(println [:a] "Got" value)
(println [:a] "Waiting for a value")
(recur (<! out)))))

(put! in 1)
(put! in 2)
(put! in 3)
(close! in)
;; Waiting 164 miliseconds for value 3
;; Waiting 304 miliseconds for value 2
;; Waiting 908 miliseconds for value 1
;; [:a] Got 908
;; [:a] Waiting for a value
;; [:a] Got 304
;; [:a] Waiting for a value
;; [:a] Got 164
;; [:a] Waiting for a value
;; [:a] I’m done!

■pipeline pipeline pipeline-async に似ていますが、⾮同期の関数を取る代わりに transducer
を使います。 transducer は各⼊⼒に独⽴して適⽤されます。

(require ’[cljs.core.async :refer [chan pipeline put! <! close!]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

(def in (chan))
(def out (chan))
(def parallelism 3)

(pipeline parallelism out (map inc) in)

(go-loop [value (<! out)]
(if (nil? value)

(println [:a] "I’m done!")
(do

(println [:a] "Got" value)
(println [:a] "Waiting for a value")
(recur (<! out)))))

(put! in 1)
(put! in 2)
(put! in 3)
(close! in)

;; [:a] Got 2
;; [:a] Waiting for a value
;; [:a] Got 3
;; [:a] Waiting for a value
;; [:a] Got 4
;; [:a] Waiting for a value
;; [:a] I’m done!

■split split は述部とチャンネルを取り、2 つのチャンネルを持つベクタを返します。最初の
チャンネルは述部が true の値を受け取り、2 番⽬のチャンネルは述部が false の値を受け取ります。
オプションで 3 番⽬ (true チャンネル)と 4番⽬ (false チャンネル)の引数を使って、チャンネルの
バッファまたは数字を渡すことができます。

(require ’[cljs.core.async :refer [chan split put! <! close!]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

(def in (chan))
(def chans (split even? in))
(def even-ch (first chans))
(def odd-ch (second chans))

(go-loop [value (<! even-ch)]
(if (nil? value)

(println [:evens] "I’m done!")
(do

(println [:evens] "Got" value)
(println [:evens] "Waiting for a value")
(recur (<! even-ch)))))

(go-loop [value (<! odd-ch)]
(if (nil? value)

(println [:odds] "I’m done!")
(do

(println [:odds] "Got" value)
(println [:odds] "Waiting for a value")
(recur (<! odd-ch)))))

(put! in 0)
(put! in 1)
(put! in 2)
(put! in 3)
(close! in)
;; [:evens] Got 0
;; [:evens] Waiting for a value
;; [:odds] Got 1
;; [:odds] Waiting for a value
;; [:odds] Got 3
;; [:odds] Waiting for a value
;; [:evens] Got 2
;; [:evens] Waiting for a value
;; [:evens] I’m done!
;; [:odds] I’m done!

■reduce reduce は、reducing 関数、初期値、および input チャンネルをとります。指定された
初期値を seed として使⽤して、input チャンネルに設定されたすべての値に reduce を⾏い、閉じ
たチャンネルを返します。

(require ’[cljs.core.async :as async :refer [chan put! <! close!]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(def in (chan))

(go
(println "Result" (<! (async/reduce + (+) in))))

(put! in 0)
(put! in 1)
(put! in 2)
(put! in 3)
(close! in)
;; Result: 6

■onto-chan onto-chan はチャンネルとコレクションを取得して、コレクションの内容をチャン
ネルに格納します。終了後にチャンネルを閉じますが、チャンネルを閉じるかどうかを指定する 3
番⽬の引数を受け⼊れます。onto-chan を⽤いて reduce の例を書きかえてみましょう。

(require ’[cljs.core.async :as async :refer [chan put! <! close! onto-chan]])
(require-macros ’[cljs.core.async.macros :refer [go]])

(def in (chan))

(go
(println "Result" (<! (async/reduce + (+) in))))

(onto-chan in [0 1 2 3])

;; Result: 6

■to-chan to-chan はコレクションを取得して、コレクション内のすべての値を格納するチャン
ネルを返し、その後、チャンネルを閉じます。

(require ’[cljs.core.async :refer [chan put! <! close! to-chan]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

(def ch (to-chan (range 3)))

(go-loop [value (<! ch)]
(if (nil? value)

(println [:a] "I’m done!")
(do

(println [:a] "Got" value)
(println [:a] "Waiting for a value")
(recur (<! ch)))))

;; [:a] Got 0
;; [:a] Waiting for a value
;; [:a] Got 1
;; [:a] Waiting for a value
;; [:a] Got 2
;; [:a] Waiting for a value
;; [:a] I’m done!

■merge merge は⼊⼒チャンネルのコレクションを取得して、⼊⼒チャンネルに設定されたすべ
ての値を格納するチャンネルを返します。すべての⼊⼒チャンネルを閉じると、返されたチャンネ
ルが閉じます。返されるチャンネルはデフォルトではバッファされませんが、最後の引数として数
値またはバッファを指定できます。

(require ’[cljs.core.async :refer [chan put! <! close! merge]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

(def in1 (chan))
(def in2 (chan))
(def in3 (chan))

(def out (merge [in1 in2 in3]))

(go-loop [value (<! out)]
(if (nil? value)

(println [:a] "I’m done!")
(do

(println [:a] "Got" value)
(println [:a] "Waiting for a value")
(recur (<! out)))))

(put! in1 1)
(close! in1)
(put! in2 2)
(close! in2)
(put! in3 3)
(close! in3)

;; [:a] Got 3
;; [:a] Waiting for a value
;; [:a] Got 2
;; [:a] Waiting for a value
;; [:a] Got 1
;; [:a] Waiting for a value
;; [:a] I’m done!

⾼度な抽象化
core.async の低レベルなプリミティブと、それらがチャンネルと動作するために提供されてい
るコンビネータについて学びました。さらに core.async は、より⾼度な機能の構成要素として機
能するチャンネルの上に、役に⽴つ⾼レベルの抽象化を提供します。

■Mult 値を多くの他のチャンネルにブロードキャストする必要があるチャンネルがある場合は、
mult を使⽤して、指定されたチャンネルを複数作成できます。モジュールができると tap を⽤い
てチャンネルに接続し、untap してチャンネルを切り離します。また、mults は untap-all を使
⽤して、タップしたすべてのチャンネルを⼀度に削除することもできます。

mult のソースチャンネルに⼊⼒されたすべての値は、タップされたすべてのチャンネルにブロー
ドキャストされて、次のアイテムがブロードキャストされる前にすべてのチャンネルがそれを受け
⼊れる必要があります。遅い taker が mult の値をブロックするのを防ぐためには、タップしたチャ
ンネルのバッファリングを慎重に使⽤する必要があります。
クローズタップしたチャンネルは⾃動的にミュートから解除されます。まだタップがない状態で
ソースチャンネルに値を設定すると、その値はドロップされます。

161



第 5章 ClojureScript 発展編 5.5 CSP (core.asyncを⽤いた場合)

(require ’[cljs.core.async :refer [chan put! <! close! timeout mult tap]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

;; Source channel and mult
(def in (chan))
(def m-in (mult in))

;; Sink channels
(def a-ch (chan))
(def another-ch (chan))

;; Taker for a-ch
(go-loop [value (<! a-ch)]

(if (nil? value)
(println [:a] "I’m done!")
(do

(println [:a] "Got" value)
(recur (<! a-ch)))))

;; Taker for another-ch , which sleeps for 3 seconds between takes
(go-loop [value (<! another-ch)]

(if (nil? value)
(println [:b] "I’m done!")
(do

(println [:b] "Got" value)
(println [:b] "Resting 3 seconds")
(<! (timeout 3000))
(recur (<! another-ch)))))

;; Tap the two channels to the mult
(tap m-in a-ch)
(tap m-in another-ch)

;; See how the values are delivered to a-ch and another-ch
(put! in 1)
(put! in 2)

;; [:a] Got 1
;; [:b] Got 1
;; [:b] Resting for 3 seconds
;; [:a] Got 2
;; [:b] Got 2
;; [:b] Resting for 3 seconds

■Pub-sub mult について学んだ後で、mult の上に pub-sub の抽象化を実装する⽅法を想像し
てみてください。core.async はすでにこの機能を実装しています。

source チャンネルから mult を作成する代わりに、pub にチャンネル、またメッセージのトピッ
クを抽出するために使⽤する関数を指定して publicztion を作成します。

sub を持つ publicztion を subscribe して、subscribe したい publicztion、関⼼のあるトピック、
そのトピックを持つメッセージを配置するチャンネルを指定できます。1 つのチャンネルを複数の
トピックに subscribe できます。

unsub には、そのようなチャンネルをトピックから unsubscribe するための publication、トピッ
ク、チャンネルを指定できます。unsub-all には、publication とトピックを指定して、指定した
トピックからすべてのチャンネルを unsubscribeできます。

(require ’[cljs.core.async :refer [chan put! <! close! pub sub]])
(require-macros ’[cljs.core.async.macros :refer [go-loop]])

;; Source channel and publication
(def in (chan))
(def publication (pub in :action))

;; Sink channels
(def a-ch (chan))
(def another-ch (chan))

;; Channel with :increment action
(sub publication :increment a-ch)

(go-loop [value (<! a-ch)]
(if (nil? value)

(println [:a] "I’m done!")
(do

(println [:a] "Increment:" (inc (:value value)))
(recur (<! a-ch)))))

;; Channel with :double action
(sub publication :double another-ch)

(go-loop [value (<! another-ch)]
(if (nil? value)

(println [:b] "I’m done!")
(do

(println [:b] "Double:" (* 2 (:value value)))
(recur (<! another-ch)))))

;; See how values are delivered to a-ch and another-ch depending on their action
(put! in {:action :increment :value 98})
(put! in {:action :double :value 21})

;; [:a] Increment: 99
;; [:b] Double: 42

■Mixer core.async のセクションで学んだように、複数のチャンネルを 1 つに結合する merge

関数を使⽤できます。複数のチャンネルを merge する場合、⼊⼒チャンネルに⼊⼒されたすべての
値はマージされたチャンネルになります。ただし、⼊⼒チャンネルに⼊⼒された値が出⼒チャンネ
ルに出⼒されるようにするには、より細かく制御する必要がある場合があります。この場合、Mixer
が便利です。

core.async では、複数の⼊⼒チャンネルを 1 つの出⼒チャンネルに結合するために使⽤できる
Mixer の抽象化が提供されます。Mixer の⾯⽩いところは、特定の input チャンネルだけをミュー
ト、⼀時停⽌、再⽣できることです。

mix で出⼒チャンネルを指定して Mixer を作成できます。Mixerができたら、admix を使って⼊
⼒チャンネルを mix に追加したり、unmix を使って削除したり、unmix-all を使ってすべての⼊
⼒チャンネルを削除したりできます。
⼊⼒チャンネルの状態を制御するために、Mixer とチャンネルからその状態へのマップを与える

toggle 関数を使⽤します。マップは Mix の現在の状態とマージされるため、toggle を使⽤して
チャンネルを Mix に追加できます。チャンネルの状態はマップで、boolean にマップされる :mute

:pause :sole のキーを持つことができます
チャンネルの mute、pause、sole の意味を⾒てみましょう。

- ミュートされた⼊⼒チャンネルは、値を取得している間は出⼒チャンネルに転送されないことを
意味します。したがって、チャンネルがミュートされている間は、そのチャンネルに⼊⼒されてい
るすべての値が破棄されます。

- ⼀時停⽌された input チャンネルは、そこから値を取得しないことを意味します。つまり、チャ
ンネルに⼊⼒された値は出⼒チャンネルに転送されず、破棄されません。

- 1 つまたは複数のチャンネルを sole にすると、出⼒チャンネルは sole に設定されたチャンネル
の値だけを受け取ります。デフォルトでは、sole されていないチャンネルはミュートされますが、
sole されていないチャンネルをミュートするか⼀時停⽌するかは sole モードで決定できます。

多くの情報があったので、理解を深めるための例を⾒てみましょう。まず、アウトチャンネルの
Mixer を設定し、ミックスに 3 つの⼊⼒チャンネルを追加します。その後、出⼒チャンネルで受信
したすべての値を出⼒して、⼊⼒チャンネルへの制御を⾒てみましょう。

(require ’[cljs.core.async :refer [chan put! <! close! mix admix
unmix toggle solo-mode]])

(require-macros ’[cljs.core.async.macros :refer [go-loop]])

;; Output channel and mixer
(def out (chan))
(def mixer (mix out))

;; Input channels
(def in-1 (chan))
(def in-2 (chan))
(def in-3 (chan))

(admix mixer in-1)
(admix mixer in-2)
(admix mixer in-3)

;; Let’s listen to the out channel and print what we get from it
(go-loop [value (<! out)]

(if (nil? value)
(println [:a] "I’m done")
(do

(println [:a] "Got" value)
(recur (<! out)))))

164



第 5章 ClojureScript 発展編 5.5 CSP (core.asyncを⽤いた場合)

デフォルトでは、⼊⼒チャンネルに設定されたすべての値が出⼒チャンネルに設定されます。

(do
(put! in-1 1)
(put! in-2 2)
(put! in-3 3))

;; [:a] Got 1
;; [:a] Got 2
;; [:a] Got 3

n-2 チャンネルを⼀時停⽌し、すべての input チャンネルに値を⼊⼒して、in-2 を再開します。

(toggle mixer {in-2 {:pause true}})
;; => true

(do
(put! in-1 1)
(put! in-2 2)
(put! in-3 3))

;; [:a] Got 1
;; [:a] Got 3

(toggle mixer {in-2 {:pause false}})

;; [:a] Got 2

上記の例からわかるように、⼀時停⽌されたチャンネルに⼊⼒された値は破棄されません。⼊⼒
チャンネルに⼊⼒された値を破棄するには、ミュートする必要があります。次に例を⽰します。

(toggle mixer {in-2 {:mute true}})
;; => true

(do
(put! in-1 1)
;; out will never get this value since it’s discarded
(put! in-2 2)
(put! in-3 3))

;; [:a] Got 1
;; [:a] Got 3

(toggle mixer {in-2 {:mute false}})

in-2 チャンネルに値 2 を設定しました。その時点でチャンネルがミュートされていたため、この
値が破棄されて out に⼊ることはありません。チャンネルが Mixer の中にある 3 番⽬の状態 sole
を⾒てみましょう。
前述したように、ミキサーのチャンネルを soleにすると、デフォルトで残りのチャンネルがミュー

トされます。

(toggle mixer {in-1 {:solo true}
in-2 {:solo true}})

;; => true

(do
(put! in-1 1)
(put! in-2 2)
;; out will never get this value since it’s discarded
(put! in-3 3))

;; [:a] Got 1
;; [:a] Got 2

(toggle mixer {in-1 {:solo false}
in-2 {:solo false}})

ただし、sole のチャンネルがある間は、sole でないチャンネルのモードを設定できます。デフォ
ルトの⾮ sole モードを、デフォルトの mute ではなく pause に設定します。

(solo-mode mixer :pause)
;; => true
(toggle mixer {in-1 {:solo true}

in-2 {:solo true}})
;; => true

(do
(put! in-1 1)
(put! in-2 2)
(put! in-3 3))

;; [:a] Got 1
;; [:a] Got 2

(toggle mixer {in-1 {:solo false}
in-2 {:solo false}})

;; [:a] Got 3

